

Lógica con Lean

José A. Alonso Jiménez

Grupo de Lógica Computacional
Dpto. de Ciencias de la Computación e Inteligencia Artificial
Universidad de Sevilla
Sevilla, 7 de noviembre de 2020

Esta obra está bajo una licencia Reconocimiento-NoComercial-CompartirIgual 2.5 Spain de Creative Commons.

Se permite:

- copiar, distribuir y comunicar públicamente la obra
- hacer obras derivadas

Bajo las condiciones siguientes:

Reconocimiento. Debe reconocer los créditos de la obra de la manera especificada por el autor.



No comercial. No puede utilizar esta obra para fines comerciales.



Compartir bajo la misma licencia. Si altera o transforma esta obra, o genera una obra derivada, sólo puede distribuir la obra generada bajo una licencia idéntica a ésta.

- Al reutilizar o distribuir la obra, tiene que dejar bien claro los términos de la licencia de esta obra.
- Algunas de estas condiciones pueden no aplicarse si se obtiene el permiso del titular de los derechos de autor.

Esto es un resumen del texto legal (la licencia completa). Para ver una copia de esta licencia, visite <http://creativecommons.org/licenses/by-nc-sa/2.5/es/> o envíe una carta a Creative Commons, 559 Nathan Abbott Way, Stanford, California 94305, USA.

Índice general

1	Introducción	7
2	Lógica proposicional	9
2.1	Reglas del condicional	9
2.1.1	Regla de eliminación del condicional en $P \rightarrow Q, P \vdash Q$	9
2.1.2	Pruebas de $P, P \rightarrow Q, P \rightarrow (Q \rightarrow R) \vdash R$	11
2.1.3	Regla de introducción del condicional en $P \rightarrow P$	12
2.1.4	Pruebas de $P \rightarrow (Q \rightarrow P)$	14
2.1.5	Pruebas del silogismo hipotético: $P \rightarrow Q, Q \rightarrow R \vdash P \rightarrow R$	16
2.2	Reglas de la conjunción	18
2.2.1	Reglas de la conjunción en $P \wedge Q, R \vdash Q \wedge R$	18
2.2.2	Pruebas de $P \wedge Q \rightarrow Q \wedge P$	22
2.3	Reglas de la negación	25
2.3.1	Reglas de la negación con $(\perp \vdash P), (P, \neg P \vdash \perp)$ y $\neg(P \wedge \neg P)$	25
2.3.2	Pruebas de $P \rightarrow Q, P \rightarrow \neg Q \vdash \neg P$	28
2.3.3	Pruebas del modus tollens: $P \rightarrow Q, \neg Q \vdash \neg P$	31
2.3.4	Pruebas de $P \rightarrow (Q \rightarrow R), P, \neg R \vdash \neg Q$	33
2.3.5	Pruebas de $P \rightarrow Q \vdash \neg Q \rightarrow \neg P$	36
2.3.6	Regla de introducción de la doble negación: $P \vdash \neg\neg P$	38
2.3.7	Pruebas de $\neg Q \rightarrow \neg P \vdash P \rightarrow \neg\neg Q$	40
2.4	Reglas de la disyunción	43
2.4.1	Reglas de introducción de la disyunción	43
2.4.2	Regla de eliminación de la disyunción	47
2.4.3	Pruebas de $P \vee Q \vdash Q \vee P$	48
2.4.4	Pruebas de $Q \rightarrow R \vdash P \vee Q \rightarrow P \vee R$	59
2.4.5	Pruebas de $\neg P \vee Q \vdash P \rightarrow Q$	62
2.5	Reglas del bicondicional	66
2.5.1	Regla de introducción del bicondicional en $P \wedge Q \leftrightarrow Q \wedge P$	66
2.5.2	Reglas de eliminación del bicondicional en $P \leftrightarrow Q, P \vee Q \vdash P \wedge Q$	69

2.6 Reglas de la lógica clásica	72
2.6.1 Pruebas de la regla de reducción al absurdo	72
2.6.2 Pruebas de la eliminación de la doble negación	74
2.6.3 Pruebas del principio del tercio excluso	75
2.6.4 Pruebas de $P \rightarrow Q \vdash \neg P \vee Q$	79
2.6.5 Pruebas de $P, \neg\neg(Q \wedge R) \vdash \neg\neg P \wedge R$	83
2.6.6 Pruebas de $\neg P \rightarrow Q, \neg Q \vdash P$	85
2.6.7 Pruebas de $(Q \rightarrow R) \rightarrow ((\neg Q \rightarrow \neg P) \rightarrow (P \rightarrow R))$	87
3 Lógica de primer orden	93
3.1 Reglas del cuantificador universal	93
3.1.1 Regla de eliminación del cuantificador universal	93
3.1.2 Regla de introducción del cuantificador universal	95
3.2 Reglas del cuantificador existencial	97
3.2.1 Regla de introducción del cuantificador existencial	97
3.2.2 Regla de eliminación del cuantificador existencial	98
3.3 Ejercicios sobre cuantificadores	102
3.3.1 Pruebas de $\neg\forall x P(x) \leftrightarrow \exists x \neg P(x)$	102
3.3.2 Pruebas de $\forall x (P(x) \wedge Q(x)) \leftrightarrow \forall x P(x) \wedge \forall x Q(x)$	108
3.3.3 Pruebas de $\exists x (P(x) \vee Q(x)) \leftrightarrow \exists x P(x) \vee \exists x Q(x)$	114
3.3.4 Pruebas de $\exists x \exists y P(x,y) \leftrightarrow \exists y \exists x P(x,y)$	120
3.4 Reglas de la igualdad	124
3.4.1 Regla de eliminación de la igualdad	124
3.4.2 Pruebas de la transitividad de la igualdad	126
3.4.3 Regla de introducción de la igualdad	128
3.4.4 Pruebas de $y = x \rightarrow y = z \rightarrow x = z$	130
3.4.5 Pruebas de $(x + y) + z = (x + z) + y$	133
3.4.6 Pruebas de desarrollo de producto de sumas	135
4 Conjuntos	137
4.1 Elementos básicos sobre conjuntos	137
4.1.1 Pruebas de la reflexividad de la inclusión de conjuntos	137
4.1.2 Pruebas de la antisimetría de la inclusión de conjuntos	139
4.1.3 Introducción de la intersección	141
4.1.4 Introducción de la unión	142
4.1.5 El conjunto vacío	144
4.1.6 Diferencia de conjuntos: $A \setminus B \subseteq A$	146
4.1.7 Complementario de un conjunto: Pruebas de $A \setminus B \subseteq B^c$	148
4.1.8 Pruebas de la conmutatividad de la intersección	150

4.2	Identities conjuntistas	153
4.2.1	Pruebas de la propiedad distributiva de la intersección sobre la unión.	153
4.2.2	Pruebas de $(A \cap B^c) \cup B = A \cup B$	157
4.3	Familias de conjuntos	159
4.3.1	Unión e intersección de familias de conjuntos	159
4.3.2	Pertenencia a uniones e intersecciones de familias	161
4.3.3	Pruebas de la distributiva de la intersección general sobre la intersección	162
4.3.4	Reglas de la intersección general	164
4.3.5	Reglas de la unión general	165
4.3.6	Pruebas de intersección sobre unión general	167
4.3.7	Pruebas de $(\bigcup_i, \bigcap_j, A \cap j) \subseteq (\bigcap_j, \bigcup_i, A \cap j)$	169
4.4	Conjunto potencia	171
4.4.1	Definición del conjunto potencia	171
4.4.2	Pruebas de $A \in \wp(A \cup B)$	171
4.4.3	Monotonía del conjunto potencia: $\wp A \subseteq \wp B \leftrightarrow A \subseteq B$	172
5	Relaciones	177
5.1	Relaciones de orden	177
5.1.1	Las irreflexivas y transitivas son asimétricas	177
5.1.2	Las partes estrictas son irreflexivas	178
5.1.3	Las partes estrictas de los órdenes parciales son transitivas	179
5.1.4	Las partes simétricas de las reflexivas son reflexivas	181
5.1.5	Las partes simétricas son simétricas	182
5.2	Órdenes sobre números	184
5.2.1	Pruebas de $n + 1 \leq m \vdash n < m + 1$	184
5.3	Relaciones de equivalencia	186
5.3.1	Las equivalencias son preórdenes simétricos	186
5.3.2	Las relaciones reflexivas y euclídeas son de equivalencia	187
6	Funciones	191
6.1	Funciones en Lean	191
6.1.1	Definición de la composición de funciones	191
6.1.2	Definición de la función identidad	192
6.1.3	Extensionalidad funcional	192
6.1.4	Propiedades de la composición de funciones (elemento neutro y asociatividad)	193
6.1.5	Funciones inyectivas, suprayectivas y biyectivas	197
6.1.6	La identidad es biyectiva	198
6.1.7	La composición de funciones inyectivas es inyectiva	200

6.1.8 La composición de funciones suprayectivas es suprayectiva	203
6.1.9 La composición de funciones biyectivas es biyectiva	205

Capítulo 1

Introducción

El objetivo de este trabajo es presentar una introducción a la Lógica usando [Lean](#) para usarla en las clases de la asignatura de [Razonamiento automático](#) del [Máster Universitario en Lógica, Computación e Inteligencia Artificial](#) de la Universidad de Sevilla. Por tanto, el único prerequisite es, como en el Máster, cierta madurez matemática como la que deben tener los alumnos de los Grados de Matemática y de Informática.

El trabajo se basa fundamentalmente en

- El [curso de "Lógica matemática y fundamentos"](#) en que se estudia la deducción natural proposicional y de primer orden (basado en el libro [Logic in computer science: Modelling and reasoning about systems](#) de Michael Huth y Mark Ryan) y su formalización en [Isabelle/HOL](#).
- Los apuntes de [Lógica y demostración con Lean](#) que son un resumen del libro [Logic and Proof](#) de Jeremy Avigad, Robert Y. Lewis y Floris van Doorn.
- Los apuntes [Deducción natural en Lean](#) en el que se presentan ejemplos de uso de las tácticas de Lean correspondientes a las reglas de la deducción natural.
- Los apuntes [Matemáticas en Lean](#) en el que se presentan la formalización en Lean de temas básicos de las matemáticas usando las librerías de [mathlib](#). Está basado en el libro [Mathematics in Lean](#) de Jeremy Avigad, Kevin Buzzard, Robert Y. Lewis y Patrick Massot.

La exposición se hará mediante una colección de ejercicios. En cada ejercicio se mostrarán distintas pruebas del mismo resultado y se comentan las tácticas conforme se van usando y los lemas utilizados en las demostraciones.

Además, en cada ejercicio hay tres enlaces: uno al código, otro que al pulsarlo abre el ejercicio en Lean Web (en una sesión del navegador) de forma

que se puede navegar por las pruebas y editar otras alternativas, y el tercero es un enlace a un vídeo explicando las soluciones del ejercicio.

El trabajo se desarrolla como un [proyecto en GitHub](#) que contiene [libro en PDF](#). Además, los vídeos correspondientes a cada uno de los ejercicios se encuentran en [YouTube](#).

Capítulo 2

Lógica proposicional

2.1. Reglas del condicional

2.1.1. Regla de eliminación del condicional en $P \rightarrow Q, P \vdash Q$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Eliminación del condicional en Lean
-- =====

-- Ej. 1. Demostrar que
--      (P → Q), P ⊢ Q.

import tactic
variables (P Q : Prop)

-- 1ª demostración
example
  (h1 : P → Q)
  (h2 : P)
  : Q :=
begin
  apply h1,
  exact h2,
end

-- 2ª demostración
example
  (h1 : P → Q)
```

```
(h2 : P)
: Q :=
begin
  exact h1 h2,
end

-- 3ª demostración
example
  (h1 : P → Q)
  (h2 : P)
  : Q :=
by exact h1 h2

-- 4ª demostración
example
  (h1 : P → Q)
  (h2 : P)
  : Q :=
h1 h2

-- 5ª demostración
example
  (h1 : P → Q)
  (h2 : P)
  : Q :=
by tauto

-- 6ª demostración
example
  (h1 : P → Q)
  (h2 : P)
  : Q :=
by finish

-- 7ª demostración
example
  (h1 : P → Q)
  (h2 : P)
  : Q :=
by solve_by_elim
```

2.1.2. Pruebas de $P, P \rightarrow Q, P \rightarrow (Q \rightarrow R) \vdash R$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Pruebas de  $P, P \rightarrow Q, P \rightarrow (Q \rightarrow R) \vdash R$ 
-- =====
```

```
-- Ej 1. Demostrar que
--  $P, P \rightarrow Q, P \rightarrow (Q \rightarrow R) \vdash R$ 
```

```
import tactic
```

```
variables (P Q R : Prop)
```

```
-- 1ª demostración
```

```
example
```

```
  (h1 : P)
  (h2 : P → Q)
  (h3 : P → (Q → R))
  : R :=
```

```
have h4 : Q,
```

```
  from h2 h1,
```

```
have h5 : Q → R,
```

```
  from h3 h1,
```

```
show R,
```

```
  from h5 h4
```

```
-- 2ª demostración
```

```
example
```

```
  (h1 : P)
  (h2 : P → Q)
  (h3 : P → (Q → R))
  : R :=
```

```
have h4 : Q      := h2 h1,
```

```
have h5 : Q → R := h3 h1,
```

```
show R, from h5 h4
```

```
-- 3ª demostración
```

```
example
```

```
  (h1 : P)
  (h2 : P → Q)
  (h3 : P → (Q → R))
  : R :=
```

```
show R, from (h3 h1) (h2 h1)
```

```

-- 4ª demostración
example
  (h1 : P)
  (h2 : P → Q)
  (h3 : P → (Q → R))
  : R :=
(h3 h1) (h2 h1)

-- 5ª demostración
example
  (h1 : P)
  (h2 : P → Q)
  (h3 : P → (Q → R))
  : R :=
by finish

```

2.1.3. Regla de introducción del condicional en $P \rightarrow P$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Introducción del condicional en Lean
-- =====

-- Ej. 1. Demostrar que
--      P → P

import tactic
variable (P : Prop)

-- 1ª demostración
example : P → P :=
assume h : P,
show P, from h

-- 2ª demostración
example : P → P :=
assume : P,
show P, from this

-- 3ª demostración
example : P → P :=
assume : P,
show P, from <P>

```

```
-- 4ª demostración
example : P → P :=
assume h : P, h

-- 5ª demostración
example : P → P :=
λ h, h

-- 6ª demostración
example : P → P :=
id

-- 7ª demostración
example : P → P :=
begin
  intro h,
  exact h,
end

-- 8ª demostración
example : P → P :=
begin
  intro,
  exact <P>,
end

-- 9ª demostración
example : P → P :=
begin
  intro h,
  assumption,
end

-- 10ª demostración
example : P → P :=
begin
  intro,
  assumption,
end

-- 11ª demostración
example : P → P :=
by tauto
```

```
-- 12ª demostración
example : P → P :=
by finish

-- 13ª demostración
example : P → P :=
by simp
```

2.1.4. Pruebas de $P \rightarrow (Q \rightarrow P)$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Pruebas de  $P \rightarrow (Q \rightarrow P)$ 
-- =====

-- Ej. 1. Demostrar
--       $P \rightarrow (Q \rightarrow P)$ 

import tactic

variables (P Q : Prop)

-- 1ª demostración
example : P → (Q → P) :=
assume (h1 : P),
show Q → P, from
  ( assume h2 : Q,
    show P, from h1)

-- 2ª demostración
example : P → (Q → P) :=
assume (h1 : P),
show Q → P, from
  ( assume h2 : Q, h1)

-- 3ª demostración
example : P → (Q → P) :=
assume (h1 : P),
show Q → P, from
  ( λ h2, h1)

-- 4ª demostración
example : P → (Q → P) :=
```

```
assume (h1 : P), (λ h2, h1)

-- 5ª demostración
example : P → (Q → P) :=
λ h1, λ h2, h1

-- 6ª demostración
example : P → (Q → P) :=
λ h1 h2, h1

-- 7ª demostración
example : P → (Q → P) :=
λ h _, h

-- 8ª demostración
example : P → (Q → P) :=
imp_intro

-- 9ª demostración
example : P → (Q → P) :=
begin
  intro h1,
  intro h2,
  exact h1,
end

-- 10ª demostración
example : P → (Q → P) :=
begin
  intros h1 h2,
  exact h1,
end

-- 6ª demostración
example : P → (Q → P) :=
λ h1 h2, h1

-- 11ª demostración
example : P → (Q → P) :=
by tauto

-- 12ª demostración
example : P → (Q → P) :=
by finish
```

2.1.5. Pruebas del silogismo hipotético: $P \rightarrow Q, Q \rightarrow R \vdash P \rightarrow R$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Pruebas del silogismo hipotético
```

```
-- -----
```

```
import tactic
```

```
variables (P Q R : Prop)
```

```
-- Ej. 1. Demostrar que
```

```
--    $P \rightarrow Q, Q \rightarrow R \vdash P \rightarrow R$ 
```

```
-- 1ª demostración
```

```
example
```

```
  (h1 : P → Q)
```

```
  (h2 : Q → R)
```

```
  : P → R :=
```

```
assume h : P,
```

```
have h3 : Q,
```

```
  from h1 h,
```

```
show R,
```

```
  from h2 h3
```

```
-- 2ª demostración
```

```
example
```

```
  (h1 : P → Q)
```

```
  (h2 : Q → R)
```

```
  : P → R :=
```

```
assume h : P,
```

```
have h3 : Q := h1 h,
```

```
show R,
```

```
  from h2 h3
```

```
-- 3ª demostración
```

```
example
```

```
  (h1 : P → Q)
```

```
  (h2 : Q → R)
```

```
  : P → R :=
```

```
assume h : P,
```


```
show R,
```

```
  from h2 (h1 h)
```



```
-- 4º demostración
example
  (h1 : P → Q)
  (h2 : Q → R)
  : P → R :=
assume h : P, h2 (h1 h)

-- 5º demostración
example
  (h1 : P → Q)
  (h2 : Q → R)
  : P → R :=
λ h, h2 (h1 h)

-- 6º demostración
example
  (h1 : P → Q)
  (h2 : Q → R)
  : P → R :=
h2  h1


-- 7º demostración
example
  (h1 : P → Q)
  (h2 : Q → R)
  : P → R :=
begin
  intro h,
  apply h2,
  apply h1,
  exact h,
end

-- 8º demostración
example
  (h1 : P → Q)
  (h2 : Q → R)
  : P → R :=
begin
  intro h,
  apply h2,
  exact h1 h,
end
```

```

-- 9º demostración
example
  (h1 : P → Q)
  (h2 : Q → R)
  : P → R :=
begin
  intro h,
  exact h2 (h1 h),
end

-- 10º demostración
example
  (h1 : P → Q)
  (h2 : Q → R)
  : P → R :=
λ h, h2 (h1 h)

-- 11º demostración
example
  (h1 : P → Q)
  (h2 : Q → R)
  : P → R :=
h2  h1

-- 12º demostración
example
  (h1 : P → Q)
  (h2 : Q → R)
  : P → R :=
by tauto

-- 13º demostración
example
  (h1 : P → Q)
  (h2 : Q → R)
  : P → R :=
by finish

```

2.2. Reglas de la conjunción

2.2.1. Reglas de la conjunción en $P \wedge Q, R \vdash Q \wedge R$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Reglas de la conjunción
-- =====
```

```
-- Ej. 1. Demostrar que
--       $P \wedge Q, R \vdash Q \wedge R$ 
```

```
import tactic
```

```
variables (P Q R : Prop)
```

```
-- 1ª demostración
-- =====
```

```
example
  (hPQ : P  $\wedge$  Q)
  (hR : R)
  : Q  $\wedge$  R :=
have hQ : Q,
  from and.right hPQ,
show Q  $\wedge$  R,
  from and.intro hQ hR
```

```
-- 2ª demostración
-- =====
```

```
example
  (hPQ : P  $\wedge$  Q)
  (hR : R)
  : Q  $\wedge$  R :=
have hQ : Q,
  from hPQ.right,
show Q  $\wedge$  R,
  from ⟨hQ, hR⟩
```

```
-- 3ª demostración
-- =====
```

```
example
  (hPQ : P  $\wedge$  Q)
  (hR : R)
  : Q  $\wedge$  R :=
have hQ : Q,
  from hPQ.2,
show Q  $\wedge$  R,
  from ⟨hQ, hR⟩
```

```
-- 4ª demostración
-- =====
```

```
example
  (hPQ : P ∧ Q)
  (hR : R)
  : Q ∧ R :=
have hQ : Q :=
  hPQ.2,
show Q ∧ R,
  from ⟨hQ, hR⟩
```

```
-- 5ª demostración
-- =====
```

```
example
  (hPQ : P ∧ Q)
  (hR : R)
  : Q ∧ R :=
show Q ∧ R,
  from ⟨hPQ.2, hR⟩
```

```
-- 6ª demostración
-- =====
```

```
example
  (hPQ : P ∧ Q)
  (hR : R)
  : Q ∧ R :=
⟨hPQ.2, hR⟩
```

```
-- 7ª demostración
-- =====
```

```
example
  (hPQ : P ∧ Q)
  (hR : R)
  : Q ∧ R :=
match hPQ with ⟨hP, hQ⟩ :=
  and.intro hQ hR
end
```

```
-- 8ª demostración
-- =====
```

```

example
  (hPQ : P ∧ Q)
  (hR : R)
  : Q ∧ R :=
begin
  split,
  { cases hPQ with hP hQ,
    clear hP,
    exact hQ, },
  { exact hR, },
end

```

```

-- 9ª demostración
-- =====

```

```

example
  (hPQ : P ∧ Q)
  (hR : R)
  : Q ∧ R :=
begin
  split,
  { cases hPQ,
    assumption, },
  { assumption, },
end

```

```

-- 10ª demostración
-- =====

```

```

example
  (hPQ : P ∧ Q)
  (hR : R)
  : Q ∧ R :=
begin
  constructor,
  { cases hPQ,
    assumption, },
  { assumption, },
end

```

```

-- 11ª demostración
-- =====

```

```

example

```

```

(hPQ : P ∧ Q)
(hR : R)
: Q ∧ R :=
by tauto

-- 12ª demostración
-- =====

example
  (hPQ : P ∧ Q)
  (hR : R)
  : Q ∧ R :=
by finish

```

2.2.2. Pruebas de $P \wedge Q \rightarrow Q \wedge P$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas en Lean de  $P \wedge Q \rightarrow Q \wedge P$ 
-- =====

-- Ej. 1. Demostrar que
--    $P \wedge Q \rightarrow Q \wedge P$ 

import tactic

variables (P Q : Prop)

-- 1ª demostración
-- =====

example : P ∧ Q → Q ∧ P :=
assume h : P ∧ Q,
have hP : P,
  from and.left h,
have hQ : Q,
  from and.right h,
show Q ∧ P,
  from and.intro hQ hP

-- 2ª demostración
-- =====

```

```

example : P ∧ Q → Q ∧ P :=
assume h : P ∧ Q,
have hP : P,
  from h.left,
have hQ : Q,
  from h.right,
show Q ∧ P,
  from ⟨hQ, hP⟩

-- 3ª demostración
-- =====

example : P ∧ Q → Q ∧ P :=
assume h : P ∧ Q,
have hP : P,
  from h.1,
have hQ : Q,
  from h.2,
show Q ∧ P,
  from ⟨hQ, hP⟩

-- 4ª demostración
-- =====

example : P ∧ Q → Q ∧ P :=
assume h : P ∧ Q,
have hP : P := h.1,
have hQ : Q := h.2,
show Q ∧ P,
  from ⟨hQ, hP⟩

-- 5ª demostración
-- =====

example : P ∧ Q → Q ∧ P :=
assume h : P ∧ Q,
show Q ∧ P,
  from ⟨h.2, h.1⟩

-- 6ª demostración
-- =====

example : P ∧ Q → Q ∧ P :=
assume h : P ∧ Q, ⟨h.2, h.1⟩

```

```

-- 7ª demostración
-- =====

example : P ∧ Q → Q ∧ P :=
λ h, ⟨h.2, h.1⟩

-- 8ª demostración
-- =====

example : P ∧ Q → Q ∧ P :=
and.comm.mp

-- 9ª demostración
-- =====

example : P ∧ Q → Q ∧ P :=
begin
  intro h,
  cases h with hP hQ,
  split,
  { exact hQ, },
  { exact hP, },
end

-- 10ª demostración
-- =====

example : P ∧ Q → Q ∧ P :=
begin
  rintro ⟨hP, hQ⟩,
  exact ⟨hQ, hP⟩,
end

-- 11ª demostración
-- =====

example : P ∧ Q → Q ∧ P :=
λ ⟨hP, hQ⟩, ⟨hQ, hP⟩

-- 12ª demostración
-- =====

example : P ∧ Q → Q ∧ P :=
by tauto

```



```
-- 13ª demostración
-- =====

example : P ∧ Q → Q ∧ P :=
by finish
```

2.3. Reglas de la negación

2.3.1. Reglas de la negación con $(\perp \vdash P)$, $(P, \neg P \vdash \perp)$ y $\neg(P \wedge \neg P)$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Reglas de la negación
-- =====

import tactic

variable (P : Prop)

-- Eliminación del falso
-- -----

-- Ej. 1. Demostrar que
--    $\perp \vdash P$ 

-- 1ª demostración
example
  (h : false)
  : P :=
false.elim h

-- 2ª demostración
example
  (h : false)
  : P :=
false.rec P h

-- 3ª demostración
example
  (h : false)
```

```

: P :=
by tauto

-- 4ª demostración
example
  (h : false)
  : P :=
by cases h

-- 5ª demostración
example
  (h : false)
  : P :=
by finish

-- 6ª demostración
example
  (h : false)
  : P :=
by solve_by_elim

-- Definición de la negación
-- -----

--  $\neg P \equiv (P \rightarrow \text{false})$ 

-- Eliminación de la negación
-- -----

-- Ej. 2. Demostrar que
--  $P, \neg P \vdash \perp$ 

-- 1ª demostración
example
  (h1: P)
  (h2:  $\neg P$ )
  : false :=
not.elim h2 h1

-- 2ª demostración
example
  (h1: P)
  (h2:  $\neg P$ )
  : false :=
h2 h1

```

```

-- Introducción de la negación
-- -----

-- Ej. 3. Demostrar
--    $\neg(P \wedge \neg P)$ 

-- 1ª demostración
example :  $\neg(P \wedge \neg P)$  :=
not.intro
  ( assume h : P  $\wedge$   $\neg P$ ,
    have h1 : P := h.1,
    have h2 :  $\neg P$  := h.2,
    show false, from h2 h1 )

-- 2ª demostración
example :  $\neg(P \wedge \neg P)$  :=
not.intro
  ( assume h : P  $\wedge$   $\neg P$ ,
    show false, from h.2 h.1 )

-- 3ª demostración
example :  $\neg(P \wedge \neg P)$  :=
not.intro
  ( assume h : P  $\wedge$   $\neg P$ , h.2 h.1 )

-- 4ª demostración
example :  $\neg(P \wedge \neg P)$  :=
not.intro ( $\lambda$  h, h.2 h.1)

-- 5ª demostración
example :  $\neg(P \wedge \neg P)$  :=
begin
  intro h,
  cases h with h1 h2,
  apply h2,
  exact h1,
end

-- 6ª demostración
example :  $\neg(P \wedge \neg P)$  :=
begin
  rintro ⟨h1, h2⟩,
  exact h2 h1,
end

```

```

-- 7ª demostración
example :  $\neg(P \wedge \neg P)$  :=
 $\lambda \langle h1, h2 \rangle, h2\ h1$ 

-- 8ª demostración
example :  $\neg(P \wedge \neg P)$  :=
(and_not_self P).mp

-- 9ª demostración
example :  $\neg(P \wedge \neg P)$  :=
by tauto

-- 10ª demostración
example :  $\neg(P \wedge \neg P)$  :=
by finish

-- 11ª demostración
example :  $\neg(P \wedge \neg P)$  :=
by simp

```

2.3.2. Pruebas de $P \rightarrow Q$, $P \rightarrow \neg Q \vdash \neg P$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de  $P \rightarrow Q$ ,  $P \rightarrow \neg Q \vdash \neg P$ 
-- =====

import tactic

variables (P Q : Prop)

-- Ej. 1. Demostrar
--    $P \rightarrow Q, P \rightarrow \neg Q \vdash \neg P$ 

-- 1ª demostración
example
  (h1 :  $P \rightarrow Q$ )
  (h2 :  $P \rightarrow \neg Q$ )
  :  $\neg P$  :=
assume h : P,
have h4 : Q,
  from h1 h,

```

```

have h5 : ¬Q,
  from h2 h,
show false,
  from h5 h4

-- 2ª demostración
example
  (h1 : P → Q)
  (h2 : P → ¬Q)
  : ¬P :=
assume h : P,
have h4 : Q := h1 h,
have h5 : ¬Q := h2 h,
show false,
  from h5 h4

-- 3ª demostración
example
  (h1 : P → Q)
  (h2 : P → ¬Q)
  : ¬P :=
assume h : P,
show false,
  from (h2 h) (h1 h)

-- 4ª demostración
example
  (h1 : P → Q)
  (h2 : P → ¬Q)
  : ¬P :=
assume h : P, (h2 h) (h1 h)

-- 5ª demostración
example
  (h1 : P → Q)
  (h2 : P → ¬Q)
  : ¬P :=
λ h, (h2 h) (h1 h)

-- 6ª demostración
example
  (h1 : P → Q)
  (h2 : P → ¬Q)
  : ¬P :=
begin

```

```
intro h,  
  have h3 :  $\neg Q := h2 h$ ,  
  apply h3,  
  apply h1,  
  exact h,  
end
```

-- 7ª demostración

```
example  
  (h1 :  $P \rightarrow Q$ )  
  (h2 :  $P \rightarrow \neg Q$ )  
  :  $\neg P :=$   
begin  
  intro h,  
  have h3 :  $\neg Q := h2 h$ ,  
  apply h3,  
  exact h1 h,  
end
```

-- 8ª demostración

```
example  
  (h1 :  $P \rightarrow Q$ )  
  (h2 :  $P \rightarrow \neg Q$ )  
  :  $\neg P :=$   
begin  
  intro h,  
  have h3 :  $\neg Q := h2 h$ ,  
  exact h3 (h1 h),  
end
```

-- 9ª demostración

```
example  
  (h1 :  $P \rightarrow Q$ )  
  (h2 :  $P \rightarrow \neg Q$ )  
  :  $\neg P :=$   
begin  
  intro h,  
  exact (h2 h) (h1 h),  
end
```

-- 10ª demostración

```
example  
  (h1 :  $P \rightarrow Q$ )  
  (h2 :  $P \rightarrow \neg Q$ )  
  :  $\neg P :=$ 
```

```

λ h, (h2 h) (h1 h)

-- 11ª demostración
example
  (h1 : P → Q)
  (h2 : P → ¬Q)
  : ¬P :=
by finish

```

2.3.3. Pruebas del modus tollens: $P \rightarrow Q, \neg Q \vdash \neg P$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas del modus tollens
-- =====

-- Ej. 1. Demostrar
--    $P \rightarrow Q, \neg Q \vdash \neg P$ 

import tactic

variables (P Q : Prop)

-- 1ª demostración
example
  (h1 : P → Q)
  (h2 : ¬Q)
  : ¬P :=
assume h3 : P,
have h4 : Q,
  from h1 h3,
show false,
  from h2 h4

-- 2ª demostración
example
  (h1 : P → Q)
  (h2 : ¬Q)
  : ¬P :=
assume h3 : P,
have h4 : Q := h1 h3,
show false,
  from h2 h4


```

```

-- 3ª demostración
example
  (h1 : P → Q)
  (h2 : ¬Q)
  : ¬P :=
assume h3 : P,
show false,
  from h2 (h1 h3)

-- 4ª demostración
example
  (h1 : P → Q)
  (h2 : ¬Q)
  : ¬P :=
assume h3 : P, h2 (h1 h3)

-- 5ª demostración
example
  (h1 : P → Q)
  (h2 : ¬Q)
  : ¬P :=
λ h, h2 (h1 h)

-- 6ª demostración
example
  (h1 : P → Q)
  (h2 : ¬Q)
  : ¬P :=
h2  h1

-- 7ª demostración
example
  (h1 : P → Q)
  (h2 : ¬Q)
  : ¬P :=
mt h1 h2

-- 8ª demostración
example
  (h1 : P → Q)
  (h2 : ¬Q)
  : ¬P :=
by tauto

```



```

-- 9ª demostración
example
  (h1 : P → Q)
  (h2 : ¬Q)
  : ¬P :=
by finish

-- 10ª demostración
example
  (h1 : P → Q)
  (h2 : ¬Q)
  : ¬P :=
begin
  intro h,
  apply h2,
  apply h1,
  exact h,
end

-- 11ª demostración
example
  (h1 : P → Q)
  (h2 : ¬Q)
  : ¬P :=
begin
  intro h,
  exact h2 (h1 h),
end

-- 12ª demostración
example
  (h1 : P → Q)
  (h2 : ¬Q)
  : ¬P :=
λ h, h2 (h1 h)

```

2.3.4. Pruebas de $P \rightarrow (Q \rightarrow R)$, $P, \neg R \vdash \neg Q$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de  $P \rightarrow (Q \rightarrow R)$ ,  $P, \neg R \vdash \neg Q$ 
-- =====

```

```
-- Ej. 1. Demostrar
-- Pruebas de  $P \rightarrow (Q \rightarrow R), P, \neg R \vdash \neg Q$ 
```

```
import tactic
```

```
variables (P Q R : Prop)
```

```
-- 1ª demostración
```

```
example
```

```
(h1 : P → (Q → R))
```

```
(h2 : P)
```

```
(h3 : ¬R)
```

```
: ¬Q :=
```

```
have h4 : Q → R,
```

```
  from h1 h2,
```

```
show ¬Q,
```

```
  from mt h4 h3
```

```
-- 2ª demostración
```

```
example
```

```
(h1 : P → (Q → R))
```

```
(h2 : P)
```

```
(h3 : ¬R)
```

```
: ¬Q :=
```

```
have h4 : Q → R := h1 h2,
```

```
show ¬Q,
```

```
  from mt h4 h3
```

```
-- 3ª demostración
```

```
example
```

```
(h1 : P → (Q → R))
```

```
(h2 : P)
```

```
(h3 : ¬R)
```

```
: ¬Q :=
```

```
show ¬Q,
```

```
  from mt (h1 h2) h3
```

```
-- 4ª demostración
```

```
example
```

```
(h1 : P → (Q → R))
```

```
(h2 : P)
```

```
(h3 : ¬R)
```

```
: ¬Q :=
```

```
mt (h1 h2) h3
```

```
-- 5ª demostración
example
  (h1 : P → (Q → R))
  (h2 : P)
  (h3 : ¬R)
  : ¬Q :=
begin
  intro h4,
  apply h3,
  apply (h1 h2),
  exact h4,
end

-- 6ª demostración
example
  (h1 : P → (Q → R))
  (h2 : P)
  (h3 : ¬R)
  : ¬Q :=
begin
  intro h4,
  apply h3,
  exact (h1 h2) h4,
end

-- 7ª demostración
example
  (h1 : P → (Q → R))
  (h2 : P)
  (h3 : ¬R)
  : ¬Q :=
begin
  intro h4,
  exact h3 ((h1 h2) h4),
end

-- 8ª demostración
example
  (h1 : P → (Q → R))
  (h2 : P)
  (h3 : ¬R)
  : ¬Q :=
λ h4, h3 ((h1 h2) h4)

-- 9ª demostración
```

```
example
  (h1 : P → (Q → R))
  (h2 : P)
  (h3 : ¬R)
  : ¬Q :=
by finish
```

2.3.5. Pruebas de $P \rightarrow Q \vdash \neg Q \rightarrow \neg P$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Pruebas de  $P \rightarrow Q \vdash \neg Q \rightarrow \neg P$ 
-- =====
```

```
-- Ej. 1. Demostrar
--  $P \rightarrow Q \vdash \neg Q \rightarrow \neg P$ 
```

```
import tactic
```

```
variables (P Q : Prop)
```

```
-- 1ª demostración
```

```
example
  (h1 : P → Q)
  : ¬Q → ¬P :=
assume h2 : ¬Q,
show ¬P,
from mt h1 h2
```

```
-- 2ª demostración
```

```
example
  (h1 : P → Q)
  : ¬Q → ¬P :=
assume h2 : ¬Q, mt h1 h2
```

```
-- 3ª demostración
```

```
example
  (h1 : P → Q)
  : ¬Q → ¬P :=
λ h2, mt h1 h2
```

```
-- 4ª demostración
```

```
example
```

```
(h1 : P → Q)
: ¬Q → ¬P :=
mt h1

-- 5ª demostración
example
  (h1 : P → Q)
  : ¬Q → ¬P :=
begin
  intro h2,
  exact mt h1 h2,
end

-- 6ª demostración
example
  (h1 : P → Q)
  : ¬Q → ¬P :=
begin
  intro h2,
  intro h3,
  apply h2,
  apply h1,
  exact h3,
end

-- 7ª demostración
example
  (h1 : P → Q)
  : ¬Q → ¬P :=
begin
  intro h2,
  intro h3,
  apply h2,
  exact h1 h3,
end

-- 8ª demostración
example
  (h1 : P → Q)
  : ¬Q → ¬P :=
begin
  intro h2,
  intro h3,
  exact h2 (h1 h3),
end
```

```

-- 9ª demostración
example
  (h1 : P → Q)
  : ¬Q → ¬P :=
begin
  intros h2 h3,
  exact h2 (h1 h3),
end

-- 10ª demostración
example
  (h1 : P → Q)
  : ¬Q → ¬P :=
λ h2 h3, h2 (h1 h3)

-- 11ª demostración
example
  (h1 : P → Q)
  : ¬Q → ¬P :=
by tauto

-- 12ª demostración
example
  (h1 : P → Q)
  : ¬Q → ¬P :=
by finish

```

2.3.6. Regla de introducción de la doble negación: $P \vdash \neg\neg P$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Regla de introducción de la doble negación
-- =====

-- Ej. 1. Demostrar
--      P ⊢ ¬¬P

import tactic
variable (P : Prop)

-- 1ª demostración

```

```
example
  (h1 : P)
  :  $\neg\neg P$  :=
not.intro
  ( assume h2:  $\neg P$ ,
    show false,
    from h2 h1)

-- 2ª demostración
example
  (h1 : P)
  :  $\neg\neg P$  :=
assume h2:  $\neg P$ ,
show false,
  from h2 h1

-- 3ª demostración
example
  (h1 : P)
  :  $\neg\neg P$  :=
assume h2:  $\neg P$ , h2 h1

-- 4ª demostración
example
  (h1 : P)
  :  $\neg\neg P$  :=
 $\lambda$  h2, h2 h1

-- 5ª demostración
example
  (h1 : P)
  :  $\neg\neg P$  :=
not_not.mpr h1

-- 6ª demostración
example
  (h1 : P)
  :  $\neg\neg P$  :=
not_not_intro h1

-- 7ª demostración
example
  (h1 : P)
  :  $\neg\neg P$  :=
begin
```

```

    intro h2,
    exact h2 h1,
end

-- 8ª demostración
example
  (h1 : P)
  :  $\neg\neg P$  :=
by tauto

-- 9ª demostración
example
  (h1 : P)
  :  $\neg\neg P$  :=
by finish

```

2.3.7. Pruebas de $\neg Q \rightarrow \neg P \vdash P \rightarrow \neg\neg Q$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de  $\neg Q \rightarrow \neg P \vdash P \rightarrow \neg\neg Q$ 
-- =====

-- Ej. 1. Demostrar
--  $\neg Q \rightarrow \neg P \vdash P \rightarrow \neg\neg Q$ 

import tactic

variables (P Q : Prop)

-- 1ª demostración
example
  (h1 :  $\neg Q \rightarrow \neg P$ )
  :  $P \rightarrow \neg\neg Q$  :=
assume h2 : P,
have h3 :  $\neg\neg P$ ,
  from not_not_intro h2,
show  $\neg\neg Q$ ,
  from mt h1 h3

-- 2ª demostración
example
  (h1 :  $\neg Q \rightarrow \neg P$ )

```



```

: P → ¬¬Q :=
assume h2 : P,
have h3 : ¬¬P := not_not_intro h2,
show ¬¬Q,
  from mt h1 h3

-- 3ª demostración
example
  (h1 : ¬Q → ¬P)
  : P → ¬¬Q :=
assume h2 : P,
show ¬¬Q,
  from mt h1 (not_not_intro h2)

-- 4ª demostración
example
  (h1 : ¬Q → ¬P)
  : P → ¬¬Q :=
assume h2 : P, mt h1 (not_not_intro h2)

-- 5ª demostración
example
  (h1 : ¬Q → ¬P)
  : P → ¬¬Q :=
λ h2, mt h1 (not_not_intro h2)

-- 6ª demostración
example
  (h1 : ¬Q → ¬P)
  : P → ¬¬Q :=
imp_not_comm.mp h1

-- 7ª demostración
example
  (h1 : ¬Q → ¬P)
  : P → ¬¬Q :=
begin
  intro h2,
  apply mt h1,
  apply not_not_intro,
  exact h2,
end

-- 8ª demostración
example

```

```

(h1 :  $\neg Q \rightarrow \neg P$ )
:  $P \rightarrow \neg\neg Q$  :=
begin
  intro h2,
  apply mt h1,
  exact not_not_intro h2,
end

-- 9ª demostración
example
(h1 :  $\neg Q \rightarrow \neg P$ )
:  $P \rightarrow \neg\neg Q$  :=
begin
  intro h2,
  exact mt h1 (not_not_intro h2),
end

-- 10ª demostración
example
(h1 :  $\neg Q \rightarrow \neg P$ )
:  $P \rightarrow \neg\neg Q$  :=
λ h2, mt h1 (not_not_intro h2)

-- 11ª demostración
example
(h1 :  $\neg Q \rightarrow \neg P$ )
:  $P \rightarrow \neg\neg Q$  :=
begin
  intro h2,
  intro h3,
  have h4 :  $\neg P$  := h1 h3,
  exact h4 h2,
end

-- 12ª demostración
example
(h1 :  $\neg Q \rightarrow \neg P$ )
:  $P \rightarrow \neg\neg Q$  :=
begin
  intros h2 h3,
  exact (h1 h3) h2,
end

-- 13ª demostración
example

```

```

(h1 :  $\neg Q \rightarrow \neg P$ )
:  $P \rightarrow \neg\neg Q$  :=
λ h2 h3, (h1 h3) h2

-- 14ª demostración
example
(h1 :  $\neg Q \rightarrow \neg P$ )
:  $P \rightarrow \neg\neg Q$  :=
by tauto

-- 15ª demostración
example
(h1 :  $\neg Q \rightarrow \neg P$ )
:  $P \rightarrow \neg\neg Q$  :=
by finish

```

2.4. Reglas de la disyunción

2.4.1. Reglas de introducción de la disyunción

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Reglas de introducción de la disyunción
-- =====

import tactic

variables (P Q R : Prop)

-- Ej. 1. Demostrar
--    $P \vdash P \vee Q$ 

-- 1ª demostración
example
(h : P)
:  $P \vee Q$  :=
or.intro_left Q h

-- 2ª demostración
example
(h : P)
:  $P \vee Q$  :=

```

```

or.inl h

-- 3ª demostración
example
  (h : P)
  : P ∨ Q :=
by tauto

-- 4ª demostración
example
  (h : P)
  : P ∨ Q :=
by finish

-- Ej. 2. Demostrar
--    $P \wedge Q \vdash P \vee R$ 

-- 1ª demostración
example
  (h1 : P ∧ Q)
  : P ∨ R :=
have h2 : P,
  from and.elim_left h1,
show P ∨ R,
  from or.inl h2

-- 2ª demostración
example
  (h1 : P ∧ Q)
  : P ∨ R :=
have h2 : P,
  from h1.1,
show P ∨ R,
  from or.inl h2

-- 3ª demostración
example
  (h1 : P ∧ Q)
  : P ∨ R :=
have h2 : P := h1.1,
show P ∨ R,
  from or.inl h2

-- 4ª demostración
example

```

```

(h1 : P ∧ Q)
: P ∨ R :=
show P ∨ R,
  from or.inl h1.1

-- 5ª demostración
example
(h1 : P ∧ Q)
: P ∨ R :=
or.inl h1.1

-- 6ª demostración
example
(h1 : P ∧ Q)
: P ∨ R :=
by tauto

-- 7ª demostración
example
(h1 : P ∧ Q)
: P ∨ R :=
by finish

-- Ej. 3. Demostrar
--   Q ⊢ P ∨ Q

-- 1ª demostración
example
(h : Q)
: P ∨ Q :=
or.intro_right P h

-- 2ª demostración
example
(h : Q)
: P ∨ Q :=
or.inr h

-- 3ª demostración
example
(h : Q)
: P ∨ Q :=
by tauto

-- 4ª demostración

```

```

example
  (h : Q)
  : P ∨ Q :=
by finish

-- Ej. 4. Demostrar
--    $P \wedge Q \vdash P \vee R$ 

-- 1ª demostración
example
  (h1 : P ∧ Q)
  : R ∨ Q :=
have h2 : Q,
  from and.elim_right h1,
show R ∨ Q,
  from or.inr h2

-- 2ª demostración
example
  (h1 : P ∧ Q)
  : R ∨ Q :=
have h2 : Q,
  from h1.2,
show R ∨ Q,
  from or.inr h2

-- 3ª demostración
example
  (h1 : P ∧ Q)
  : R ∨ Q :=
have h2 : Q := h1.2,
show R ∨ Q,
  from or.inr h2

-- 4ª demostración
example
  (h1 : P ∧ Q)
  : R ∨ Q :=
show R ∨ Q,
  from or.inr h1.2

-- 5ª demostración
example
  (h1 : P ∧ Q)
  : R ∨ Q :=

```

```

or.inr h1.2

-- 6ª demostración
example
  (h1 : P ∧ Q)
  : R ∨ Q :=
by tauto

-- 7ª demostración
example
  (h1 : P ∧ Q)
  : R ∨ Q :=
by finish

```

2.4.2. Regla de eliminación de la disyunción

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Regla de eliminación de la disyunción
-- =====

import tactic

variables (P Q R : Prop)

-- Ej. 1. Demostrar
--    $P \vee Q, P \rightarrow R, Q \rightarrow R \vdash R$ 

-- 1ª demostración
example
  (h1 : P ∨ Q)
  (h2 : P → R)
  (h3 : Q → R)
  : R :=
or.elim h1 h2 h3

-- 2ª demostración
example
  (h1 : P ∨ Q)
  (h2 : P → R)
  (h3 : Q → R)
  : R :=
h1.elim h2 h3

```

```

-- 3ª demostración
example
  (h1 : P ∨ Q)
  (h2 : P → R)
  (h3 : Q → R)
  : R :=
or.rec h2 h3 h1

-- 4ª demostración
example
  (h1 : P ∨ Q)
  (h2 : P → R)
  (h3 : Q → R)
  : R :=
begin
  cases h1 with hP hQ,
  { exact h2 hP, },
  { exact h3 hQ, },
end

-- 5ª demostración
example
  (h1 : P ∨ Q)
  (h2 : P → R)
  (h3 : Q → R)
  : R :=
by tauto

-- 6ª demostración
example
  (h1 : P ∨ Q)
  (h2 : P → R)
  (h3 : Q → R)
  : R :=
by finish

```

2.4.3. Pruebas de $P \vee Q \vdash Q \vee P$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de  $P \vee Q \vdash Q \vee P$ 
-- =====

```



```

import tactic

variables (P Q R : Prop)

-- Ej. 1. Demostrar
--    $P \vee Q \vdash Q \vee P$ 

-- 1ª demostración
example
  (h1 : P  $\vee$  Q)
  : Q  $\vee$  P :=
or.elim h1
  ( assume h2 : P,
    show Q  $\vee$  P,
    from or.inr h2 )
  ( assume h3 : Q,
    show Q  $\vee$  P,
    from or.inl h3 )

-- 2ª demostración
example
  (h1 : P  $\vee$  Q)
  : Q  $\vee$  P :=
or.elim h1
  (  $\lambda$  h, or.inr h )
  (  $\lambda$  h, or.inl h )

-- 3ª demostración
example
  (h1 : P  $\vee$  Q)
  : Q  $\vee$  P :=
or.elim h1 or.inr or.inl

example
  (h1 : P  $\vee$  Q)
  : Q  $\vee$  P :=
h1.elim or.inr or.inl

-- 4ª demostración
example
  (h1 : P  $\vee$  Q)
  : Q  $\vee$  P :=
or.rec or.inr or.inl h1

```

```

-- 5ª demostración
example
  (h1 : P ∨ Q)
  : Q ∨ P :=
or.swap h1

-- 6ª demostración
example
  (h1 : P ∨ Q)
  : Q ∨ P :=
begin
  cases h1 with h2 h3,
  { exact or.inr h2, },
  { exact or.inl h3, },
end

-- 7ª demostración
example
  (P ∨ Q)
  : Q ∨ P :=
begin
  cases <P ∨ Q>,
  { exact or.inr <P>, },
  { exact or.inl <Q>, },
end

-- 8ª demostración
example
  (h1 : P ∨ Q)
  : Q ∨ P :=
begin
  cases h1 with h2 h3,
  { right,
    exact h2, },
  { left,
    exact h3, },
end

-- 9ª demostración
example
  (h1 : P ∨ Q)
  : Q ∨ P :=
by tauto

```

```

-- 10ª demostración
example
  (h1 : P ∨ Q)
  : Q ∨ P :=
by finish

-----

-- Ej. 1. Demostrar
--    $P \vee Q \vdash Q \vee P$ 
-----

-- 1ª demostración
example
  (h1 : P ∨ Q)
  : Q ∨ P :=
or.elim h1
  ( assume h2 : P,
    show Q ∨ P,
    from or.inr h2 )
  ( assume h3 : Q,
    show Q ∨ P,
    from or.inl h3 )

-- 2ª demostración
example
  (h1 : P ∨ Q)
  : Q ∨ P :=
or.elim h1
  ( λ h, or.inr h )
  ( λ h, or.inl h )

-- 3ª demostración
example
  (h1 : P ∨ Q)
  : Q ∨ P :=
or.elim h1 or.inr or.inl

example
  (h1 : P ∨ Q)
  : Q ∨ P :=
h1.elim or.inr or.inl

-- 4ª demostración
example
  (h1 : P ∨ Q)

```

```

: Q ∨ P :=
or.rec or.inr or.inl h1

-- 5ª demostración
example
  (h1 : P ∨ Q)
  : Q ∨ P :=
or.swap h1

-- 6ª demostración
example
  (h1 : P ∨ Q)
  : Q ∨ P :=
begin
  cases h1 with h2 h3,
  { exact or.inr h2, },
  { exact or.inl h3, },
end

-- 7ª demostración
example
  (P ∨ Q)
  : Q ∨ P :=
begin
  cases <P ∨ Q>,
  { exact or.inr <P>, },
  { exact or.inl <Q>, },
end

-- 8ª demostración
example
  (h1 : P ∨ Q)
  : Q ∨ P :=
begin
  cases h1 with h2 h3,
  { right,
    exact h2, },
  { left,
    exact h3, },
end

-- 9ª demostración
example
  (h1 : P ∨ Q)

```

```

: Q ∨ P :=
by tauto

-- 10ª demostración
example
  (h1 : P ∨ Q)
  : Q ∨ P :=
by finish

-----

-- Ej. 1. Demostrar
--    $Q \rightarrow R \vdash P \vee Q \rightarrow P \vee R$ 
-----

-- 1ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
assume h2 : P ∨ Q,
or.elim h2
  ( assume h3 : P,
    show P ∨ R,
    from or.inl h3 )
  ( assume h4 : Q,
    have h5 : R := h1 h4,
    show P ∨ R,
    from or.inr h5 )

-- 2ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
assume h2 : P ∨ Q,
or.elim h2
  ( assume h3 : P, or.inl h3 )
  ( assume h4 : Q,
    show P ∨ R,
    from or.inr (h1 h4) )

-- 3ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
assume h2 : P ∨ Q,
or.elim h2

```

```

( assume h3 : P, or.inl h3 )
( assume h4 : Q, or.inr (h1 h4) )

-- 4ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
assume h2 : P ∨ Q,
or.elim h2
  ( λ h3, or.inl h3 )
  ( λ h4, or.inr (h1 h4) )

-- 5ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
assume h2 : P ∨ Q,
or.elim h2
  or.inl
  ( λ h, or.inr (h1 h) )

-- 6ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
λ h2, or.elim h2 or.inl (λ h, or.inr (h1 h))

example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
λ h2, h2.elim or.inl (λ h, or.inr (h1 h))

-- 7ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
or.imp_right h1

-- 8ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
begin
  intro h2,
  cases h2 with h3 h4,

```

```

{ exact or.inl h3, },
{ exact or.inr (h1 h4), },
end

```

```
-- 9ª demostración
```

```
example
```

```

(h1 : Q → R)
: P ∨ Q → P ∨ R :=

```

```
begin
```

```

  intro h2,
  cases h2 with h3 h4,
  { left,
    exact h3, },
  { right,
    exact (h1 h4), },
end

```

```
-- 10ª demostración
```

```
example
```

```

(h1 : Q → R)
: P ∨ Q → P ∨ R :=

```

```
begin
```

```

  rintro (h3 | h4),
  { left,
    exact h3, },
  { right,
    exact (h1 h4), },
end

```

```
-- 11ª demostración
```

```
example
```

```

(h1 : Q → R)
: P ∨ Q → P ∨ R :=

```

```
by tauto
```

```
-- 12ª demostración
```

```
example
```

```

(h1 : Q → R)
: P ∨ Q → P ∨ R :=

```

```
by finish
```

```

-----
-- Ej. 1. Demostrar
--    $\neg P \vee Q \vdash P \rightarrow Q$ 
-----

```

```

-- 1ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
  assume h2 : P,
  or.elim h1
    ( assume h3 :  $\neg P$ ,
      have h4 : false,
      from h3 h2,
      show Q,
      from false.elim h4)
    ( assume h5 : Q,
      show Q, from h5)

-- 2ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
  assume h2 : P,
  or.elim h1
    ( assume h3 :  $\neg P$ ,
      have h4 : false,
      from h3 h2,
      show Q,
      from false.elim h4)
    ( assume h5 : Q, h5)

-- 3ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
  assume h2 : P,
  or.elim h1
    ( assume h3 :  $\neg P$ ,
      have h4 : false,
      from h3 h2,
      show Q,
      from false.elim h4)
    (  $\lambda$  h5, h5)

-- 4ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=

```



```

assume h2 : P,
or.elim h1
  ( assume h3 : ¬P,
    have h4 : false,
      from h3 h2,
    show Q,
      from false.elim h4)
  id

-- 5ª demostración
example
  (h1 : ¬P ∨ Q)
  : P → Q :=
assume h2 : P,
or.elim h1
  ( assume h3 : ¬P,
    show Q,
      from false.elim (h3 h2))
  id

-- 6ª demostración
example
  (h1 : ¬P ∨ Q)
  : P → Q :=
assume h2 : P,
or.elim h1
  ( assume h3 : ¬P, false.elim (h3 h2))
  id

-- 7ª demostración
example
  (h1 : ¬P ∨ Q)
  : P → Q :=
assume h2 : P,
or.elim h1
  ( λ h3, false.elim (h3 h2))
  id

-- 8ª demostración
example
  (h1 : ¬P ∨ Q)
  : P → Q :=
λ h2, or.elim h1 (λ h3, false.elim (h3 h2)) id

example

```

```

(h1 :  $\neg P \vee Q$ )
:  $P \rightarrow Q$  :=
λ h2, h1.elim (λ h3, false.elim (h3 h2)) id

```

example

```

(h1 :  $\neg P \vee Q$ )
:  $P \rightarrow Q$  :=
λ h2, h1.elim (λ h3, (h3 h2).elim) id

```

-- 9ª demostración

example

```

(h1 :  $\neg P \vee Q$ )
:  $P \rightarrow Q$  :=
imp_iff_not_or.mpr h1

```

-- 10ª demostración

example

```

(h1 :  $\neg P \vee Q$ )
:  $P \rightarrow Q$  :=
begin
  intro h2,
  cases h1 with h3 h4,
  { apply false.rec,
    exact h3 h2, },
  { exact h4, },
end

```

-- 11ª demostración

example

```

(h1 :  $\neg P \vee Q$ )
:  $P \rightarrow Q$  :=
begin
  intro h2,
  cases h1 with h3 h4,
  { exact false.elim (h3 h2), },
  { exact h4, },
end

```

-- 12ª demostración

example

```

(h1 :  $\neg P \vee Q$ )
:  $P \rightarrow Q$  :=
begin
  intro h2,
  cases h1 with h3 h4,

```

```

{ exfalso,
  exact h3 h2, },
{ exact h4, },
end

```

```
-- 13ª demostración
```

```
example
  (h1 : ¬P ∨ Q)
  : P → Q :=
by tauto

```

```
-- 14ª demostración
```

```
example
  (h1 : ¬P ∨ Q)
  : P → Q :=
by finish

```

2.4.4. Pruebas de $Q \rightarrow R \vdash P \vee Q \rightarrow P \vee R$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Pruebas de  $Q \rightarrow R \vdash P \vee Q \rightarrow P \vee R$ 
-- =====

```

```
-- Ej. 1. Demostrar
```

```
--  $Q \rightarrow R \vdash P \vee Q \rightarrow P \vee R$ 

```

```
import tactic

```

```
variables (P Q R : Prop)

```

```
-- 1ª demostración
```

```
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
assume h2 : P ∨ Q,
or.elim h2
( assume h3 : P,
  show P ∨ R,
  from or.inl h3 )
( assume h4 : Q,
  have h5 : R := h1 h4,
  show P ∨ R,

```

```

    from or.inr h5 )

-- 2ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
assume h2 : P ∨ Q,
or.elim h2
  ( assume h3 : P, or.inl h3 )
  ( assume h4 : Q,
    show P ∨ R,
    from or.inr (h1 h4) )

-- 3ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
assume h2 : P ∨ Q,
or.elim h2
  ( assume h3 : P, or.inl h3 )
  ( assume h4 : Q, or.inr (h1 h4) )

-- 4ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
assume h2 : P ∨ Q,
or.elim h2
  ( λ h3, or.inl h3 )
  ( λ h4, or.inr (h1 h4) )

-- 5ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
assume h2 : P ∨ Q,
or.elim h2
  or.inl
  ( λ h, or.inr (h1 h) )

-- 6ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
λ h2, or.elim h2 or.inl (λ h, or.inr (h1 h))

```

```
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
λ h2, h2.elim or.inl (λ h, or.inr (h1 h))
```

-- 7ª demostración

```
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
or.imp_right h1
```

-- 8ª demostración

```
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
begin
  intro h2,
  cases h2 with h3 h4,
  { exact or.inl h3, },
  { exact or.inr (h1 h4), },
end
```

-- 9ª demostración

```
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
begin
  intro h2,
  cases h2 with h3 h4,
  { left,
    exact h3, },
  { right,
    exact (h1 h4), },
end
```

-- 10ª demostración

```
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
begin
  rintro (h3 | h4),
  { left,
    exact h3, },
  { right,
```

```

    exact (h1 h4), },
end

-- 11ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
by tauto

-- 12ª demostración
example
  (h1 : Q → R)
  : P ∨ Q → P ∨ R :=
by finish

```

2.4.5. Pruebas de $\neg P \vee Q \vdash P \rightarrow Q$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Prueba de  $\neg P \vee Q \vdash P \rightarrow Q$ 
-- =====

-- Ej. 1. Demostrar
--  $\neg P \vee Q \vdash P \rightarrow Q$ 

import tactic

variables (P Q : Prop)

-- 1ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  : P → Q :=
assume h2 : P,
or.elim h1
( assume h3 :  $\neg P$ ,
  have h4 : false,
  from h3 h2,
  show Q,
  from false.elim h4)
( assume h5 : Q,
  show Q, from h5)

```

```
-- 2ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
assume h2 : P,
or.elim h1
  ( assume h3 :  $\neg P$ ,
    have h4 : false,
      from h3 h2,
    show Q,
      from false.elim h4)
  ( assume h5 : Q, h5)

-- 3ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
assume h2 : P,
or.elim h1
  ( assume h3 :  $\neg P$ ,
    have h4 : false,
      from h3 h2,
    show Q,
      from false.elim h4)
  (  $\lambda$  h5, h5)

-- 4ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
assume h2 : P,
or.elim h1
  ( assume h3 :  $\neg P$ ,
    have h4 : false,
      from h3 h2,
    show Q,
      from false.elim h4)
  id

-- 5ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
assume h2 : P,
or.elim h1
```

```

( assume h3 :  $\neg P$ ,
  show Q,
  from false.elim (h3 h2))
id

-- 6ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
assume h2 : P,
or.elim h1
  ( assume h3 :  $\neg P$ , false.elim (h3 h2))
  id

-- 7ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
assume h2 : P,
or.elim h1
  (  $\lambda$  h3, false.elim (h3 h2))
  id

-- 8ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
 $\lambda$  h2, or.elim h1 ( $\lambda$  h3, false.elim (h3 h2)) id

example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
 $\lambda$  h2, h1.elim ( $\lambda$  h3, false.elim (h3 h2)) id

example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
 $\lambda$  h2, h1.elim ( $\lambda$  h3, (h3 h2).elim) id

-- 9ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
imp_iff_not_or.mpr h1

```



```
-- 10ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
begin
  intro h2,
  cases h1 with h3 h4,
  { apply false.rec,
    exact h3 h2, },
  { exact h4, },
end

-- 11ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
begin
  intro h2,
  cases h1 with h3 h4,
  { exact false.elim (h3 h2), },
  { exact h4, },
end

-- 12ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
begin
  intro h2,
  cases h1 with h3 h4,
  { exfalso,
    exact h3 h2, },
  { exact h4, },
end

-- 13ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
by tauto

-- 14ª demostración
example
  (h1 :  $\neg P \vee Q$ )
  :  $P \rightarrow Q$  :=
```

by finish

2.5. Reglas del bicondicional

2.5.1. Regla de introducción del bicondicional en $P \wedge Q \leftrightarrow Q \wedge P$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Regla de introducción del bicondicional
-- =====
```

```
-- Ej. 1. Demostrar
--    $P \wedge Q \leftrightarrow Q \wedge P$ 
```

```
import tactic
```

```
variables (P Q : Prop)
```

```
-- 1ª demostración
```

```
example : P ∧ Q ↔ Q ∧ P :=
iff.intro
  ( assume h1 : P ∧ Q,
    have h2 : P,
      from and.elim_left h1,
    have h3 : Q,
      from and.elim_right h1,
    show Q ∧ P,
      from and.intro h3 h2)
  ( assume h4 : Q ∧ P,
    have h5 : Q,
      from and.elim_left h4,
    have h6 : P,
      from and.elim_right h4,
    show P ∧ Q,
      from and.intro h6 h5)
```

```
-- 2ª demostración
```

```
example : P ∧ Q ↔ Q ∧ P :=
iff.intro
  ( assume h1 : P ∧ Q,
```

```

    have h2 : P,
      from h1.1,
    have h3 : Q,
      from h1.2,
    show Q  $\wedge$  P,
      from and.intro h3 h2)
  ( assume h4 : Q  $\wedge$  P,
    have h5 : Q,
      from h4.1,
    have h6 : P,
      from h4.2,
    show P  $\wedge$  Q,
      from and.intro h6 h5)

-- 3ª demostración
example : P  $\wedge$  Q  $\leftrightarrow$  Q  $\wedge$  P :=
iff.intro
  ( assume h1 : P  $\wedge$  Q,
    have h2 : P := h1.1,
    have h3 : Q := h1.2,
    show Q  $\wedge$  P,
      from and.intro h3 h2)
  ( assume h4 : Q  $\wedge$  P,
    have h5 : Q := h4.1,
    have h6 : P := h4.2,
    show P  $\wedge$  Q,
      from and.intro h6 h5)

-- 4ª demostración
example : P  $\wedge$  Q  $\leftrightarrow$  Q  $\wedge$  P :=
iff.intro
  ( assume h1 : P  $\wedge$  Q,
    show Q  $\wedge$  P,
      from and.intro h1.2 h1.1)
  ( assume h4 : Q  $\wedge$  P,
    show P  $\wedge$  Q,
      from and.intro h4.2 h4.1)

-- 5ª demostración
example : P  $\wedge$  Q  $\leftrightarrow$  Q  $\wedge$  P :=
iff.intro
  ( assume h1 : P  $\wedge$  Q, and.intro h1.2 h1.1)
  ( assume h4 : Q  $\wedge$  P, and.intro h4.2 h4.1)

-- 6ª demostración

```

```

example : P ∧ Q ↔ Q ∧ P :=
iff.intro
  ( assume h1 : P ∧ Q, ⟨h1.2, h1.1⟩ )
  ( assume h4 : Q ∧ P, ⟨h4.2, h4.1⟩ )

```

-- 7ª demostración

```

example : P ∧ Q ↔ Q ∧ P :=
iff.intro
  ( λ h, ⟨h.2, h.1⟩ )
  ( λ h, ⟨h.2, h.1⟩ )

```

-- 8ª demostración

```

lemma aux :
  P ∧ Q → Q ∧ P :=
λ h, ⟨h.2, h.1⟩

```

```

example : P ∧ Q ↔ Q ∧ P :=
iff.intro (aux P Q) (aux Q P)

```

-- 9ª demostración

```

example : P ∧ Q ↔ Q ∧ P :=
and.comm

```

-- 10ª demostración

```

example : P ∧ Q ↔ Q ∧ P :=
begin
  split,
  { intro h1,
    cases h1 with h2 h3,
    split,
    { exact h3, },
    { exact h2, }},
  { intro h4,
    cases h4 with h5 h6,
    split,
    { exact h6, },
    { exact h5, }},
end

```

-- 11ª demostración

```

example : P ∧ Q ↔ Q ∧ P :=
begin
  split,
  { rintro ⟨h2, h3⟩,
    split,

```

```

    { exact h3, },
    { exact h2, }},
  { rintro ⟨h5, h6⟩,
    split,
    { exact h6, },
    { exact h5, }},
end

-- 12ª demostración
example : P ∧ Q ↔ Q ∧ P :=
begin
  constructor,
  { rintro ⟨h2, h3⟩,
    constructor,
    { exact h3, },
    { exact h2, }},
  { rintro ⟨h5, h6⟩,
    constructor,
    { exact h6, },
    { exact h5, }},
end

-- 13ª demostración
example : P ∧ Q ↔ Q ∧ P :=
by tauto

-- 14ª demostración
example : P ∧ Q ↔ Q ∧ P :=
by finish

```

2.5.2. Reglas de eliminación del bicondicional en $P \leftrightarrow Q$, $P \vee Q \vdash P \wedge Q$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Reglas de eliminacion del condicional
-- =====

-- Ej. 1. Demostrar
--      P ↔ Q, P ∨ Q ⊢ P ∧ Q

import tactic

```

```

variables (P Q : Prop)

-- 1ª demostración
example
  (h1 : P ↔ Q)
  (h2 : P ∨ Q)
  : P ∧ Q :=
or.elim h2
  ( assume h3 : P,
    have h4 : P → Q,
      from iff.elim_left h1,
    have h5 : Q,
      from h4 h3,
    show P ∧ Q,
      from and.intro h3 h5 )
  ( assume h6 : Q,
    have h7 : Q → P,
      from iff.elim_right h1,
    have h8 : P,
      from h7 h6,
    show P ∧ Q,
      from and.intro h8 h6 )

-- 2ª demostración
example
  (h1 : P ↔ Q)
  (h2 : P ∨ Q)
  : P ∧ Q :=
or.elim h2
  ( assume h3 : P,
    have h4 : P → Q := h1.1,
    have h5 : Q := h4 h3,
    show P ∧ Q, from ⟨h3, h5⟩ )
  ( assume h6 : Q,
    have h7 : Q → P := h1.2,
    have h8 : P := h7 h6,
    show P ∧ Q, from ⟨h8, h6⟩ )

-- 3ª demostración
example
  (h1 : P ↔ Q)
  (h2 : P ∨ Q)
  : P ∧ Q :=
or.elim h2

```

```

( assume h3 : P,
  show P  $\wedge$  Q, from  $\langle$ h3, (h1.1 h3) $\rangle$  )
( assume h6 : Q,
  show P  $\wedge$  Q, from  $\langle$ h1.2 h6, h6 $\rangle$  )

-- 4ª demostración
example
  (h1 : P  $\leftrightarrow$  Q)
  (h2 : P  $\vee$  Q)
  : P  $\wedge$  Q :=
or.elim h2
  ( $\lambda$ h,  $\langle$ h, (h1.1 h) $\rangle$ )
  ( $\lambda$ h,  $\langle$ h1.2 h, h $\rangle$ )

example
  (h1 : P  $\leftrightarrow$  Q)
  (h2 : P  $\vee$  Q)
  : P  $\wedge$  Q :=
h2.elim
  ( $\lambda$ h,  $\langle$ h, (h1.1 h) $\rangle$ )
  ( $\lambda$ h,  $\langle$ h1.2 h, h $\rangle$ )

-- 5ª demostración
example
  (h1 : P  $\leftrightarrow$  Q)
  (h2 : P  $\vee$  Q)
  : P  $\wedge$  Q :=
begin
  cases h2 with h3 h4,
  { split,
    { exact h3, },
    { apply h1.mp,
      exact h3, }},
  { split,
    { apply h1.mpr,
      exact h4, },
    { exact h4, }},
end

-- 6ª demostración
example
  (h1 : P  $\leftrightarrow$  Q)
  (h2 : P  $\vee$  Q)
  : P  $\wedge$  Q :=
begin

```

```

cases h2 with h3 h4,
{ split,
  { exact h3, },
  { rw ← h1,
    exact h3, }},
{ split,
  { rw h1,
    exact h4, },
  { exact h4, }},
end

-- 7ª demostración
example
  (h1 : P ↔ Q)
  (h2 : P ∨ Q)
  : P ∧ Q :=
by tauto

-- 8ª demostración
example
  (h1 : P ↔ Q)
  (h2 : P ∨ Q)
  : P ∧ Q :=
by finish

-- 9ª demostración
example
  (h1 : P ↔ Q)
  (h2 : P ∨ Q)
  : P ∧ Q :=
begin
  simp [h1] at h2 |-,
  assumption,
end

```

2.6. Reglas de la lógica clásica

2.6.1. Pruebas de la regla de reducción al absurdo

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).


```

-- Prueba de la regla de reducción al absurdo
-- =====

import tactic

variable (P : Prop)

-- 1ª demostración
example
  (h1 :  $\neg P \rightarrow \text{false}$ )
  : P :=
have h2 :  $\neg\neg P$ , from
  assume h3 :  $\neg P$ ,
  show false, from h1 h3,
show P, from not_not.mp h2

-- 2ª demostración
example
  (h1 :  $\neg P \rightarrow \text{false}$ )
  : P :=
begin
  apply not_not.mp,
  intro h2,
  exact h1 h2,
end

-- 3ª demostración
example
  (h1 :  $\neg P \rightarrow \text{false}$ )
  : P :=
begin
  apply not_not.mp,
  exact  $\lambda$  h2, h1 h2,
end

-- 4ª demostración
example
  (h1 :  $\neg P \rightarrow \text{false}$ )
  : P :=
not_not.mp ( $\lambda$  h2, h1 h2)

#print axioms not_not

-- 5ª demostración
example

```

```

(h1 : ¬P → false)
: P :=
by_contra h1

#print axioms by_contra

-- 6ª demostración
Lemma RAA
(h1 : ¬P → false)
: P :=
by finish

#print axioms RAA

```

2.6.2. Pruebas de la eliminación de la doble negación

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de la eliminación de la doble negación
-- =====

-- Ej. 1. Demostrar
--      ¬¬P ⊢ P

import tactic

variable (P : Prop)

open_locale classical

-- 1ª demostración
example
(h1 : ¬¬P)
: P :=
by_contra
( assume h2 : ¬P,
  show false,
  from h1 h2 )

-- 2ª demostración
example
(h1 : ¬¬P)
: P :=

```

```

by_contra
  ( assume h2 :  $\neg$ P,
    h1 h2 )

-- 3ª demostración
example
  (h1 :  $\neg\neg$ P)
  : P :=
by_contra ( $\lambda$  h2, h1 h2)

-- 4ª demostración
example
  (h1 :  $\neg\neg$ P)
  : P :=
not_not.mp h1

-- 5ª demostración
example
  (h1 :  $\neg\neg$ P)
  : P :=
begin
  by_contradiction h2,
  exact h1 h2,
end

-- 6ª demostración
example
  (h1 :  $\neg\neg$ P)
  : P :=
by tauto

-- 7ª demostración
lemma aux
  (h1 :  $\neg\neg$ P)
  : P :=
by finish

#print axioms aux

```

2.6.3. Pruebas del principio del tercio excluso

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas del principio del tercio excluso
-- =====

-- Demostrar que
--    $F \vee \neg F$ 

import tactic

variable (F : Prop)

open_locale classical

-- 1ª demostración
example : F  $\vee$   $\neg F$  :=
by_contradiction
  ( assume h1 :  $\neg(F \vee \neg F)$ ,
    have h2 :  $\neg F$ , from
      assume h3 : F,
      have h4 : F  $\vee$   $\neg F$ , from or.inl h3,
      show false, from h1 h4,
    have h5 : F  $\vee$   $\neg F$ , from or.inr h2,
    show false, from h1 h5 )

-- 2ª demostración
example : F  $\vee$   $\neg F$  :=
by_contradiction
  ( assume h1 :  $\neg(F \vee \neg F)$ ,
    have h2 :  $\neg F$ , from
      assume h3 : F,
      have h4 : F  $\vee$   $\neg F$ , from or.inl h3,
      show false, from h1 h4,
    have h5 : F  $\vee$   $\neg F$ , from or.inr h2,
    h1 h5 )

-- 3ª demostración
example : F  $\vee$   $\neg F$  :=
by_contradiction
  ( assume h1 :  $\neg(F \vee \neg F)$ ,
    have h2 :  $\neg F$ , from
      assume h3 : F,
      have h4 : F  $\vee$   $\neg F$ , from or.inl h3,
      show false, from h1 h4,
    h1 (or.inr h2) )

-- 4ª demostración

```

```

example : F  $\vee$   $\neg$ F :=
by_contradiction
  ( assume h1 :  $\neg$ (F  $\vee$   $\neg$ F),
    have h2 :  $\neg$ F, from
      assume h3 : F,
      have h4 : F  $\vee$   $\neg$ F, from or.inl h3,
      h1 h4,
      h1 (or.inr h2) )

-- 5ª demostración
example : F  $\vee$   $\neg$ F :=
by_contradiction
  ( assume h1 :  $\neg$ (F  $\vee$   $\neg$ F),
    have h2 :  $\neg$ F, from
      assume h3 : F,
      h1 (or.inl h3),
      h1 (or.inr h2) )

-- 6ª demostración
example : F  $\vee$   $\neg$ F :=
by_contradiction
  ( assume h1 :  $\neg$ (F  $\vee$   $\neg$ F),
    have h2 :  $\neg$ F, from
       $\lambda$  h3, h1 (or.inl h3),
      h1 (or.inr h2) )

-- 7ª demostración
example : F  $\vee$   $\neg$ F :=
by_contradiction
  ( assume h1 :  $\neg$ (F  $\vee$   $\neg$ F),
    h1 (or.inr ( $\lambda$  h3, h1 (or.inl h3))) )

-- 8ª demostración
example : F  $\vee$   $\neg$ F :=
by_contradiction
  (  $\lambda$  h1, h1 (or.inr ( $\lambda$  h3, h1 (or.inl h3))) )

-- 9ª demostración
example : F  $\vee$   $\neg$ F :=
em F

#print axioms em

-- 10ª demostración
example : F  $\vee$   $\neg$ F :=

```

```
begin
  by_contra h1,
  apply h1,
  apply or.inr,
  intro h2,
  apply h1,
  exact or.inl h2,
end

-- 11ª demostración
example : F ∨ ¬F :=
begin
  by_contra h1,
  apply h1,
  apply or.inr,
  intro h2,
  exact h1 (or.inl h2),
end

-- 12ª demostración
example : F ∨ ¬F :=
begin
  by_contra h1,
  apply h1,
  apply or.inr,
  exact λ h2, h1 (or.inl h2),
end

-- 13ª demostración
example : F ∨ ¬F :=
begin
  by_contra h1,
  apply h1,
  exact or.inr (λ h2, h1 (or.inl h2)),
end

-- 14ª demostración
example : F ∨ ¬F :=
begin
  by_contra h1,
  exact h1 (or.inr (λ h2, h1 (or.inl h2))),
end

-- 15ª demostración
example : F ∨ ¬F :=
```

```

by_contra (λ h1, h1 (or.inr (λ h2, h1 (or.inl h2))))

-- 16ª demostración
example : F ∨ ¬F :=
begin
  by_contra h1,
  apply h1,
  right,
  intro h2,
  apply h1,
  left,
  exact h2,
end

-- 17ª demostración
example : F ∨ ¬F :=
by tauto

-- 18ª demostración
example : F ∨ ¬F :=
by finish

```

2.6.4. Pruebas de $P \rightarrow Q \vdash \neg P \vee Q$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de  $P \rightarrow Q \vdash \neg P \vee Q$ 
-- =====

-- Ej. 1. Demostrar
--    $P \rightarrow Q \vdash \neg P \vee Q$ 

import tactic

variables (P Q : Prop)

open_locale classical

-- 1ª demostración
example
  (h1 : P → Q)
  : ¬P ∨ Q :=
have h2 : P ∨ ¬P,

```

```

    from em P,
or.elim h2
  ( assume h3 : P,
    have h4 : Q,
      from h1 h3,
    show  $\neg P \vee Q$ ,
      from or.inr h4)
  ( assume h5 :  $\neg P$ ,
    show  $\neg P \vee Q$ ,
      from or.inl h5)

-- 2ª demostración
example
  (h1 :  $P \rightarrow Q$ )
  :  $\neg P \vee Q :=$ 
have h2 :  $P \vee \neg P$ ,
  from em P,
or.elim h2
  ( assume h3 : P,
    have h4 : Q,
      from h1 h3,
    show  $\neg P \vee Q$ ,
      from or.inr h4)
  ( assume h5 :  $\neg P$ ,
    or.inl h5)

-- 3ª demostración
example
  (h1 :  $P \rightarrow Q$ )
  :  $\neg P \vee Q :=$ 
have h2 :  $P \vee \neg P$ ,
  from em P,
or.elim h2
  ( assume h3 : P,
    have h4 : Q,
      from h1 h3,
    show  $\neg P \vee Q$ ,
      from or.inr h4)
  (  $\lambda$  h5, or.inl h5)

-- 4ª demostración
example
  (h1 :  $P \rightarrow Q$ )
  :  $\neg P \vee Q :=$ 
have h2 :  $P \vee \neg P$ ,

```



```

    from em P,
or.elim h2
  ( assume h3 : P,
    have h4 : Q,
      from h1 h3,
    or.inr h4)
  ( λ h5, or.inl h5)

-- 5ª demostración
example
  (h1 : P → Q)
  : ¬P ∨ Q :=
have h2 : P ∨ ¬P,
  from em P,
or.elim h2
  ( assume h3 : P,
    or.inr (h1 h3))
  ( λ h5, or.inl h5)

-- 6ª demostración
example
  (h1 : P → Q)
  : ¬P ∨ Q :=
have h2 : P ∨ ¬P,
  from em P,
or.elim h2
  ( λ h3, or.inr (h1 h3))
  ( λ h5, or.inl h5)

-- 7ª demostración
example
  (h1 : P → Q)
  : ¬P ∨ Q :=
or.elim (em P)
  ( λ h3, or.inr (h1 h3))
  ( λ h5, or.inl h5)

example
  (h1 : P → Q)
  : ¬P ∨ Q :=
(em P).elim
  ( λ h3, or.inr (h1 h3))
  ( λ h5, or.inl h5)

-- 8ª demostración

```

```

example
  (h1 : P → Q)
  : ¬P ∨ Q :=
-- by library_search
not_or_of_imp h1

-- 9ª demostración
example
  (h1 : P → Q)
  : ¬P ∨ Q :=
if h3 : P then or.inr (h1 h3) else or.inl h3

-- 10ª demostración
example
  (h1 : P → Q)
  : ¬P ∨ Q :=
begin
  by_cases h2 : P,
  { apply or.inr,
    exact h1 h2, },
  { exact or.inl h2, },
end

-- 11ª demostración
example
  (h1 : P → Q)
  : ¬P ∨ Q :=
begin
  by_cases h2 : P,
  { exact or.inr (h1 h2), },
  { exact or.inl h2, },
end

-- 12ª demostración
example
  (h1 : P → Q)
  : ¬P ∨ Q :=
begin
  by_cases h2 : P,
  { right,
    exact h1 h2, },
  { left,
    exact h2, },
end

```

```
-- 13ª demostración
```

```
example
```

```
(h1 : P → Q)
```

```
: ¬P ∨ Q :=
```

```
-- by hint
```

```
by tauto
```

```
-- 14ª demostración
```

```
example
```

```
(h1 : P → Q)
```

```
: ¬P ∨ Q :=
```

```
-- by hint
```

```
by finish
```

2.6.5. Pruebas de $P, \neg\neg(Q \wedge R) \vdash \neg\neg P \wedge R$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al vídeo.

```
-- Pruebas de  $P, \neg\neg(Q \wedge R) \vdash \neg\neg P \wedge R$ 
```

```
-- =====
```

```
-- Ej. 1. Demostrar
```

```
--  $P, \neg\neg(Q \wedge R) \vdash \neg\neg P \wedge R$ 
```

```
import tactic
```

```
variables (P Q R : Prop)
```

```
open_locale classical
```

```
-- 1ª demostración
```

```
example
```

```
(h1 : P)
```

```
(h2 : ¬¬(Q ∧ R))
```

```
: ¬¬P ∧ R :=
```

```
have h3 : ¬¬P, from not_not_intro h1,
```

```
have h4 : Q ∧ R, from not_not.mp h2,
```

```
have h5 : R, from and.elim_right h4,
```

```
show ¬¬P ∧ R, from and.intro h3 h5
```

```
-- 2ª demostración
```

```
example
```

```
(h1 : P)
```

```

(h2 :  $\neg\neg(Q \wedge R)$ )
:  $\neg\neg P \wedge R :=$ 
have h3 :  $\neg\neg P$ , from not_not_intro h1,
have h4 :  $Q \wedge R$ , from not_not.mp h2,
have h5 :  $R$ , from and.elim_right h4,
and.intro h3 h5

-- 3ª demostración
example
(h1 : P)
(h2 :  $\neg\neg(Q \wedge R)$ )
:  $\neg\neg P \wedge R :=$ 
have h3 :  $\neg\neg P$ , from not_not_intro h1,
have h4 :  $Q \wedge R$ , from not_not.mp h2,
have h5 :  $R$ , from h4.2,
and.intro h3 h5

-- 5ª demostración
example
(h1 : P)
(h2 :  $\neg\neg(Q \wedge R)$ )
:  $\neg\neg P \wedge R :=$ 
and.intro (not_not_intro h1) (not_not.mp h2).2

-- 6ª demostración
example
(h1 : P)
(h2 :  $\neg\neg(Q \wedge R)$ )
:  $\neg\neg P \wedge R :=$ 
begin
  split,
  { exact not_not_intro h1, },
  { push_neg at h2,
    exact h2.2, },
end

-- 7ª demostración
example
(h1 : P)
(h2 :  $\neg\neg(Q \wedge R)$ )
:  $\neg\neg P \wedge R :=$ 
-- by hint
by tauto

-- 8ª demostración

```

```

lemma aux
  (h1 : P)
  (h2 :  $\neg\neg(Q \wedge R)$ )
  :  $\neg\neg P \wedge R$  :=
by finish

#print axioms aux

```

2.6.6. Pruebas de $\neg P \rightarrow Q$, $\neg Q \vdash P$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al vídeo.

```

-- Pruebas de  $\neg P \rightarrow Q$ ,  $\neg Q \vdash P$ 
-- =====

-- Ej. 1. Demostrar
--    $\neg P \rightarrow Q$ ,  $\neg Q \vdash P$ 

import tactic

variables (P Q : Prop)

open_locale classical

-- 1ª demostración
example
  (h1 :  $\neg P \rightarrow Q$ )
  (h2 :  $\neg Q$ )
  : P :=
have h3 :  $\neg\neg P$ , from mt h1 h2,
show P, from not_not.mp h3

-- 2ª demostración
example
  (h1 :  $\neg P \rightarrow Q$ )
  (h2 :  $\neg Q$ )
  : P :=
not_not.mp (mt h1 h2)

-- 3ª demostración
example
  (h1 :  $\neg P \rightarrow Q$ )
  (h2 :  $\neg Q$ )

```

```

: P :=
begin
  by_contra h3,
  apply h2,
  exact h1 h3,
end

-- 4ª demostración
example
  (h1 :  $\neg P \rightarrow Q$ )
  (h2 :  $\neg Q$ )
  : P :=
begin
  by_contra h3,
  exact h2 (h1 h3),
end

-- 5ª demostración
example
  (h1 :  $\neg P \rightarrow Q$ )
  (h2 :  $\neg Q$ )
  : P :=
by_contra ( $\lambda$  h3, h2 (h1 h3))

-- 6ª demostración
example
  (h1 :  $\neg P \rightarrow Q$ )
  (h2 :  $\neg Q$ )
  : P :=
by_contra ( $\lambda$  h3, (h2  $\circ$  h1) h3)

-- 7ª demostración
example
  (h1 :  $\neg P \rightarrow Q$ )
  (h2 :  $\neg Q$ )
  : P :=
by_contra (h2  $\circ$  h1)

-- 8ª demostración
example
  (h1 :  $\neg P \rightarrow Q$ )
  (h2 :  $\neg Q$ )
  : P :=
-- by library_search
not_not.mp (mt h1 h2)

```

```

-- 9ª demostración
example
  (h1 : ¬P → Q)
  (h2 : ¬Q)
  : P :=
-- by hint
by tauto

-- 10ª demostración
lemma aux
  (h1 : ¬P → Q)
  (h2 : ¬Q)
  : P :=
-- by hint
by finish

#print axioms aux

```

2.6.7. Pruebas de $(Q \rightarrow R) \rightarrow ((\neg Q \rightarrow \neg P) \rightarrow (P \rightarrow R))$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al vídeo.

```

-- Pruebas de  $(Q \rightarrow R) \rightarrow ((\neg Q \rightarrow \neg P) \rightarrow (P \rightarrow R))$ 
-- =====

-- Ej. 1. Demostrar
--  $(Q \rightarrow R) \rightarrow ((\neg Q \rightarrow \neg P) \rightarrow (P \rightarrow R))$ 

import tactic

variables (P Q R : Prop)

-- 1ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
assume h1 : Q → R,
show (¬Q → ¬P) → (P → R), from
  ( assume h2 : ¬Q → ¬P,
    show P → R, from
      ( assume h3 : P,
        have h4 : ¬¬P, from not_not_intro h3,
        have h5 : ¬¬Q, from mt h2 h4,

```

```

    have h6 : Q, from not_not.mp h5,
    show R, from h1 h6))

-- 2ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
assume h1 : Q → R,
show (¬Q → ¬P) → (P → R), from
  ( assume h2 : ¬Q → ¬P,
    show P → R, from
      ( assume h3 : P,
        have h4 : ¬¬P, from not_not_intro h3,
        have h5 : ¬¬Q, from mt h2 h4,
        have h6 : Q, from not_not.mp h5,
        h1 h6))

-- 3ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
assume h1 : Q → R,
show (¬Q → ¬P) → (P → R), from
  ( assume h2 : ¬Q → ¬P,
    show P → R, from
      ( assume h3 : P,
        have h4 : ¬¬P, from not_not_intro h3,
        have h5 : ¬¬Q, from mt h2 h4,
        h1 (not_not.mp h5)))

-- 4ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
assume h1 : Q → R,
show (¬Q → ¬P) → (P → R), from
  ( assume h2 : ¬Q → ¬P,
    show P → R, from
      ( assume h3 : P,
        have h4 : ¬¬P, from not_not_intro h3,
        h1 (not_not.mp (mt h2 h4))))

-- 5ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
assume h1 : Q → R,
show (¬Q → ¬P) → (P → R), from
  ( assume h2 : ¬Q → ¬P,

```



```

    show P → R, from
      ( assume h3 : P,
        h1 (not_not.mp (mt h2 (not_not_intro h3))))))

-- 6ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
assume h1 : Q → R,
show (¬Q → ¬P) → (P → R), from
  ( assume h2 : ¬Q → ¬P,
    show P → R, from
      (λh3, h1 (not_not.mp (mt h2 (not_not_intro h3))))))

-- 7ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
assume h1 : Q → R,
show (¬Q → ¬P) → (P → R), from
  ( assume h2 : ¬Q → ¬P,
    (λh3, h1 (not_not.mp (mt h2 (not_not_intro h3))))))

-- 8ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
assume h1 : Q → R,
show (¬Q → ¬P) → (P → R), from
  ( λh2,
    (λh3, h1 (not_not.mp (mt h2 (not_not_intro h3))))))

-- 9ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
assume h1 : Q → R,
(λ h2 h3, h1 (not_not.mp (mt h2 (not_not_intro h3))))

-- 10ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
λ h1 h2 h3, h1 (not_not.mp (mt h2 (not_not_intro h3)))

-- 11ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
begin
  intro h1,

```

```

    intro h2,
    intro h3,
    apply h1,
    apply not_not.mp,
    apply mt h2,
    exact not_not_intro h3,
end

-- 12ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
begin
  intros h1 h2 h3,
  apply h1,
  apply not_not.mp,
  apply mt h2,
  exact not_not_intro h3,
end

-- 13ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
begin
  intros h1 h2 h3,
  apply h1,
  apply not_not.mp,
  exact mt h2 (not_not_intro h3),
end

-- 14ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
begin
  intros h1 h2 h3,
  exact h1 (not_not.mp (mt h2 (not_not_intro h3))),
end

-- 15ª demostración
example :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=
λ h1 h2 h3, h1 (not_not.mp (mt h2 (not_not_intro h3)))

-- 16ª demostración
lemma aux :
  (Q → R) → ((¬Q → ¬P) → (P → R)) :=

```

```
-- by hint  
by finish  
  
#print axioms aux
```


Capítulo 3

Lógica de primer orden

3.1. Reglas del cuantificador universal

3.1.1. Regla de eliminación del cuantificador universal

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Regla de eliminación del cuantificador universal
-- =====

-- Ej. 1. Demostrar
--    $P(c), \forall x (P(x) \rightarrow \neg Q(x)) \vdash \neg Q(c)$ 

import tactic

variable U : Type
variable c : U
variables P Q : U → Prop

-- 1ª demostración
example
  (h1 : P c)
  (h2 :  $\forall x, P x \rightarrow \neg Q x$ )
  :  $\neg Q c :=$ 
have h3 : P c  $\rightarrow \neg Q c$ , from h2 c,
show  $\neg Q c$ , from h3 h1

-- 2ª demostración
example
  (h1 : P c)
  (h2 :  $\forall x, P x \rightarrow \neg Q x$ )
```

```

:  $\neg Q\ c :=$ 
have h3 :  $P\ c \rightarrow \neg Q\ c$ , from h2 c,
h3 h1

-- 3ª demostración
example
  (h1 :  $P\ c$ )
  (h2 :  $\forall x, P\ x \rightarrow \neg Q\ x$ )
  :  $\neg Q\ c :=$ 
  (h2 c) h1

-- 4ª demostración
example
  (h1 :  $P\ c$ )
  (h2 :  $\forall x, P\ x \rightarrow \neg Q\ x$ )
  :  $\neg Q\ c :=$ 
  -- by library_search
  h2 c h1

-- 5ª demostración
example
  (h1 :  $P\ c$ )
  (h2 :  $\forall x, P\ x \rightarrow \neg Q\ x$ )
  :  $\neg Q\ c :=$ 
  -- by hint
  by tauto

-- 6ª demostración
example
  (h1 :  $P\ c$ )
  (h2 :  $\forall x, P\ x \rightarrow \neg Q\ x$ )
  :  $\neg Q\ c :=$ 
  by finish

-- 7ª demostración
example
  (h1 :  $P\ c$ )
  (h2 :  $\forall x, P\ x \rightarrow \neg Q\ x$ )
  :  $\neg Q\ c :=$ 
begin
  apply h2,
  exact h1,
end

```

3.1.2. Regla de introducción del cuantificador universal

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Regla de introducción del cuantificador universal
-- =====

-- Ej. 1. Demostrar
--    $\forall x [P(x) \rightarrow \neg Q(x)], \forall x P(x) \vdash \forall x \neg Q(x)$ 

import tactic

variable U : Type
variables P Q : U → Prop

-- 1ª demostración
example
  (h1 :  $\forall x, P\ x \rightarrow \neg Q\ x$ )
  (h2 :  $\forall x, P\ x$ )
  :  $\forall x, \neg Q\ x :=$ 
  assume x₀,
  have h4 : P x₀ →  $\neg Q\ x₀$ , from h1 x₀,
  have h5 : P x₀,           from h2 x₀,
  show  $\neg Q\ x₀$ ,           from h4 h5

-- 2ª demostración
example
  (h1 :  $\forall x, P\ x \rightarrow \neg Q\ x$ )
  (h2 :  $\forall x, P\ x$ )
  :  $\forall x, \neg Q\ x :=$ 
  assume x₀, (h1 x₀) (h2 x₀)

-- 3ª demostración
example
  (h1 :  $\forall x, P\ x \rightarrow \neg Q\ x$ )
  (h2 :  $\forall x, P\ x$ )
  :  $\forall x, \neg Q\ x :=$ 
   $\lambda\ x₀, (h1\ x₀) (h2\ x₀)$ 

-- 4ª demostración
example
  (h1 :  $\forall x, P\ x \rightarrow \neg Q\ x$ )
  (h2 :  $\forall x, P\ x$ )
  :  $\forall x, \neg Q\ x :=$ 
  begin
```

```

    intro x₀,
    apply h1,
    apply h2,
end

-- 5ª demostración
example
  (h1 :  $\forall x, P\ x \rightarrow \neg Q\ x$ )
  (h2 :  $\forall x, P\ x$ )
  :  $\forall x, \neg Q\ x :=$ 
begin
  intro x₀,
  specialize h1 x₀,
  specialize h2 x₀,
  apply h1,
  exact h2,
end

-- 6ª demostración
example
  (h1 :  $\forall x, P\ x \rightarrow \neg Q\ x$ )
  (h2 :  $\forall x, P\ x$ )
  :  $\forall x, \neg Q\ x :=$ 
begin
  intro x₀,
  specialize h1 x₀,
  specialize h2 x₀,
  exact h1 h2,
end

-- 7ª demostración
example
  (h1 :  $\forall x, P\ x \rightarrow \neg Q\ x$ )
  (h2 :  $\forall x, P\ x$ )
  :  $\forall x, \neg Q\ x :=$ 
-- by hint
by tauto

-- 8ª demostración
example
  (h1 :  $\forall x, P\ x \rightarrow \neg Q\ x$ )
  (h2 :  $\forall x, P\ x$ )
  :  $\forall x, \neg Q\ x :=$ 
by finish

```


3.2. Reglas del cuantificador existencial

3.2.1. Regla de introducción del cuantificador existencial

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Regla de introducción del cuantificador existencial
-- =====

-- Ej. 1. Demostrar
--    $\forall x P(x) \vdash \exists x P(x)$ 

import tactic

variable U : Type
variable c : U
variable P : U -> Prop

-- 1ª demostración
example
  (h1 :  $\forall x, P x$ )
  :  $\exists x, P x :=$ 
have h2 : P c, from h1 c,
show  $\exists x, P x$ , from exists.intro c h2

-- 2ª demostración
example
  (h1 :  $\forall x, P x$ )
  :  $\exists x, P x :=$ 
show  $\exists x, P x$ , from exists.intro c (h1 c)

-- 3ª demostración
example
  (h1 :  $\forall x, P x$ )
  :  $\exists x, P x :=$ 
exists.intro c (h1 c)

-- 4ª demostración
example
  (h1 :  $\forall x, P x$ )
  :  $\exists x, P x :=$ 
⟨c, h1 c⟩
```

```

-- 5ª demostración
example
  (a : U)
  (h1 :  $\forall x, P\ x$ )
  :  $\exists x, P\ x :=$ 
begin
  use a,
  apply h1,
end

-- 6ª demostración
example
  (a : U)
  (h1 :  $\forall x, P\ x$ )
  :  $\exists x, P\ x :=$ 
begin
  constructor,
  apply h1 a,
end

-- 7ª demostración
example
  [inhabited U]
  (h1 :  $\forall x, P\ x$ )
  :  $\exists x, P\ x :=$ 
begin
  constructor,
  apply h1 (default U),
end

-- 8ª demostración
example
  (h : nonempty U)
  (h1 :  $\forall x, P\ x$ )
  :  $\exists x, P\ x :=$ 
begin
  use (classical.choice h),
  apply h1,
end

```

3.2.2. Regla de eliminación del cuantificador existencial

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Regla de eliminación del cuantificador existencial
-- =====

-- Ej. 1. Demostrar
--    $\forall x [P(x) \rightarrow Q(x)], \exists x P(x) \vdash \exists x Q(x)$ 

import tactic

variable U : Type
variables P Q : U -> Prop

-- 1ª demostración
-- =====

example
  (h1 :  $\forall x, P\ x \rightarrow Q\ x$ )
  (h2 :  $\exists x, P\ x$ )
  :  $\exists x, Q\ x :=$ 
exists.elim h2
  ( assume x0 (h3 : P x0),
    have h4 : P x0  $\rightarrow$  Q x0, from h1 x0,
    have h5 : Q x0,          from h4 h3,
    show  $\exists x, Q\ x$ ,          from exists.intro x0 h5 )

-- 2ª demostración
-- =====

example
  (h1 :  $\forall x, P\ x \rightarrow Q\ x$ )
  (h2 :  $\exists x, P\ x$ )
  :  $\exists x, Q\ x :=$ 
exists.elim h2
  ( assume x0 (h3 : P x0),
    have h4 : P x0  $\rightarrow$  Q x0, from h1 x0,
    have h5 : Q x0,          from h4 h3,
    show  $\exists x, Q\ x$ ,          from  $\langle x0, h5 \rangle$  )

-- 3ª demostración
-- =====

example
  (h1 :  $\forall x, P\ x \rightarrow Q\ x$ )
  (h2 :  $\exists x, P\ x$ )
  :  $\exists x, Q\ x :=$ 
exists.elim h2

```

```

( assume x0 (h3 : P x0),
  have h4 : P x0 → Q x0, from h1 x0,
  have h5 : Q x0, from h4 h3,
  ⟨x0, h5⟩ )

-- 4ª demostración
-- =====

example
(h1 : ∀x, P x → Q x)
(h2 : ∃x, P x)
: ∃x, Q x :=
exists.elim h2
( assume x0 (h3 : P x0),
  have h4 : P x0 → Q x0, from h1 x0,
  ⟨x0, h4 h3⟩ )

-- 5ª demostración
-- =====

example
(h1 : ∀x, P x → Q x)
(h2 : ∃x, P x)
: ∃x, Q x :=
exists.elim h2
( assume x0 (h3 : P x0),
  ⟨x0, h1 x0 h3⟩ )

-- 6ª demostración
-- =====

example
(h1 : ∀x, P x → Q x)
(h2 : ∃x, P x)
: ∃x, Q x :=
exists.elim h2 (λ x0 h3, ⟨x0, h1 x0 h3⟩)

-- 7ª demostración
-- =====

example
(h1 : ∀x, P x → Q x)
(h2 : ∃x, P x)
: ∃x, Q x :=
-- by library_search

```

```

Exists.imp h1 h2

-- 8ª demostración
-- =====

example
  (h1 :  $\forall x, P x \rightarrow Q x$ )
  (h2 :  $\exists x, P x$ )
  :  $\exists x, Q x$  :=
begin
  cases h2 with x0 h3,
  use x0,
  apply h1,
  exact h3,
end

-- 9ª demostración
-- =====

example
  (h1 :  $\forall x, P x \rightarrow Q x$ )
  (h2 :  $\exists x, P x$ )
  :  $\exists x, Q x$  :=
begin
  cases h2 with x0 h3,
  use x0,
  specialize h1 x0,
  apply h1,
  exact h3,
end

-- 10ª demostración
-- =====

example
  (h1 :  $\forall x, P x \rightarrow Q x$ )
  (h2 :  $\exists x, P x$ )
  :  $\exists x, Q x$  :=
-- by hint
by tauto

-- 11ª demostración
-- =====

example

```

```

(h1 : ∀x, P x → Q x)
(h2 : ∃x, P x)
: ∃x, Q x :=
by finish

```

3.3. Ejercicios sobre cuantificadores

3.3.1. Pruebas de $\neg\forall x P(x) \leftrightarrow \exists x \neg P(x)$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de  $\neg\forall x P(x) \leftrightarrow \exists x \neg P(x)$ 
-- =====

import tactic

variable {U : Type}
variable {P : U -> Prop}

-----

-- Ej. 1. Demostrar que
--       $\neg\forall x P(x) \vdash \exists x \neg P(x)$ 
-- -----

-- 1ª demostración
example
  (h1 : ¬∀x, P x)
  : ∃x, ¬P x :=
by_contra
  ( assume h2 : ¬∃x, ¬P x,
    have h8 : ∀x, P x, from
      ( assume x0,
        show P x0, from
          by_contra
            ( assume h4 : ¬P x0,
              have h5 : ∃x, ¬P x, from exists.intro x0 h4,
              show false, from h2 h5 )),
        show false, from h1 h8)

-- 2ª demostración
example
  (h1 : ¬∀x, P x)

```

```

:  $\exists x, \neg P x :=$ 
by_contra
  ( assume h2 :  $\neg \exists x, \neg P x$ ,
    have h8 :  $\forall x, P x$ , from
      ( assume x0,
        show  $P x_0$ , from
          by_contra
            ( assume h4 :  $\neg P x_0$ ,
              have h5 :  $\exists x, \neg P x$ , from exists.intro x0 h4,
              show false, from h2 h5 )) ,
      h1 h8)

-- 3ª demostración
example
  (h1 :  $\neg \forall x, P x$ )
  :  $\exists x, \neg P x :=$ 
by_contra
  ( assume h2 :  $\neg \exists x, \neg P x$ ,
    have h8 :  $\forall x, P x$ , from
      ( assume x0,
        show  $P x_0$ , from
          by_contra
            ( assume h4 :  $\neg P x_0$ ,
              have h5 :  $\exists x, \neg P x$ , from exists.intro x0 h4,
              h2 h5 )) ,
      h1 h8)

-- 4ª demostración
example
  (h1 :  $\neg \forall x, P x$ )
  :  $\exists x, \neg P x :=$ 
by_contra
  ( assume h2 :  $\neg \exists x, \neg P x$ ,
    have h8 :  $\forall x, P x$ , from
      ( assume x0,
        show  $P x_0$ , from
          by_contra
            ( assume h4 :  $\neg P x_0$ ,
              have h5 :  $\exists x, \neg P x$ , from ⟨x0, h4⟩,
              h2 h5 )) ,
      h1 h8)

-- 5ª demostración
example
  (h1 :  $\neg \forall x, P x$ )

```

```

:  $\exists x, \neg P x :=$ 
by_contra
  ( assume h2 :  $\neg \exists x, \neg P x$ ,
    have h8 :  $\forall x, P x$ , from
      ( assume x0,
        show  $P x_0$ , from
          by_contra
            ( assume h4 :  $\neg P x_0$ ,
              h2  $\langle x_0, h4 \rangle$  ) ),
        h1 h8)

-- 6ª demostración
example
  (h1 :  $\neg \forall x, P x$ )
  :  $\exists x, \neg P x :=$ 
by_contra
  ( assume h2 :  $\neg \exists x, \neg P x$ ,
    have h8 :  $\forall x, P x$ , from
      ( assume x0,
        show  $P x_0$ , from
          by_contra ( $\lambda h4, h2 \langle x_0, h4 \rangle$ )),
        h1 h8)

-- 7ª demostración
example
  (h1 :  $\neg \forall x, P x$ )
  :  $\exists x, \neg P x :=$ 
by_contra
  ( assume h2 :  $\neg \exists x, \neg P x$ ,
    have h8 :  $\forall x, P x$ , from
      ( assume x0,
        by_contra ( $\lambda h4, h2 \langle x_0, h4 \rangle$ )),
        h1 h8)

-- 8ª demostración
example
  (h1 :  $\neg \forall x, P x$ )
  :  $\exists x, \neg P x :=$ 
by_contra
  ( assume h2 :  $\neg \exists x, \neg P x$ ,
    have h8 :  $\forall x, P x$ , from
      ( $\lambda x_0, \text{by\_contra } (\lambda h4, h2 \langle x_0, h4 \rangle)$ ),
        h1 h8)

-- 9ª demostración

```



```

example
  (h1 :  $\neg \forall x, P x$ )
  :  $\exists x, \neg P x :=$ 
by_contra
  ( assume h2 :  $\neg \exists x, \neg P x$ ,
    h1 ( $\lambda x_0$ , by_contra ( $\lambda h4$ , h2  $\langle x_0, h4 \rangle$ )))

-- 10ª demostración
example
  (h1 :  $\neg \forall x, P x$ )
  :  $\exists x, \neg P x :=$ 
by_contra ( $\lambda h2$ , h1 ( $\lambda x_0$ , by_contra ( $\lambda h4$ , h2  $\langle x_0, h4 \rangle$ )))

-- 11ª demostración
example
  (h1 :  $\neg \forall x, P x$ )
  :  $\exists x, \neg P x :=$ 
-- by library_search
not_forall.mp h1

-- 12ª demostración
lemma aux1
  (h1 :  $\neg \forall x, P x$ )
  :  $\exists x, \neg P x :=$ 
-- by hint
by finish

-----

-- Ej. 2. Demostrar que
--    $\exists x \neg P(x) \vdash \neg \forall x P(x)$ 
--   -----

-- 1ª demostración
example
  (h1 :  $\exists x, \neg P x$ )
  :  $\neg \forall x, P x :=$ 
assume h2 :  $\forall x, P x$ ,
exists.elim h1
  ( assume x0 (h3 :  $\neg P x_0$ ),
    have h4 :  $P x_0$ , from h2 x0,
    show false, from h3 h4)

-- 2ª demostración
example
  (h1 :  $\exists x, \neg P x$ )

```

```

:  $\neg\forall x, P\ x :=$ 
assume h2 :  $\forall x, P\ x,$ 
exists.elim h1
  ( assume x0 (h3 :  $\neg P\ x_0$ ),
    have h4 :  $P\ x_0$ , from h2 x0,
    h3 h4)

-- 3ª demostración
example
  (h1 :  $\exists x, \neg P\ x$ )
  :  $\neg\forall x, P\ x :=$ 
assume h2 :  $\forall x, P\ x,$ 
exists.elim h1
  ( assume x0 (h3 :  $\neg P\ x_0$ ),
    h3 (h2 x0))

-- 4ª demostración
example
  (h1 :  $\exists x, \neg P\ x$ )
  :  $\neg\forall x, P\ x :=$ 
assume h2 :  $\forall x, P\ x,$ 
exists.elim h1
  ( $\lambda\ x_0\ h3, h3\ (h2\ x_0)$ )

-- 5ª demostración
example
  (h1 :  $\exists x, \neg P\ x$ )
  :  $\neg\forall x, P\ x :=$ 
 $\lambda\ h2, exists.elim\ h1\ (\lambda\ x_0\ h3, h3\ (h2\ x_0))$ 

-- 6ª demostración
example
  (h1 :  $\exists x, \neg P\ x$ )
  :  $\neg\forall x, P\ x :=$ 
-- by library_search
not_forall.mpr h1

-- 7ª demostración
example
  (h1 :  $\exists x, \neg P\ x$ )
  :  $\neg\forall x, P\ x :=$ 
assume h2 :  $\forall x, P\ x,$ 
match h1 with  $\langle x_0, (h3 : \neg P\ x_0) \rangle :=$ 
  ( have h4 :  $P\ x_0$ , from h2 x0,
    show false, from h3 h4)

```

```

end

-- 8ª demostración
example
  (h1 :  $\exists x, \neg P x$ )
  :  $\neg \forall x, P x :=$ 
begin
  intro h2,
  cases h1 with x0 h3,
  apply h3,
  apply h2,
end

example
  (h1 :  $\exists x, \neg P x$ )
  :  $\neg \forall x, P x :=$ 
begin
  intro h2,
  obtain ⟨x0, h3⟩ := h1,
  apply h3,
  apply h2,
end

-- 9ª demostración
example
  (h1 :  $\exists x, \neg P x$ )
  :  $\neg \forall x, P x :=$ 
-- by hint
by tauto

-- 10ª demostración
lemma aux2
  (h1 :  $\exists x, \neg P x$ )
  :  $\neg \forall x, P x :=$ 
by finish

#print axioms aux2

-----
-- Ej. 3. Demostrar que
--  $\neg \forall x P(x) \leftrightarrow \exists x \neg P(x)$ 
-----

-- 1ª demostración
example :

```

```

( $\neg \forall x, P\ x$ )  $\leftrightarrow$  ( $\exists x, \neg P\ x$ ) :=
iff.intro
  ( assume h1 :  $\neg \forall x, P\ x$ ,
    show  $\exists x, \neg P\ x$ , from aux1 h1)
  ( assume h2 :  $\exists x, \neg P\ x$ ,
    show  $\neg \forall x, P\ x$ , from aux2 h2)

-- 2ª demostración
example :
  ( $\neg \forall x, P\ x$ )  $\leftrightarrow$  ( $\exists x, \neg P\ x$ ) :=
iff.intro aux1 aux2

-- 3ª demostración
example :
  ( $\neg \forall x, P\ x$ )  $\leftrightarrow$  ( $\exists x, \neg P\ x$ ) :=
-- by library_search
not_forall

-- 4ª demostración
example :
  ( $\neg \forall x, P\ x$ )  $\leftrightarrow$  ( $\exists x, \neg P\ x$ ) :=
begin
  split,
  { exact aux1, },
  { exact aux2, },
end

-- 5ª demostración
example :
  ( $\neg \forall x, P\ x$ )  $\leftrightarrow$  ( $\exists x, \neg P\ x$ ) :=
-- by hint
by finish

```

3.3.2. Pruebas de $\forall x (P(x) \wedge Q(x)) \leftrightarrow \forall x P(x) \wedge \forall x Q(x)$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de  $\forall x (P(x) \wedge Q(x)) \leftrightarrow \forall x P(x) \wedge \forall x Q(x)$ 
-- =====

import tactic

section

```

```

variable {U : Type}
variables {P Q : U -> Prop}

-----

-- Ej. 1. Demostrar
--    $\forall x (P(x) \wedge Q(x)) \vdash \forall x P(x) \wedge \forall x Q(x)$ 
-----

-- 1ª demostración
example
  (h1 :  $\forall x, P\ x \wedge Q\ x$ )
  : ( $\forall x, P\ x$ )  $\wedge$  ( $\forall x, Q\ x$ ) :=
have h5 :  $\forall x, P\ x$ , from
  assume x0,
  have h3 :  $P\ x_0 \wedge Q\ x_0$ , from h1 x0,
  show P x0, from and.elim_left h3,
have h9 :  $\forall x, Q\ x$ , from
  assume x1,
  have h7 :  $P\ x_1 \wedge Q\ x_1$ , from h1 x1,
  show Q x1, from and.elim_right h7,
show ( $\forall x, P\ x$ )  $\wedge$  ( $\forall x, Q\ x$ ), from and.intro h5 h9

-- 2ª demostración
example
  (h1 :  $\forall x, P\ x \wedge Q\ x$ )
  : ( $\forall x, P\ x$ )  $\wedge$  ( $\forall x, Q\ x$ ) :=
have h5 :  $\forall x, P\ x$ , from
  assume x0,
  have h3 :  $P\ x_0 \wedge Q\ x_0$ , from h1 x0,
  show P x0, from h3.left,
have h9 :  $\forall x, Q\ x$ , from
  assume x1,
  have h7 :  $P\ x_1 \wedge Q\ x_1$ , from h1 x1,
  show Q x1, from h7.right,
show ( $\forall x, P\ x$ )  $\wedge$  ( $\forall x, Q\ x$ ), from ⟨h5, h9⟩

-- 3ª demostración
example
  (h1 :  $\forall x, P\ x \wedge Q\ x$ )
  : ( $\forall x, P\ x$ )  $\wedge$  ( $\forall x, Q\ x$ ) :=
have h5 :  $\forall x, P\ x$ , from
  assume x0,
  have h3 :  $P\ x_0 \wedge Q\ x_0$ , from h1 x0,
  h3.left,

```

```

have h9 :  $\forall x, Q\ x$ , from
  assume x1,
  have h7 :  $P\ x_1 \wedge Q\ x_1$ , from h1 x1,
  h7.right,
show  $(\forall x, P\ x) \wedge (\forall x, Q\ x)$ , from ⟨h5, h9⟩

-- 4ª demostración
example
  (h1 :  $\forall x, P\ x \wedge Q\ x$ )
  :  $(\forall x, P\ x) \wedge (\forall x, Q\ x) :=$ 
have h5 :  $\forall x, P\ x$ , from
  assume x0,
  (h1 x0).left,
have h9 :  $\forall x, Q\ x$ , from
  assume x1,
  (h1 x1).right,
show  $(\forall x, P\ x) \wedge (\forall x, Q\ x)$ , from ⟨h5, h9⟩

-- 5ª demostración
example
  (h1 :  $\forall x, P\ x \wedge Q\ x$ )
  :  $(\forall x, P\ x) \wedge (\forall x, Q\ x) :=$ 
have h5 :  $\forall x, P\ x$ , from
   $\lambda\ x_0, (h1\ x_0).left,$ 
have h9 :  $\forall x, Q\ x$ , from
   $\lambda\ x_1, (h1\ x_1).right,$ 
show  $(\forall x, P\ x) \wedge (\forall x, Q\ x)$ , from ⟨h5, h9⟩

-- 6ª demostración
example
  (h1 :  $\forall x, P\ x \wedge Q\ x$ )
  :  $(\forall x, P\ x) \wedge (\forall x, Q\ x) :=$ 
have h5 :  $\forall x, P\ x$ , from
   $\lambda\ x_0, (h1\ x_0).left,$ 
have h9 :  $\forall x, Q\ x$ , from
   $\lambda\ x_1, (h1\ x_1).right,$ 
⟨h5, h9⟩

-- 7ª demostración
example
  (h1 :  $\forall x, P\ x \wedge Q\ x$ )
  :  $(\forall x, P\ x) \wedge (\forall x, Q\ x) :=$ 
⟨ $\lambda\ x_0, (h1\ x_0).left, \lambda\ x_1, (h1\ x_1).right$ ⟩

-- 8ª demostración

```

```

example
  (h1 :  $\forall x, P\ x \wedge Q\ x$ )
  : ( $\forall x, P\ x$ )  $\wedge$  ( $\forall x, Q\ x$ ) :=
-- by library_search
forall_and_distrib.mp h1

-- 9ª demostración
example
  (h1 :  $\forall x, P\ x \wedge Q\ x$ )
  : ( $\forall x, P\ x$ )  $\wedge$  ( $\forall x, Q\ x$ ) :=
begin
  split,
  { intro x0,
    specialize h1 x0,
    exact h1.left, },
  { intro x1,
    specialize h1 x1,
    exact h1.right, },
end

-- 9ª demostración
lemma aux1
  (h1 :  $\forall x, P\ x \wedge Q\ x$ )
  : ( $\forall x, P\ x$ )  $\wedge$  ( $\forall x, Q\ x$ ) :=
-- by hint
by finish

-----

-- Ej. 2. Demostrar
--  $\forall x\ P(x) \wedge \forall x\ Q(x) \vdash \forall x\ (P(x) \wedge Q(x))$ 
-----

-- 1ª demostración
example
  (h1 : ( $\forall x, P\ x$ )  $\wedge$  ( $\forall x, Q\ x$ ))
  :  $\forall x, P\ x \wedge Q\ x$  :=
assume x0,
have h3 :  $\forall x, P\ x$ , from and.elim_left h1,
have h4 :  $P\ x_0$ , from h3 x0,
have h5 :  $\forall x, Q\ x$ , from and.elim_right h1,
have h6 :  $Q\ x_0$ , from h5 x0,
show  $P\ x_0 \wedge Q\ x_0$ , from and.intro h4 h6

-- 2ª demostración
example

```

```

(h1 : (∀x, P x) ∧ (∀x, Q x))
: ∀x, P x ∧ Q x :=
assume x₀,
have h3 : ∀x, P x, from h1.left,
have h4 : P x₀, from h3 x₀,
have h5 : ∀x, Q x, from h1.right,
have h6 : Q x₀, from h5 x₀,
show P x₀ ∧ Q x₀, from ⟨h4, h6⟩

-- 3ª demostración
example
(h1 : (∀x, P x) ∧ (∀x, Q x))
: ∀x, P x ∧ Q x :=
assume x₀,
have h3 : ∀x, P x, from h1.left,
have h4 : P x₀, from h3 x₀,
have h5 : ∀x, Q x, from h1.right,
have h6 : Q x₀, from h5 x₀,
⟨h4, h6⟩

-- 4ª demostración
example
(h1 : (∀x, P x) ∧ (∀x, Q x))
: ∀x, P x ∧ Q x :=
assume x₀,
have h3 : ∀x, P x, from h1.left,
have h4 : P x₀, from h3 x₀,
have h5 : ∀x, Q x, from h1.right,
⟨h4, h5 x₀⟩

-- 5ª demostración
example
(h1 : (∀x, P x) ∧ (∀x, Q x))
: ∀x, P x ∧ Q x :=
assume x₀,
have h3 : ∀x, P x, from h1.left,
have h4 : P x₀, from h3 x₀,
⟨h4, h1.right x₀⟩

-- 6ª demostración
example
(h1 : (∀x, P x) ∧ (∀x, Q x))
: ∀x, P x ∧ Q x :=
assume x₀,
have h3 : ∀x, P x, from h1.left,

```



```

⟨h3 x0, h1.right x0⟩

-- 7ª demostración
example
  (h1 : (∀x, P x) ∧ (∀x, Q x))
  : ∀x, P x ∧ Q x :=
assume x0,
⟨h1.left x0, h1.right x0⟩

-- 8ª demostración
example
  (h1 : (∀x, P x) ∧ (∀x, Q x))
  : ∀x, P x ∧ Q x :=
λ x0, ⟨h1.left x0, h1.right x0⟩

-- 9ª demostración
example
  (h1 : (∀x, P x) ∧ (∀x, Q x))
  : ∀x, P x ∧ Q x :=
-- by library_search
forall_and_distrib.mpr h1

-- 10ª demostración
example
  (h1 : (∀x, P x) ∧ (∀x, Q x))
  : ∀x, P x ∧ Q x :=
begin
  cases h1 with h2 h3,
  intro x0,
  split,
  { apply h2, },
  { apply h3, },
end

-- 11ª demostración
example
  (h1 : (∀x, P x) ∧ (∀x, Q x))
  : ∀x, P x ∧ Q x :=
-- by hint
by tauto

-- 12ª demostración
lemma aux2
  (h1 : (∀x, P x) ∧ (∀x, Q x))
  : ∀x, P x ∧ Q x :=

```

```

by finish

-- -----
-- Ej. 3. Demostrar
--    $\forall x (P(x) \wedge Q(x)) \leftrightarrow \forall x P(x) \wedge \forall x Q(x)$ 
-- -----

-- 1ª demostración
example :
  ( $\forall x, P\ x \wedge Q\ x$ )  $\leftrightarrow$  ( $\forall x, P\ x$ )  $\wedge$  ( $\forall x, Q\ x$ ) :=
iff.intro aux1 aux2

-- 2ª demostración
example :
  ( $\forall x, P\ x \wedge Q\ x$ )  $\leftrightarrow$  ( $\forall x, P\ x$ )  $\wedge$  ( $\forall x, Q\ x$ ) :=
-- by library_search
forall_and_distrib

-- 3ª demostración
example :
  ( $\forall x, P\ x \wedge Q\ x$ )  $\leftrightarrow$  ( $\forall x, P\ x$ )  $\wedge$  ( $\forall x, Q\ x$ ) :=
-- by hint
by finish

end

```

3.3.3. Pruebas de $\exists x (P(x) \vee Q(x)) \leftrightarrow \exists x P(x) \vee \exists x Q(x)$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de  $\exists x (P(x) \vee Q(x)) \leftrightarrow \exists x P(x) \vee \exists x Q(x)$ 
-- =====

import tactic

section

variable {U : Type}
variables {P Q : U -> Prop}

-- -----
-- Ej. 1. Demostrar
--    $\exists x (P(x) \vee Q(x)) \vdash \exists x P(x) \vee \exists x Q(x)$ 

```

-- 1ª demostración

example

```
(h1 :  $\exists x, P\ x \vee Q\ x$ )
: ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ) :=
exists.elim h1
( assume x0 (h2 :  $P\ x_0 \vee Q\ x_0$ ),
  or.elim h2
    ( assume h3 :  $P\ x_0$ ,
      have h4 :  $\exists x, P\ x$ , from exists.intro x0 h3,
      show ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ), from or.inl h4 )
    ( assume h6 :  $Q\ x_0$ ,
      have h7 :  $\exists x, Q\ x$ , from exists.intro x0 h6,
      show ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ), from or.inr h7 ))
```

-- 2ª demostración

example

```
(h1 :  $\exists x, P\ x \vee Q\ x$ )
: ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ) :=
exists.elim h1
( assume x0 (h2 :  $P\ x_0 \vee Q\ x_0$ ),
  or.elim h2
    ( assume h3 :  $P\ x_0$ ,
      have h4 :  $\exists x, P\ x$ , from ⟨x0, h3⟩,
      show ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ), from or.inl h4 )
    ( assume h6 :  $Q\ x_0$ ,
      have h7 :  $\exists x, Q\ x$ , from ⟨x0, h6⟩,
      show ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ), from or.inr h7 ))
```

-- 3ª demostración

example

```
(h1 :  $\exists x, P\ x \vee Q\ x$ )
: ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ) :=
exists.elim h1
( assume x0 (h2 :  $P\ x_0 \vee Q\ x_0$ ),
  or.elim h2
    ( assume h3 :  $P\ x_0$ ,
      have h4 :  $\exists x, P\ x$ , from ⟨x0, h3⟩,
      or.inl h4 )
    ( assume h6 :  $Q\ x_0$ ,
      have h7 :  $\exists x, Q\ x$ , from ⟨x0, h6⟩,
      or.inr h7 ))
```

-- 4ª demostración

```

example
  (h1 :  $\exists x, P\ x \vee Q\ x$ )
  : ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ) :=
exists.elim h1
  ( assume x0 (h2 :  $P\ x_0 \vee Q\ x_0$ ),
    or.elim h2
      ( assume h3 :  $P\ x_0$ ,
        or.inl  $\langle x_0, h3 \rangle$  )
      ( assume h6 :  $Q\ x_0$ ,
        or.inr  $\langle x_0, h6 \rangle$  ))

-- 5ª demostración
example
  (h1 :  $\exists x, P\ x \vee Q\ x$ )
  : ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ) :=
exists.elim h1
  ( assume x0 (h2 :  $P\ x_0 \vee Q\ x_0$ ),
    or.elim h2
      (  $\lambda$  h3, or.inl  $\langle x_0, h3 \rangle$  )
      (  $\lambda$  h6, or.inr  $\langle x_0, h6 \rangle$  ))

-- 6ª demostración
example
  (h1 :  $\exists x, P\ x \vee Q\ x$ )
  : ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ) :=
exists.elim h1
  (  $\lambda$  x0 h2, h2.elim (  $\lambda$  h3, or.inl  $\langle x_0, h3 \rangle$  )
                     (  $\lambda$  h6, or.inr  $\langle x_0, h6 \rangle$  ))

-- 7ª demostración
example
  (h1 :  $\exists x, P\ x \vee Q\ x$ )
  : ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ) :=
-- by library_search
exists_or_distrib.mp h1

-- 8ª demostración
example
  (h1 :  $\exists x, P\ x \vee Q\ x$ )
  : ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ) :=
match h1 with  $\langle x_0, (h2 : P\ x_0 \vee Q\ x_0) \rangle$  :=
  ( or.elim h2
    ( assume h3 :  $P\ x_0$ ,
      have h4 :  $\exists x, P\ x$ , from exists.intro x0 h3,
      show ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ), from or.inl h4 )

```

```

    ( assume h6 : Q x0,
      have h7 :  $\exists x, Q\ x$ , from exists.intro x0 h6,
      show ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ), from or.inr h7 ))
end

-- 9ª demostración
example
  (h1 :  $\exists x, P\ x \vee Q\ x$ )
  : ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ) :=
begin
  cases h1 with x0 h3,
  cases h3 with hp hq,
  { left,
    use x0,
    exact hp, },
  { right,
    use x0,
    exact hq, },
end

-- 10ª demostración
example
  (h1 :  $\exists x, P\ x \vee Q\ x$ )
  : ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ) :=
begin
  rcases h1 with ⟨x0, hp | hq⟩,
  { left,
    use x0,
    exact hp, },
  { right,
    use x0,
    exact hq, },
end

-- 11ª demostración
lemma aux1
  (h1 :  $\exists x, P\ x \vee Q\ x$ )
  : ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ) :=
-- by hint
by finish

-----
-- Ej. 2. Demostrar
--  $\exists x\ P(x) \vee \exists x\ Q(x) \vdash \exists x\ (P(x) \vee Q(x))$ 
-----

```

```

-- 1ª demostración
example
  (h1 : ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ))
  :  $\exists x, P\ x \vee Q\ x :=$ 
or.elim h1
  ( assume h2 :  $\exists x, P\ x$ ,
    exists.elim h2
      ( assume x0 (h3 :  $P\ x_0$ ),
        have h4 :  $P\ x_0 \vee Q\ x_0$ , from or.inl h3,
        show  $\exists x, P\ x \vee Q\ x$ , from exists.intro x0 h4 ))
  ( assume h2 :  $\exists x, Q\ x$ ,
    exists.elim h2
      ( assume x0 (h3 :  $Q\ x_0$ ),
        have h4 :  $P\ x_0 \vee Q\ x_0$ , from or.inr h3,
        show  $\exists x, P\ x \vee Q\ x$ , from exists.intro x0 h4 ))

-- 2ª demostración
example
  (h1 : ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ))
  :  $\exists x, P\ x \vee Q\ x :=$ 
h1.elim
  ( assume ⟨x0, (h3 :  $P\ x_0$ )⟩,
    have h4 :  $P\ x_0 \vee Q\ x_0$ , from or.inl h3,
    show  $\exists x, P\ x \vee Q\ x$ , from ⟨x0, h4⟩ )
  ( assume ⟨x0, (h3 :  $Q\ x_0$ )⟩,
    have h4 :  $P\ x_0 \vee Q\ x_0$ , from or.inr h3,
    show  $\exists x, P\ x \vee Q\ x$ , from ⟨x0, h4⟩ )

-- 3ª demostración
example
  (h1 : ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ))
  :  $\exists x, P\ x \vee Q\ x :=$ 
h1.elim
  ( assume ⟨x0, (h3 :  $P\ x_0$ )⟩,
    have h4 :  $P\ x_0 \vee Q\ x_0$ , from or.inl h3,
    ⟨x0, h4⟩ )
  ( assume ⟨x0, (h3 :  $Q\ x_0$ )⟩,
    have h4 :  $P\ x_0 \vee Q\ x_0$ , from or.inr h3,
    ⟨x0, h4⟩ )

-- 4ª demostración
example
  (h1 : ( $\exists x, P\ x$ )  $\vee$  ( $\exists x, Q\ x$ ))
  :  $\exists x, P\ x \vee Q\ x :=$ 

```

```

h1.elim
  ( assume  $\langle x_0, (h3 : P\ x_0) \rangle$ ,
     $\langle x_0, \text{or.inl } h3 \rangle$  )
  ( assume  $\langle x_0, (h3 : Q\ x_0) \rangle$ ,
     $\langle x_0, \text{or.inr } h3 \rangle$  )

-- 5ª demostración
example
  (h1 :  $(\exists x, P\ x) \vee (\exists x, Q\ x)$ )
  :  $\exists x, P\ x \vee Q\ x :=$ 
h1.elim
  ( $\lambda \langle x_0, h3 \rangle, \langle x_0, \text{or.inl } h3 \rangle$ )
  ( $\lambda \langle x_0, h3 \rangle, \langle x_0, \text{or.inr } h3 \rangle$ )

-- 6ª demostración
example
  (h1 :  $(\exists x, P\ x) \vee (\exists x, Q\ x)$ )
  :  $\exists x, P\ x \vee Q\ x :=$ 
-- by library_search
exists_or_distrib.mpr h1

-- 7ª demostración
example
  (h1 :  $(\exists x, P\ x) \vee (\exists x, Q\ x)$ )
  :  $\exists x, P\ x \vee Q\ x :=$ 
begin
  cases h1 with hp hq,
  { cases hp with x0 hx0,
    use x0,
    left,
    exact hx0, },
  { cases hq with x1 hx1,
    use x1,
    right,
    exact hx1, },
end

-- 8ª demostración
example
  (h1 :  $(\exists x, P\ x) \vee (\exists x, Q\ x)$ )
  :  $\exists x, P\ x \vee Q\ x :=$ 
begin
  rcases h1 with  $\langle x_0, hx_0 \rangle \mid \langle x_1, hx_1 \rangle$ ,
  { use x0,
    left,

```

```

    exact hx0, },
  { use x1,
    right,
    exact hx1, },
end

-- 9ª demostración
lemma aux2
  (h1 : (∃x, P x) ∨ (∃x, Q x))
  : ∃x, P x ∨ Q x :=
-- by hint
by finish

-----

-- Ej. 3. Demostrar
--   ∃x (P(x) ∨ Q(x)) ↔ ∃x P(x) ∨ ∃x Q(x)
-----

-- 1ª demostración
example :
  (∃x, P x ∨ Q x) ↔ (∃x, P x) ∨ (∃x, Q x) :=
iff.intro aux1 aux2

-- 2ª demostración
example :
  (∃x, P x ∨ Q x) ↔ (∃x, P x) ∨ (∃x, Q x) :=
⟨aux1, aux2⟩

-- 3ª demostración
example :
  (∃x, P x ∨ Q x) ↔ (∃x, P x) ∨ (∃x, Q x) :=
-- by library_search
exists_or_distrib

end

```

3.3.4. Pruebas de $\exists x \exists y P(x,y) \leftrightarrow \exists y \exists x P(x,y)$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de  $\exists x \exists y P(x,y) \leftrightarrow \exists y \exists x P(x,y)$ 
-- =====

```



```

import tactic

section

variable {U : Type}
variable {P : U -> U -> Prop}

-----

-- Ej. 1. Demostrar que
--    $\exists x \exists y P(x, y) \rightarrow \exists y \exists x P(x, y)$ 
-----

-- 1ª demostración
example :
  ( $\exists x, \exists y, P x y \rightarrow \exists y, \exists x, P x y$ ) :=
  assume h1 :  $\exists x, \exists y, P x y$ ,
  exists.elim h1
    ( assume x0 (h2 :  $\exists y, P x0 y$ ),
      exists.elim h2
        ( assume y0 (h3 :  $P x0 y0$ ),
          have h4 :  $\exists x, P x y0$ , from exists.intro x0 h3,
          show  $\exists y, \exists x, P x y$ , from exists.intro y0 h4))

-- 2ª demostración
example :
  ( $\exists x, \exists y, P x y \rightarrow \exists y, \exists x, P x y$ ) :=
  assume ⟨x0, y0, (h1 :  $P x0 y0$ )⟩,
  have h2 :  $\exists x, P x y0$ , from ⟨x0, h1⟩,
  show ( $\exists y, \exists x, P x y$ ), from ⟨y0, h2⟩

-- 3ª demostración
example :
  ( $\exists x, \exists y, P x y \rightarrow \exists y, \exists x, P x y$ ) :=
  assume ⟨x0, y0, (h1 :  $P x0 y0$ )⟩,
  show ( $\exists y, \exists x, P x y$ ), from ⟨y0, ⟨x0, h1⟩⟩

-- 4ª demostración
example :
  ( $\exists x, \exists y, P x y \rightarrow \exists y, \exists x, P x y$ ) :=
  assume ⟨x0, y0, (h1 :  $P x0 y0$ )⟩,
  show ( $\exists y, \exists x, P x y$ ), from ⟨y0, x0, h1⟩

-- 5ª demostración
example :
  ( $\exists x, \exists y, P x y \rightarrow \exists y, \exists x, P x y$ ) :=

```

```

assume ⟨x0, y0, (h1 : P x0 y0)⟩,
⟨y0, x0, h1⟩

-- 6ª demostración
example :
  (∃ x, ∃ y, P x y) → (∃ y, ∃ x, P x y) :=
λ ⟨x0, y0, h1⟩, ⟨y0, x0, h1⟩

-- 7ª demostración
example :
  (∃ x y, P x y) → (∃ y x, P x y) :=
-- by library_search
exists_comm.mp

-- 8ª demostración
example :
  (∃ x, ∃ y, P x y) → (∃ y, ∃ x, P x y) :=
begin
  intro h1,
  cases h1 with x0 h2,
  cases h2 with y0 h3,
  use y0,
  use x0,
  exact h3,
end

-- 9ª demostración
example :
  (∃ x, ∃ y, P x y) → (∃ y, ∃ x, P x y) :=
begin
  intro h1,
  cases h1 with x0 h2,
  cases h2 with y0 h3,
  use [y0, x0],
  exact h3,
end

-- 10ª demostración
example :
  (∃ x, ∃ y, P x y) → (∃ y, ∃ x, P x y) :=
begin
  intro h1,
  rcases h1 with ⟨x0, y0, h2⟩,
  use [y0, x0],
  exact h2,

```

```

end

-- 11ª demostración
example :
  ( $\exists x, \exists y, P\ x\ y$ )  $\rightarrow$  ( $\exists y, \exists x, P\ x\ y$ ) :=
begin
  intro h1,
  rcases h1 with  $\langle x_0, y_0, h2 \rangle$ ,
  exact  $\langle y_0, x_0, h2 \rangle$ ,
end

-- 12ª demostración
example :
  ( $\exists x, \exists y, P\ x\ y$ )  $\rightarrow$  ( $\exists y, \exists x, P\ x\ y$ ) :=
begin
  rintro  $\langle x_0, y_0, h2 \rangle$ ,
  exact  $\langle y_0, x_0, h2 \rangle$ ,
end

-- 13ª demostración
example :
  ( $\exists x, \exists y, P\ x\ y$ )  $\rightarrow$  ( $\exists y, \exists x, P\ x\ y$ ) :=
-- by hint
by tauto

-- 14ª demostración
lemma aux :
  ( $\exists x, \exists y, P\ x\ y$ )  $\rightarrow$  ( $\exists y, \exists x, P\ x\ y$ ) :=
by finish

-----

-- Ej. 2. Demostrar que
--    $\exists x \exists y P(x,y) \leftrightarrow \exists y \exists x P(x,y)$ 
-----

-- 1ª demostración
example :
  ( $\exists x\ y, P\ x\ y$ )  $\leftrightarrow$  ( $\exists y\ x, P\ x\ y$ ) :=
iff.intro aux aux

-- 2ª demostración
example :
  ( $\exists x\ y, P\ x\ y$ )  $\leftrightarrow$  ( $\exists y\ x, P\ x\ y$ ) :=
 $\langle \text{aux}, \text{aux} \rangle$ 

```

```

-- 3ª demostración
example :
  ( $\exists x y, P x y$ )  $\leftrightarrow$  ( $\exists y x, P x y$ ) :=
-- by library_search
exists_comm

-- 4ª demostración
example :
  ( $\exists x y, P x y$ )  $\leftrightarrow$  ( $\exists y x, P x y$ ) :=
-- by hint
by tauto

end

```

3.4. Reglas de la igualdad

3.4.1. Regla de eliminación de la igualdad

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Regla de eliminación de la igualdad
-- =====

import tactic

#check @eq.subst

-- -----
-- Ej. 1. Demostrar que
--        $x + 1 = 1 + x$ 
--        $x + 1 > 1 \rightarrow x + 1 > 0$ 
--        $\vdash 1 + x > 1 \rightarrow 1 + x > 0$ 
-- -----

variable (U : Type)
variable (x : U)
variable [has_add U]
variable [has_one U]
variable [has_lt U]
variable [has_zero U]

-- 1ª demostración

```

```
example
  (h1 : x + 1 = 1 + x)
  (h2 : x + 1 > 1 → x + 1 > 0)
  : 1 + x > 1 → 1 + x > 0 :=
eq.subst h1 h2
```

-- 2ª demostración

```
example
  (h1 : x + 1 = 1 + x)
  (h2 : x + 1 > 1 → x + 1 > 0)
  : 1 + x > 1 → 1 + x > 0 :=
h1 ► h2
```

-- 3ª demostración

```
example
  (h1 : x + 1 = 1 + x)
  (h2 : x + 1 > 1 → x + 1 > 0)
  : 1 + x > 1 → 1 + x > 0 :=
begin
  rw h1 at h2,
  exact h2,
end
```

-- 4ª demostración

```
example
  (h1 : x + 1 = 1 + x)
  (h2 : x + 1 > 1 → x + 1 > 0)
  : 1 + x > 1 → 1 + x > 0 :=
begin
  rw ←h1,
  exact h2,
end
```

-- 5ª demostración

```
example
  (h1 : x + 1 = 1 + x)
  (h2 : x + 1 > 1 → x + 1 > 0)
  : 1 + x > 1 → 1 + x > 0 :=
-- by hint
by finish
```

3.4.2. Pruebas de la transitividad de la igualdad

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Pruebas de la transitividad de la igualdad
-- =====

import tactic

variable (U : Type)
variables (x y z : U)

-----

-- Ej. 1. Demostrar que
--    $x = y, y = z \vdash x = z$ 
-----

-- 1ª demostración
example
  (h1 : x = y)
  (h2 : y = z)
  : x = z :=
eq.subst h2 h1

-- 2ª demostración
example
  (h1 : x = y)
  (h2 : y = z)
  : x = z :=
h2 ► h1

-- 3ª demostración
example
  (h1 : x = y)
  (h2 : y = z)
  : x = z :=
eq.substr h1 h2

-- 4ª demostración
example
  (h1 : x = y)
  (h2 : y = z)
  : x = z :=
eq.trans h1 h2
```

-- 5ª demostración

```
example
  (h1 : x = y)
  (h2 : y = z)
  : x = z :=
```

```
begin
  rw h1,
  exact h2,
end
```

-- 6ª demostración

```
example
  (h1 : x = y)
  (h2 : y = z)
  : x = z :=
```

```
begin
  rw h1,
  assumption,
end
```

-- 7ª demostración

```
example
  (h1 : x = y)
  (h2 : y = z)
  : x = z :=
```

```
begin
  rwa h1,
end
```

-- 8ª demostración

```
example
  (h1 : x = y)
  (h2 : y = z)
  : x = z :=
```

```
by rwa h1
```

-- 9ª demostración

```
example
  (h1 : x = y)
  (h2 : y = z)
  : x = z :=
```

```
begin
  rwa ←h2,
end
```

```
-- 10ª demostración
```

```
example
```

```
  (h1 : x = y)
```

```
  (h2 : y = z)
```

```
  : x = z :=
```

```
begin
```

```
  rwa h2 at h1,
```

```
end
```

```
-- 11ª demostración
```

```
example
```

```
  (h1 : x = y)
```

```
  (h2 : y = z)
```

```
  : x = z :=
```

```
by simp *
```

```
-- 12ª demostración
```

```
example
```

```
  (h1 : x = y)
```

```
  (h2 : y = z)
```

```
  : x = z :=
```

```
-- by hint
```

```
by finish
```

3.4.3. Regla de introducción de la igualdad

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Regla de introducción de la igualdad
```

```
-- =====
```

```
import tactic
```

```
#check @eq.refl
```

```
variable (U : Type)
```

```
variables (x y : U)
```

```
-- -----
```

```
-- Ej. Demostrar
```

```
--   x = y ⊢ y = x
```

```
-- -----
```



```
-- 1ª demostración
example
  (h1 : x = y)
  : y = x :=
have h2 : x = x, from eq.refl x,
show y = x,      from eq.subst h1 h2

-- 2ª demostración
example
  (h1 : x = y)
  : y = x :=
have h2 : x = x, from eq.refl x,
show y = x,      from h1 ► h2

-- 3ª demostración
example
  (h1 : x = y)
  : y = x :=
have h2 : x = x, from eq.refl x,
h1 ► h2

-- 4ª demostración
example
  (h1 : x = y)
  : y = x :=
h1 ► eq.refl x

-- 5ª demostración
example
  (h1 : x = y)
  : y = x :=
h1 ► rfl

-- 6ª demostración
example
  (h1 : x = y)
  : y = x :=
-- by library_search
eq.symm h1

-- 7ª demostración
example
  (h1 : x = y)
  : y = x :=
begin
```

```

    rw h1,
end

-- 8ª demostración
example
  (h1 : x = y)
  : y = x :=
by rw h1

-- 9ª demostración
example
  (h1 : x = y)
  : y = x :=
by simp *

-- 10ª demostración
example
  (h1 : x = y)
  : y = x :=
-- by hint
by tauto

-- 11ª demostración
example
  (h1 : x = y)
  : y = x :=
by finish

-- 12ª demostración
example
  (h1 : x = y)
  : y = x :=
by solve_by_elim

```

3.4.4. Pruebas de $y = x \rightarrow y = z \rightarrow x = z$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de  $y = x \rightarrow y = z \rightarrow x = z$ 
-- =====

-- -----
-- Ej. 1. Demostrar

```

```

--       $y = x \rightarrow y = z \rightarrow x = z$ 
-----

import tactic

variables (U : Type)
variables (x y z : U)

-- 1ª demostración
example :  $y = x \rightarrow y = z \rightarrow x = z$  :=
assume h1 :  $y = x$ ,
assume h2 :  $y = z$ ,
have h3 :  $x = y$ , from eq.symm h1,
show  $x = z$ ,      from eq.trans h3 h2

-- 2ª demostración
example :  $y = x \rightarrow y = z \rightarrow x = z$  :=
assume h1 :  $y = x$ ,
assume h2 :  $y = z$ ,
have h3 :  $x = y$ , from eq.symm h1,
eq.trans h3 h2

-- 3ª demostración
example :  $y = x \rightarrow y = z \rightarrow x = z$  :=
assume h1 :  $y = x$ ,
assume h2 :  $y = z$ ,
eq.trans (eq.symm h1) h2

-- 4ª demostración
example :  $y = x \rightarrow y = z \rightarrow x = z$  :=
λ h1 h2, eq.trans (eq.symm h1) h2

-- 5ª demostración
example :  $y = x \rightarrow y = z \rightarrow x = z$  :=
λ h1 h2, eq.trans h1.symm h2

-- 6ª demostración
example :  $y = x \rightarrow y = z \rightarrow x = z$  :=
-- by library_search
λ h, h.congr_left.mp

-- 7ª demostración
example :  $y = x \rightarrow y = z \rightarrow x = z$  :=
begin
  intros h1 h2,

```

```

    rwa h1,
end

-- 8ª demostración
example : y = x → y = z → x = z :=
begin
  intros h1 h2,
  rw h1 at h2,
  assumption,
end

-- 9ª demostración
example : y = x → y = z → x = z :=
begin
  intros h1 h2,
  rwa h1 at h2,
end

-- 10ª demostración
example : y = x → y = z → x = z :=
begin
  intros h1 h2,
  calc x = y : h1.symm
    ... = z : h2,
end

-- 11ª demostración
example : y = x → y = z → x = z :=
-- by hint
by finish

-- 12ª demostración
example : y = x → y = z → x = z :=
assume h1 : y = x,
assume h2 : y = z,
show x = z,
begin
  rw h1,
  rw h2
end

-- 13ª demostración
example : y = x → y = z → x = z :=
assume h1 : y = x,
assume h2 : y = z,

```

```

show x = z,
begin
  rw [←h1, h2]
end

-- 14ª demostración
example : y = x → y = z → x = z :=
assume h1 : y = x,
assume h2 : y = z,
show x = z, by rw [←h1, h2]

```

3.4.5. Pruebas de $(x + y) + z = (x + z) + y$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de  $(x + y) + z = (x + z) + y$ 
-- =====

-- -----
-- Ej. 1. Demostrar que para todo  $x, y, z$  en  $\mathbb{Z}$  se tiene
--  $(x + y) + z = (x + z) + y$ 
-- -----

import tactic
import data.int.basic

variables (x y z :  $\mathbb{Z}$ )

-- 1ª demostración
example :  $(x + y) + z = (x + z) + y :=$ 
calc
   $(x + y) + z = x + (y + z)$  : add_assoc x y z
  ... =  $x + (z + y)$  : eq.subst (add_comm y z) rfl
  ... =  $(x + z) + y$  : eq.symm (add_assoc x z y)

-- 2ª demostración
example :  $(x + y) + z = (x + z) + y :=$ 
calc
   $(x + y) + z = x + (y + z)$  : by rw add_assoc
  ... =  $x + (z + y)$  : by rw [add_comm y z]
  ... =  $(x + z) + y$  : by rw add_assoc

```

```

-- 3ª demostración
example : (x + y) + z = (x + z) + y :=
begin
  rw add_assoc,
  rw add_comm y z,
  rw add_assoc,
end

example : (x + y) + z = (x + z) + y :=
begin
  rw [add_assoc, add_comm y z, add_assoc],
end

-- 4ª demostración
example : (x + y) + z = (x + z) + y :=
by rw [add_assoc, add_comm y z, add_assoc]

-- 5ª demostración
example : (x + y) + z = (x + z) + y :=
-- by library_search
add_right_comm x y z

-- 6ª demostración
example : (x + y) + z = (x + z) + y :=
-- by hint
by omega

-- 7ª demostración
example : (x + y) + z = (x + z) + y :=
by linarith

-- 8ª demostración
example : (x + y) + z = (x + z) + y :=
by nlinarith

-- 9ª demostración
example : (x + y) + z = (x + z) + y :=
by ring

-- 10ª demostración
example : (x + y) + z = (x + z) + y :=
by finish

```

3.4.6. Pruebas de desarrollo de producto de sumas

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Pruebas de desarrollo de producto de sumas
-- =====

-- -----
-- Ej. 1. Sean a, b, c y d números enteros. Demostrar que
--      (a + b) * (c + d) = a * c + b * c + a * d + b * d
-- -----

import tactic
import data.int.basic

variables a b c d : ℤ

-- 1ª demostración
example : (a + b) * (c + d) = a * c + b * c + a * d + b * d :=
calc
  (a + b) * (c + d)
    = (a + b) * c + (a + b) * d      : by rw left_distrib
  ... = (a * c + b * c) + (a + b) * d : by rw right_distrib
  ... = (a * c + b * c) + (a * d + b * d) : by rw right_distrib
  ... = a * c + b * c + a * d + b * d : by rw ←add_assoc

-- 2ª demostración
example : (a + b) * (c + d) = a * c + b * c + a * d + b * d :=
by rw [left_distrib, right_distrib, right_distrib, ←add_assoc]

-- 3ª demostración
example : (a + b) * (c + d) = a * c + b * c + a * d + b * d :=
begin
  rw left_distrib,
  rw right_distrib,
  rw right_distrib,
  rw ←add_assoc,
end

-- 4ª demostración
example : (a + b) * (c + d) = a * c + b * c + a * d + b * d :=
calc
  (a + b) * (c + d)
    = (a + b) * c + (a + b) * d      : by rw mul_add
  ... = (a * c + b * c) + (a + b) * d : by rw add_mul
```

```

... = (a * c + b * c) + (a * d + b * d) : by rw add_mul
... = a * c + b * c + a * d + b * d      : by rw ←add_assoc

-- 5ª demostración
example : (a + b) * (c + d) = a * c + b * c + a * d + b * d :=
-- by hint
by linarith

-- 6ª demostración
example : (a + b) * (c + d) = a * c + b * c + a * d + b * d :=
by nlinarith

-- 7ª demostración
example : (a + b) * (c + d) = a * c + b * c + a * d + b * d :=
by ring

```


Capítulo 4

Conjuntos

4.1. Elementos básicos sobre conjuntos

4.1.1. Pruebas de la reflexividad de la inclusión de conjuntos

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Prueba de la reflexividad de la inclusión de conjuntos
-- =====

-- -----
-- Ej. 1. Demostrar
--    $A \subseteq A$ 
-- -----

import data.set
variable U : Type
variable x : U
variables A B C : set U

-- #reduce x ∈ A
-- #reduce B ⊆ C

-- 1ª demostración
example : A ⊆ A :=
begin
  intros x h,
  exact h,
end
```

```

-- 2ª demostración
example : A ⊆ A :=
assume x,
assume h : x ∈ A,
show x ∈ A, from h

-- 3ª demostración
example : A ⊆ A :=
assume x,
assume h : x ∈ A,
h

-- 4ª demostración
example : A ⊆ A :=
assume x,
λ h : x ∈ A, h

-- 5ª demostración
example : A ⊆ A :=
assume x,
id

-- 6ª demostración
example : A ⊆ A :=
λ x, id

-- 7ª demostración
example : A ⊆ A :=
-- by library_search
set.subset.rfl

open set

-- 8ª demostración
example : A ⊆ A :=
subset.rfl

-- 9ª demostración
example : A ⊆ A :=
-- by hint
by tauto

-- 10ª demostración
example : A ⊆ A :=

```

```
by finish

-- 11ª demostración
example : A ⊆ A :=
by refl
```

4.1.2. Pruebas de la antisimetría de la inclusión de conjuntos

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Pruebas de la antisimetría de la inclusión de conjuntos
-- =====

-- -----
-- Ej. 1. Demostrar
--    $A \subseteq B, B \subseteq A \vdash A = B$ 
-- -----

import data.set

variable U : Type
variables A B : set U

open set

-- 1ª demostración
example
  (h1 : A ⊆ B)
  (h2 : B ⊆ A)
  : A = B :=
begin
  ext,
  split,
  { intro h,
    exact h1 h, },
  { intro h,
    exact h2 h, },
end

-- 2ª demostración
example
```

```

(h1 : A ⊆ B)
(h2 : B ⊆ A)
: A = B :=
ext
( assume x,
  iff.intro
    ( assume h : x ∈ A,
      show x ∈ B, from h1 h)
    ( assume h : x ∈ B,
      show x ∈ A, from h2 h))

-- 3ª demostración
example
(h1 : A ⊆ B)
(h2 : B ⊆ A)
: A = B :=
ext
(λ x,
  iff.intro
    (λ h, h1 h)
    (λ h, h2 h))

-- 4ª demostración
example
(h1 : A ⊆ B)
(h2 : B ⊆ A)
: A = B :=
eq_of_subset_of_subset
( assume x,
  assume h : x ∈ A,
  show x ∈ B, from h1 h)
( assume x,
  assume h : x ∈ B,
  show x ∈ A, from h2 h)

-- 5ª demostración
example
(h1 : A ⊆ B)
(h2 : B ⊆ A)
: A = B :=
eq_of_subset_of_subset h1 h2

-- 6ª demostración

```

```
example
  (h1 : A ⊆ B)
  (h2 : B ⊆ A)
  : A = B :=
-- by library_search
subset.antisymm h1 h2
```

4.1.3. Introducción de la intersección

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Introducción de la intersección
-- =====

-- -----
-- Ej. 1. Demostrar
--    $x \in A \rightarrow x \in B \rightarrow x \in A \cap B$ 
-- -----

import data.set

variable U : Type
variables A B : set U
variable x : U

open set

-- #reduce  $x \in A \cap B$ 

-- 1ª demostración
example :  $x \in A \rightarrow x \in B \rightarrow x \in A \cap B :=$ 
begin
  intros h1 h2,
  simp,
  split,
  { exact h1, },
  { exact h2, },
end

-- 2ª demostración
example :  $x \in A \rightarrow x \in B \rightarrow x \in A \cap B :=$ 
begin
```

```

intros h1 h2,
split,
{ exact h1, },
{ exact h2, },
end

-- 3ª demostración
example :  $x \in A \rightarrow x \in B \rightarrow x \in A \cap B :=$ 
assume h1 :  $x \in A$ ,
assume h2 :  $x \in B$ ,
show  $x \in A \cap B$ , from and.intro h1 h2

-- 4ª demostración
example :  $x \in A \rightarrow x \in B \rightarrow x \in A \cap B :=$ 
assume h1 :  $x \in A$ ,
assume h2 :  $x \in B$ ,
show  $x \in A \cap B$ , from ⟨h1, h2⟩

-- 5ª demostración
example :  $x \in A \rightarrow x \in B \rightarrow x \in A \cap B :=$ 
assume h1 :  $x \in A$ ,
assume h2 :  $x \in B$ ,
⟨h1, h2⟩

-- 6ª demostración
example :  $x \in A \rightarrow x \in B \rightarrow x \in A \cap B :=$ 
λ h1 h2, ⟨h1, h2⟩

-- 7ª demostración
example :  $x \in A \rightarrow x \in B \rightarrow x \in A \cap B :=$ 
-- by library_search
mem_inter

```

4.1.4. Introducción de la unión

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Regla de introducción de la unión
-- =====

-- -----
-- Ej. 1. Demostrar

```

```

--       $A \subseteq A \cup B$ 
-----

import data.set

variable U : Type
variables A B : set U
variable x : U

open set

-- #reduce  $x \in A \cup B$ 

-- 1ª demostración
example :  $A \subseteq A \cup B$  :=
begin
  intros x h,
  simp,
  left,
  exact h,
end

-- 2ª demostración
example :  $A \subseteq A \cup B$  :=
begin
  intros x h,
  left,
  exact h,
end

-- 3ª demostración
example :  $A \subseteq A \cup B$  :=
assume x,
assume h :  $x \in A$ ,
show  $x \in A \cup B$ , from or.inl h

-- 4ª demostración
example :  $A \subseteq A \cup B$  :=
assume x,
assume h :  $x \in A$ ,
or.inl h

-- 5ª demostración
example :  $A \subseteq A \cup B$  :=
assume x,

```

```

λ h : x ∈ A, or.inl h

-- 6ª demostración
example : A ⊆ A ∪ B :=
assume x, or.inl

-- 7ª demostración
example : A ⊆ A ∪ B :=
λ x, or.inl

-- 8ª demostración
example : A ⊆ A ∪ B :=
-- by library_search
subset_union_left A B

-- 9ª demostración
example : A ⊆ A ∪ B :=
λ x, mem_union_left B

-- 10ª demostración
example : A ⊆ A ∪ B :=
-- by hint
by finish

-- 11ª demostración
example : A ⊆ A ∪ B :=
by simp

```

4.1.5. El conjunto vacío

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Regla del conjunto vacío
-- =====

-- -----
-- Ej. 1. Demostrar
--    $\emptyset \subseteq A$ 
-- -----

import data.set

```



```

variable U : Type
variables A : set U
variable x : U

open set

-- #reduce ( $\emptyset$  : set U)
-- #reduce  $x \in (\emptyset$  : set U)

-- 1ª demostración
example :  $\emptyset \subseteq A$  :=
begin
  intros x h,
  simp at h,
  exfalso,
  exact h,
end

-- 2ª demostración
example :  $\emptyset \subseteq A$  :=
begin
  intros x h,
  exfalso,
  exact h,
end

-- 3ª demostración
example :  $\emptyset \subseteq A$  :=
assume x,
assume h :  $x \in (\emptyset$  : set U),
show  $x \in A$ , from false.elim h

-- 4ª demostración
example :  $\emptyset \subseteq A$  :=
λ x, λ h, false.elim h

-- 5ª demostración
example :  $\emptyset \subseteq A$  :=
λ _, false.elim

-- 6ª demostración
example :  $\emptyset \subseteq A$  :=
-- by library_search
empty_subset A

```

```

-- 7ª demostración
example :  $\emptyset \subseteq A :=$ 
assume x,
assume h : x  $\in$  ( $\emptyset$  : set U),
show x  $\in$  A, from absurd h (not_mem_empty x)

-- 8ª demostración
example :  $\emptyset \subseteq A :=$ 
λ x h, absurd h (not_mem_empty x)

-- 9ª demostración
example :  $\emptyset \subseteq A :=$ 
-- by hint
by tauto

-- 10ª demostración
example :  $\emptyset \subseteq A :=$ 
by finish

-- 11ª demostración
example :  $\emptyset \subseteq A :=$ 
by simp

```

4.1.6. Diferencia de conjuntos: $A \setminus B \subseteq A$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Diferencia de conjuntos:  $A \setminus B \subseteq A$ 
-- =====

-- -----
-- Ej. 1. Demostrar
--    $A \setminus B \subseteq A$ 
-- -----

import data.set

variable U : Type
variables A B : set U
variable x : U

open set

```

```

-- #reduce (A \ B)
-- #reduce x ∈ A \ B

-- 1ª demostración
example : A \ B ⊆ A :=
begin
  intros x h,
  simp at h,
  exact h.left,
end

-- 2ª demostración
example : A \ B ⊆ A :=
begin
  intros x h,
  exact h.left,
end

-- 3ª demostración
example : A \ B ⊆ A :=
assume x,
assume h : x ∈ A \ B,
show x ∈ A, from h.left

-- 4ª demostración
example : A \ B ⊆ A :=
assume x,
assume h : x ∈ A \ B,
and.left h

-- 5ª demostración
example : A \ B ⊆ A :=
assume x,
λ h, and.left h

-- 6ª demostración
example : A \ B ⊆ A :=
assume x, and.left

-- 7ª demostración
example : A \ B ⊆ A :=
λ _, and.left

-- 8ª demostración

```

```

example : A \ B ⊆ A :=
-- by library_search
diff_subset A B

-- 9ª demostración
example : A \ B ⊆ A :=
assume x,
assume h : x ∈ A \ B,
show x ∈ A, from mem_of_mem_diff h

-- 10ª demostración
example : A \ B ⊆ A :=
λ _, mem_of_mem_diff

-- 11ª demostración
example : A \ B ⊆ A :=
by finish [subset_def]

```

4.1.7. Complementario de un conjunto: Pruebas de $A \setminus B \subseteq B^c$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Complementario de un conjunto: Pruebas de  $A \setminus B \subseteq B^c$ 
-- =====

-- -----
-- Ej. 1. Demostrar
--    $A \setminus B \subseteq B^c$ 
-- -----

import data.set

variable {U : Type}
variables A B : set U
variable x : U

open set

-- #reduce  $x \in B^c$ 
-- #reduce  $B^c$ 

```

```

-- 1ª demostración
example :  $A \setminus B \subseteq B^c$  :=
begin
  intros x h,
  simp at *,
  exact h.right,
end

-- 2ª demostración
example :  $A \setminus B \subseteq B^c$  :=
begin
  intros x h,
  exact h.right,
end

-- 3ª demostración
example :  $A \setminus B \subseteq B^c$  :=
assume x,
assume h1 :  $x \in A \setminus B$ ,
have h2 :  $x \notin B$ , from and.right h1,
show  $x \in B^c$ , from h2

-- 4ª demostración
example :  $A \setminus B \subseteq B^c$  :=
assume x,
assume h1 :  $x \in A \setminus B$ ,
show  $x \in B^c$ , from and.right h1

-- 5ª demostración
example :  $A \setminus B \subseteq B^c$  :=
assume x,
λ h1, and.right h1

-- 6ª demostración
example :  $A \setminus B \subseteq B^c$  :=
assume x,
and.right

-- 7ª demostración
example :  $A \setminus B \subseteq B^c$  :=
λ _, and.right

-- 8ª demostración
example :  $A \setminus B \subseteq B^c$  :=

```

```
λ _, not_mem_of_mem_diff
```

4.1.8. Pruebas de la conmutatividad de la intersección

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Pruebas de la conmutatividad de la intersección
-- =====
```

```
-- Ej. 1. Demostrar
--    $A \cap B \subseteq B \cap A$ 
```

```
import data.set
```

```
variable {U : Type}
variables A B : set U
variable x : U
```

```
open set
```

```
-- 1ª demostración
```

```
example : A ∩ B ⊆ B ∩ A :=
begin
  intros x h,
  simp at *,
  split,
  { exact h.right, },
  { exact h.left, },
end
```

```
-- 2ª demostración
```

```
example : A ∩ B ⊆ B ∩ A :=
begin
  intros x h,
  split,
  { exact h.right, },
  { exact h.left, },
end
```

```
-- 3ª demostración
```

```
example : A ∩ B ⊆ B ∩ A :=
```

```

begin
  rintros x ⟨h1, h2⟩,
  split,
  { exact h2, },
  { exact h1, },
end

-- 4ª demostración
example : A ∩ B ⊆ B ∩ A :=
begin
  rintros x ⟨h1, h2⟩,
  exact ⟨h2, h1⟩,
end

-- 5ª demostración
example : A ∩ B ⊆ B ∩ A :=
assume x,
assume h : x ∈ A ∩ B,
have h1 : x ∈ A, from and.left h,
have h2 : x ∈ B, from and.right h,
show x ∈ B ∩ A, from and.intro h2 h1

-- 6ª demostración
example : A ∩ B ⊆ B ∩ A :=
assume x,
assume h : x ∈ A ∩ B,
have h1 : x ∈ A ∧ x ∈ B, from h,
have h2 : x ∈ B ∧ x ∈ A, from and.comm.mp h1,
show x ∈ B ∩ A, from h2

-- 7ª demostración
example : A ∩ B ⊆ B ∩ A :=
assume x,
assume h : x ∈ A ∩ B,
show x ∈ B ∩ A, from and.comm.mp h

-- 8ª demostración
example : A ∩ B ⊆ B ∩ A :=
assume x,
assume h : x ∈ A ∩ B,
and.comm.mp h

-- 9ª demostración
example : A ∩ B ⊆ B ∩ A :=

```

```

assume x,
λ h, and.comm.mp h

-- 10ª demostración
example :  $A \cap B \subseteq B \cap A$  :=
assume x,
and.comm.mp

-- 10ª demostración
example :  $A \cap B \subseteq B \cap A$  :=
λ _, and.comm.mp

-- 11ª demostración
example :  $A \cap B \subseteq B \cap A$  :=
-- by hint
by finish

-- 12ª demostración
lemma aux :  $A \cap B \subseteq B \cap A$  :=
by simp

-----

-- Ej. 2. Demostrar
--  $A \cap B = B \cap A$ 
-----

-- 1ª demostración
example :  $A \cap B = B \cap A$  :=
begin
  apply eq_of_subset_of_subset,
  { exact aux A B, },
  { exact aux B A, },
end

-- 2ª demostración
example :  $A \cap B = B \cap A$  :=
eq_of_subset_of_subset (aux A B) (aux B A)

-- 3ª demostración
example :  $A \cap B = B \cap A$  :=
-- by library_search
inter_comm A B

-- 4ª demostración
example :  $A \cap B = B \cap A$  :=

```



```
-- by hint
by finish
```

4.2. Identidades conjuntistas

4.2.1. Pruebas de la propiedad distributiva de la intersección sobre la unión.

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Pruebas de  $A \cap (B \cup C) = (A \cap B) \cup (A \cap C)$ 
-- =====
```

```
import data.set
open set
```

```
variable {U : Type}
variables A B C : set U
```

```
-- -----
-- Ej. 1. Demostrar
--  $A \cap (B \cup C) \subseteq (A \cap B) \cup (A \cap C)$ 
-- -----
```

```
-- 1ª demostración
```

```
example :
```

```
  A  $\cap$  (B  $\cup$  C)  $\subseteq$  (A  $\cap$  B)  $\cup$  (A  $\cap$  C) :=
begin
```

```
  intros x h,
  cases h with ha hbc,
  cases hbc with hb hc,
  { left,
    split,
    { exact ha, },
    { exact hb, }},
  { right,
    split,
    { exact ha, },
    { exact hc, }},
end
```

```
end
```

```

-- 2ª demostración
example :
  A ⊆ (B ∪ C) ⊆ (A ∩ B) ∪ (A ∩ C) :=
begin
  intros x h,
  cases h with ha hbc,
  cases hbc with hb hc,
  { left,
    split,
    { assumption, },
    { assumption, }},
  { right,
    split,
    { assumption, },
    { assumption, }},
end

-- 3ª demostración
example :
  A ⊆ (B ∪ C) ⊆ (A ∩ B) ∪ (A ∩ C) :=
begin
  intros x h,
  cases h with ha hbc,
  cases hbc with hb hc,
  { left,
    split,
    assumption', },
  { right,
    split,
    assumption', },
end

-- 4ª demostración
example :
  A ⊆ (B ∪ C) ⊆ (A ∩ B) ∪ (A ∩ C) :=
begin
  rintros x ⟨ha, (hb | hc)⟩,
  { left,
    split,
    assumption', },
  { right,
    split,
    assumption', },
end

```

```

-- 5ª demostración
example :
   $A \cap (B \cup C) \subseteq (A \cap B) \cup (A \cap C) :=$ 
  assume x,
  assume h :  $x \in A \cap (B \cup C)$ ,
  have x ∈ A, from and.left h,
  have x ∈ B ∪ C, from and.right h,
  or.elim (⟨x ∈ B ∪ C⟩)
  ( assume : x ∈ B,
    have x ∈ A ∩ B, from and.intro ⟨x ∈ A⟩ ⟨x ∈ B⟩,
    show x ∈ (A ∩ B) ∪ (A ∩ C), from or.inl this)
  ( assume : x ∈ C,
    have x ∈ A ∩ C, from and.intro ⟨x ∈ A⟩ ⟨x ∈ C⟩,
    show x ∈ (A ∩ B) ∪ (A ∩ C), from or.inr this)

-- 6ª demostración
lemma inter_union_l1 :
   $A \cap (B \cup C) \subseteq (A \cap B) \cup (A \cap C) :=$ 
  assume x,
  assume h :  $x \in A \cap (B \cup C)$ ,
  have ha : x ∈ A, from and.left h,
  have hbc : x ∈ B ∪ C, from and.right h,
  or.elim hbc
  ( assume hb : x ∈ B,
    have hab : x ∈ A ∩ B, from and.intro ha hb,
    show x ∈ (A ∩ B) ∪ (A ∩ C), from or.inl hab)
  ( assume hc : x ∈ C,
    have hac : x ∈ A ∩ C, from and.intro ha hc,
    show x ∈ (A ∩ B) ∪ (A ∩ C), from or.inr hac)

-----
-- Ej. 2. Demostrar
--    $(A \cap B) \cup (A \cap C) \subseteq A \cap (B \cup C)$ 
-----

-- 1ª demostración
example :
   $(A \cap B) \cup (A \cap C) \subseteq A \cap (B \cup C) :=$ 
  begin
    intros x h,
    cases h with hab hac,
    { split,
      { exact hab.left, },

```

```

    { left,
      exact hab.right, }},
  { split,
    { exact hac.left, },
    { right,
      exact hac.right, }},
end

-- 2ª demostración
example :
  (A ∩ B) ∪ (A ∩ C) ⊆ A ∩ (B ∪ C) :=
begin
  rintros x (<ha, hb> | <ha, hc>),
  { split,
    { exact ha, },
    { left,
      exact hb, }},
  { split,
    { exact ha, },
    { right,
      exact hc, }},
end

-- 3ª demostración
lemma inter_union_l2 :
  (A ∩ B) ∪ (A ∩ C) ⊆ A ∩ (B ∪ C) :=
assume x,
assume : x ∈ (A ∩ B) ∪ (A ∩ C),
or.elim this
( assume h : x ∈ A ∩ B,
  have x ∈ A, from and.left h,
  have x ∈ B, from and.right h,
  have x ∈ B ∪ C, from or.inl this,
  show x ∈ A ∩ (B ∪ C), from and.intro <x ∈ A> this)
( assume h : x ∈ A ∩ C,
  have x ∈ A, from and.left h,
  have x ∈ C, from and.right h,
  have x ∈ B ∪ C, from or.inr this,
  show x ∈ A ∩ (B ∪ C), from and.intro <x ∈ A> this)

-----
-- Ej. 3. Demostrar
--   (A ∩ B) ∪ (A ∩ C) = A ∩ (B ∪ C)
-----

```

```

-- 1ª demostración
example :
  A ∩ (B ∪ C) = (A ∩ B) ∪ (A ∩ C) :=
-- by library_search
inter_distrib_left A B C

-- 2ª demostración
theorem inter_union :
  A ∩ (B ∪ C) = (A ∩ B) ∪ (A ∩ C) :=
eq_of_subset_of_subset
  (inter_union_l1 A B C)
  (inter_union_l2 A B C)

-- 3ª demostración
example :
  A ∩ (B ∪ C) = (A ∩ B) ∪ (A ∩ C) :=
begin
  ext,
  simp,
  exact and_or_distrib_left,
end

-- 4ª demostración
example :
  A ∩ (B ∪ C) = (A ∩ B) ∪ (A ∩ C) :=
begin
  ext,
  exact and_or_distrib_left,
end

-- 5ª demostración
example :
  A ∩ (B ∪ C) = (A ∩ B) ∪ (A ∩ C) :=
ext (λ x, and_or_distrib_left)

```

4.2.2. Pruebas de $(A \cap B^c) \cup B = A \cup B$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de  $(A \cap B^c) \cup B = A \cup B$ 
-- =====

```

```

-----
-- Ej. 1. Demostrar
--    $(A \cap B^c) \cup B = A \cup B$ 
-----

import data.set

open set

variable U : Type
variables A B C : set U

-- 1ª demostración
-- =====

example : (A  $\cap$  Bc)  $\cup$  B = A  $\cup$  B :=
calc
  (A  $\cap$  Bc)  $\cup$  B = (A  $\cup$  B)  $\cap$  (Bc  $\cup$  B) : by rw union_distrib_right
  ... = (A  $\cup$  B)  $\cap$  univ : by rw compl_union_self
  ... = A  $\cup$  B : by rw inter_univ

example : (A  $\cap$  B)  $\cup$  C = (A  $\cup$  C)  $\cap$  (B  $\cup$  C) :=
-- by library_search
union_distrib_right A B C

example : Bc  $\cup$  B = univ :=
-- by library_search
compl_union_self B

example : A  $\cap$  univ = A :=
-- by library_search
inter_univ A

-- 2ª demostración
-- =====

example : (A  $\cap$  Bc)  $\cup$  B = A  $\cup$  B :=
begin
  rw union_distrib_right,
  rw compl_union_self,
  rw inter_univ,
end

-- 3ª demostración
-- =====

```

```

example : (A ∩ Bc) ∪ B = A ∪ B :=
by rw [union_distrib_right, compl_union_self, inter_univ]

-- 4ª demostración
-- =====

example : (A ∩ Bc) ∪ B = A ∪ B :=
by simp [union_distrib_right]

```

4.3. Familias de conjuntos

4.3.1. Unión e intersección de familias de conjuntos

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Unión e intersección de familias de conjuntos
-- =====

-----
-- Ej. 1. Declarar I y U como variables de tipo.
-----

variables {I U : Type}

-----
-- Ej. 2. Definir la función
--   Union : (I → set U) → set U
-- tal que (Union A) es la unión de de los conjuntos de
-- la familia A.
-----

def Union (A : I → set U) : set U :=
  { x | ∃ i : I, x ∈ A i }

-----
-- Ej. 3. Definir la función
--   Inter : (I → set U) → set U
-- tal que (Inter A) es la intersección de de los
-- conjuntos de la familia A.
-----

```

```

def Inter (A : I → set U) : set U :=
  { x | ∀ i : I, x ∈ A i }

-----

-- Ej. 4. Declarar
-- + x como una variable sobre U y
-- + A como una variable sobre familias de conjuntos de
--   U con índice en A.
-----

variable x : U
variable (A : I → set U)

-----

-- Ej. 5. Demostrar que
--   x ∈ Union A ⊢ ∃ i, x ∈ A i
-----

example
  (h : x ∈ Union A)
  : ∃ i, x ∈ A i :=
h

-----

-- Ej. 6. Demostrar que
--   x ∈ x ∈ Inter A ⊢ ∀ i, x ∈ A i
-----

example
  (h : x ∈ Inter A)
  : ∀ i, x ∈ A i :=
h

-----

-- Ej 7. Usar (⋃ i, A i) como notación para (Union A).
-----

notation `⋃` binds ``, ` r:(scoped f, Union f) := r

-----

-- Ej 8. Usar (⋂ i, A i) como notación para (Inter A).
-----

notation `⋂` binds ``, ` r:(scoped f, Inter f) := r

```



```

-----
-- Ej. 9. Demostrar que
--    $x \in \bigcup i, A i \vdash \exists i, x \in A i$ 
-----

example
  (h : x ∈ ⋃ i, A i)
  : ∃ i, x ∈ A i :=
h

-----
-- Ej. 10. Demostrar que
--    $x \in x \in \bigcap i, A i \vdash \forall i, x \in A i$ 
-----

example
  (h : x ∈ ⋂ i, A i)
  : ∀ i, x ∈ A i :=
h

```

4.3.2. Pertenencia a uniones e intersecciones de familias

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pertenencia a uniones e intersecciones de familias
-- =====

import data.set

open set

variables {I U : Type}
variables {A : I → set U}
variable {x : U}

-----
-- Ej. 1. Demostrar que
--    $(x \in \bigcup i, A i) \leftrightarrow (\exists i, x \in A i)$ 
-----

```

```

-- 1ª demostración
example :
  (x ∈ ⋃ i, A i) ↔ (∃ i, x ∈ A i) :=
-- by library_search
mem_Union

-- 2ª demostración
example :
  (x ∈ ⋃ i, A i) ↔ (∃ i, x ∈ A i) :=
by simp

-----

-- Ej. 2. Demostrar que
--   (x ∈ ⋂ i, A i) ↔ (∀ i, x ∈ A i)
-----

-- 1ª demostración
example :
  (x ∈ ⋂ i, A i) ↔ (∀ i, x ∈ A i) :=
-- by library_search
mem_Inter

-- 2ª demostración
example :
  (x ∈ ⋂ i, A i) ↔ (∀ i, x ∈ A i) :=
by simp

```

4.3.3. Pruebas de la distributiva de la intersección general sobre la intersección

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de la distributiva de la intersección general sobre la intersección
-- =====

-----

-- Ej. 1. Demostrar
--   (⋂ i, A i ∩ B i) = (⋂ i, A i) ∩ (⋂ i, B i)
-----

import data.set
import tactic

```

```

open set

variables {I U : Type}
variables {A B : I → set U}

-- 1ª demostración
example :
  ( $\bigcap i, A i \cap B i$ ) = ( $\bigcap i, A i$ )  $\cap$  ( $\bigcap i, B i$ ) :=
begin
  ext,
  split,
  { intro h,
    rw mem_Inter at h,
    split,
    { rw mem_Inter,
      intro i,
      exact (h i).left, },
    { rw mem_Inter,
      intro i,
      exact (h i).right, }},
  { rintro ⟨h1, h2⟩,
    rw mem_Inter at *,
    intro i,
    exact ⟨h1 i, h2 i⟩, },
end

-- 2ª demostración
example :
  ( $\bigcap i, A i \cap B i$ ) = ( $\bigcap i, A i$ )  $\cap$  ( $\bigcap i, B i$ ) :=
ext $
assume x : U,
iff.intro
( assume h : x  $\in$   $\bigcap i, A i \cap B i$ ,
  have h1 :  $\forall i, x \in A i \cap B i$ ,
    from mem_Inter.mp h,
  have h2 :  $\forall i, x \in A i$ ,
    from assume i, and.left (h1 i),
  have h3 :  $\forall i, x \in B i$ ,
    from assume i, and.right (h1 i),
  have h4 : x  $\in$   $\bigcap i, A i$ ,
    from mem_Inter.mpr h2,
  have h5 : x  $\in$   $\bigcap i, B i$ ,
    from mem_Inter.mpr h3,
  show x  $\in$  ( $\bigcap i, A i$ )  $\cap$  ( $\bigcap i, B i$ ),

```

```

    from and.intro h4 h5)
( assume h : x ∈ (⋂ i, A i) ∩ (⋂ i, B i),
  have h1 : ∀ i, x ∈ A i,
    from mem_Inter.mp (and.left h),
  have h2 : ∀ i, x ∈ B i,
    from mem_Inter.mp (and.right h),
  have h3 : ∀ i, x ∈ A i ∩ B i,
    from assume i, and.intro (h1 i) (h2 i),
  show x ∈ ⋂ i, A i ∩ B i,
    from mem_Inter.mpr h3)

-- 3ª demostración
example :
  (⋂ i, A i ∩ B i) = (⋂ i, A i) ∩ (⋂ i, B i) :=
-- by library_search
Inter_inter_distrib A B

-- 4ª demostración
example :
  (⋂ i, A i ∩ B i) = (⋂ i, A i) ∩ (⋂ i, B i) :=
ext (by finish)

```

4.3.4. Reglas de la intersección general

- Enlaces al [código](#) y a la [sesión en Lean Web](#).

```

-- Reglas de la intersección general
-- =====

import data.set
open set

section
variables {I U : Type}
variables {A : I → set U}
variable {x : U}

-- Regla de introducción de la intersección
-- =====

-- 1ª demostración
example

```

```

(h :  $\forall i, x \in A i$ )
:  $x \in \bigcap i, A i :=$ 
begin
  simp,
  assumption,
end

-- 2ª demostración
theorem Inter.intro
  (h :  $\forall i, x \in A i$ )
  :  $x \in \bigcap i, A i :=$ 
by simp; assumption

-- Regla de eliminación de la intersección
-- =====

-- 1ª demostración
example
  (h :  $x \in \bigcap i, A i$ )
  (i : I)
  :  $x \in A i :=$ 
begin
  simp at h,
  apply h,
end

-- 2ª demostración
@[elab_simple]
theorem Inter.elim
  (h :  $x \in \bigcap i, A i$ )
  (i : I)
  :  $x \in A i :=$ 
by simp at h; apply h

end

```

4.3.5. Reglas de la unión general

- Enlaces al [código](#) y a la [sesión en Lean Web](#).

```

-- Reglas de la unión general
-- =====

```

```

import data.set

open set

variables {I U : Type}
variables {A : I → set U}
variable {x : U}
variable (i : I)

-- Regla de introducción de la unión
-- =====

-- 1ª demostración
example
  (h : x ∈ A i)
  : x ∈ ⋃ i, A i :=
begin
  simp,
  existsi i,
  exact h
end

-- 2ª demostración
theorem Union.intro
  (h : x ∈ A i)
  : x ∈ ⋃ i, A i :=
by {simp, existsi i, exact h}

-- Regla de eliminación de la unión
-- =====

-- 1ª demostración
example
  {b : Prop}
  (h₁ : x ∈ ⋃ i, A i)
  (h₂ : ∀ (i : I), x ∈ A i → b)
  : b :=
begin
  simp at h₁,
  cases h₁ with i h,
  exact h₂ i h,
end

-- 2ª demostración

```

```

theorem Union.elim
  {b : Prop}
  (h1 : x ∈ ⋃ i, A i)
  (h2 : ∀ (i : I), x ∈ A i → b)
  : b :=
by {simp at h1, cases h1 with i h, exact h2 i h}

```

4.3.6. Pruebas de intersección sobre unión general

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de intersección sobre unión general
-- =====

import data.set

open set

variables {I U : Type}
variables {A : I → set U}
variable {C : set U}

-----
-- Ej. 1. Demostrar
--   C ∩ (⋃ i, A i) ⊆ (⋃ i, C ∩ A i)
-----

-- 1ª demostración
example :
  C ∩ (⋃ i, A i) ⊆ (⋃ i, C ∩ A i) :=
begin
  rintro x ⟨hC, hU⟩,
  rw mem_Union at hU,
  cases hU with i hA,
  apply mem_Union.mpr,
  use i,
  split,
  assumption',
end

-- 2ª demostración
example :

```

```

C ∩ (⋃ i, A i) ⊆ (⋃ i, C ∩ A i) :=
begin
  intros x h,
  simp * at *,
end

-- 3ª demostración
lemma inter_Uni_l1 :
  C ∩ (⋃ i, A i) ⊆ (⋃ i, C ∩ A i) :=
by {intros x h, simp * at *}

-----

-- Ej. 2. Demostrar
--   (⋃ i, C ∩ A i) ⊆ C ∩ (⋃ i, A i)
-----

-- 1ª demostración
example :
  (⋃ i, C ∩ A i) ⊆ C ∩ (⋃ i, A i) :=
begin
  intros x h,
  rw mem_Union at h,
  cases h with i hi,
  cases hi with hC hA,
  split,
  { exact hC, },
  { apply mem_Union.mpr,
    use i,
    exact hA, },
end

-- 2ª demostración
example : (⋃ i, C ∩ A i) ⊆ C ∩ (⋃ i, A i) :=
begin
  intros x h,
  rw mem_Union at h,
  rcases h with ⟨i, hC, hA⟩,
  split,
  { exact hC, },
  { apply mem_Union.mpr,
    use i,
    exact hA, },
end

-- 3ª demostración

```



```

example :
   $(\bigcup i, C \cap A i) \subseteq C \cap (\bigcup i, A i) :=$ 
begin
  intros x h,
  simp * at *,
end

-- 4ª demostración
lemma inter_Uni_l2 :
   $(\bigcup i, C \cap A i) \subseteq C \cap (\bigcup i, A i) :=$ 
by {intros x h, simp * at *}

-----

-- Ej. 3. Demostrar
--  $C \cap (\bigcup i, A i) = (\bigcup i, C \cap A i)$ 
-----

-- 1ª demostración
example :
   $C \cap (\bigcup i, A i) = (\bigcup i, C \cap A i) :=$ 
eq_of_subset_of_subset inter_Uni_l1 inter_Uni_l2

-- 2ª demostración
example :
   $C \cap (\bigcup i, A i) = (\bigcup i, C \cap A i) :=$ 
-- by library_search
inter_Union C A

-- 3ª demostración
example :
   $C \cap (\bigcup i, A i) = (\bigcup i, C \cap A i) :=$ 
ext $ by simp

-- 4ª demostración
example :
   $C \cap (\bigcup i, A i) = (\bigcup i, C \cap A i) :=$ 
by {ext, simp}

```

4.3.7. Pruebas de $(\bigcup i, \bigcap j, A i j) \subseteq (\bigcap j, \bigcup i, A i j)$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Pruebas de  $(\bigcup i, \bigcap j, A \ i \ j) \subseteq (\bigcap j, \bigcup i, A \ i \ j)$ 
-- =====
```

```
-- Ej. 1. Demostrar
--  $(\bigcup i, \bigcap j, A \ i \ j) \subseteq (\bigcap j, \bigcup i, A \ i \ j)$ 
-- =====
```

```
import data.set
```

```
open set
```

```
variables {I J U : Type}
```

```
variables (A : I → J → set U)
```

```
-- 1ª demostración
```

```
example :  $(\bigcup i, \bigcap j, A \ i \ j) \subseteq (\bigcap j, \bigcup i, A \ i \ j) :=$ 
```

```
begin
```

```
  intros x h,
  rw mem_Union at h,
  cases h with i hi,
  rw mem_Inter at hi,
  apply mem_Inter.mpr,
  intro j,
  apply mem_Union.mpr,
  use i,
  exact (hi j),
```

```
end
```

```
-- 2ª demostración
```

```
example :  $(\bigcup i, \bigcap j, A \ i \ j) \subseteq (\bigcap j, \bigcup i, A \ i \ j) :=$ 
```

```
begin
```

```
  intros x h,
  simp * at *,
  cases h with i hi,
  intro j,
  use i,
  exact (hi j),
```

```
end
```

4.4. Conjunto potencia

4.4.1. Definición del conjunto potencia

- Enlaces al [código](#) y a la [sesión en Lean Web](#).

```
-- Definición del conjunto potencia
-- =====

variable {U : Type}

def powerset (A : set U) : set (set U) :=
  {B : set U | B ⊆ A}

example
  (A B : set U)
  (h : B ∈ powerset A)
  : B ⊆ A :=
h
```

4.4.2. Pruebas de $A \in \wp(A \cup B)$

- Enlaces al [código](#) y a la [sesión en Lean Web](#).

```
-- Pruebas de  $A \in \wp(A \cup B)$ 
-- =====

import data.set
open set

variable {U : Type}
variables (A B : set U)

#reduce powerset A
#reduce B ∈ powerset A
#reduce ∅ A
#reduce B ∈ ∅ A

-- ?ª demostración
example : A ∈ ∅ (A ∪ B) :=
begin
```

```

intros x h,
simp,
left,
exact h,
end

-- ?ª demostración
example : A ∈ ∅ (A ∪ B) :=
begin
  intros x h,
  exact or.inl h,
end

-- ?ª demostración
example : A ∈ ∅ (A ∪ B) :=
λ x, or.inl

-- ?ª demostración
example : A ∈ ∅ (A ∪ B) :=
assume x,
assume : x ∈ A,
show x ∈ A ∪ B, from or.inl <x ∈ A>

```

4.4.3. Monotonía del conjunto potencia: $\wp A \subseteq \wp B \leftrightarrow A \subseteq B$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Monotonía del conjunto potencia:  $\wp A \subseteq \wp B \leftrightarrow A \subseteq B$ 
-- =====

import data.set
open set

variable {U : Type}
variables {A B C : set U}

-- #reduce  $\wp A$ 
-- #reduce  $B \in \wp A$ 

-- -----
-- Ej. 1. Demostrar

```

```

--       $\wp A \subseteq \wp B \rightarrow A \subseteq B$ 
-----

-- 1ª demostración
example :  $\wp A \subseteq \wp B \rightarrow A \subseteq B :=$ 
begin
  intro h,
  apply subset_of_mem_powerset,
  apply h,
  apply mem_powerset,
  exact subset.rfl,
end

-- 2ª demostración
example :  $\wp A \subseteq \wp B \rightarrow A \subseteq B :=$ 
begin
  intro h,
  apply h,
  exact subset.rfl,
end

-- 3ª demostración
example :  $\wp A \subseteq \wp B \rightarrow A \subseteq B :=$ 
begin
  intro h,
  exact (h subset.rfl),
end

-- 4ª demostración
example :  $\wp A \subseteq \wp B \rightarrow A \subseteq B :=$ 
λ h, h subset.rfl

-- 5ª demostración
example :  $\wp A \subseteq \wp B \rightarrow A \subseteq B :=$ 
assume h1 :  $\wp A \subseteq \wp B$ ,
have h2 :  $A \subseteq A$ , from subset.rfl,
have h3 :  $A \in \wp A$ , from h2,
have h4 :  $A \in \wp B$ , from h1 h3,
show  $A \subseteq B$ , from h4

-- 6ª demostración
example :  $\wp A \subseteq \wp B \rightarrow A \subseteq B :=$ 
assume h1 :  $\wp A \subseteq \wp B$ ,
have h2 :  $A \subseteq A$ , from subset.rfl,

```

```

have h3 : A ∈ ∅ A, from h2,
h1 h3

-- 7ª demostración
example : ∅ A ⊆ ∅ B → A ⊆ B :=
assume h1 : ∅ A ⊆ ∅ B,
have h2 : A ⊆ A, from subset.rfl,
h1 h2

-- 8ª demostración
example : ∅ A ⊆ ∅ B → A ⊆ B :=
assume h1 : ∅ A ⊆ ∅ B,
h1 subset.rfl

-- 9ª demostración
lemma aux1 : ∅ A ⊆ ∅ B → A ⊆ B :=
λ h, h subset.rfl

-- 10ª demostración
example : ∅ A ⊆ ∅ B → A ⊆ B :=
powerset_mono.mp

-----

-- Ej. 2. Demostrar
--   A ⊆ B → ∅ A ⊆ ∅ B
-----

-- 1ª demostración
example : A ⊆ B → ∅ A ⊆ ∅ B :=
begin
  intro h,
  intros C hCA,
  apply mem_powerset,
  apply subset.trans hCA h,
end

-- 2ª demostración
example : A ⊆ B → ∅ A ⊆ ∅ B :=
begin
  intros h C hCA,
  apply subset.trans hCA h,
end

-- 3ª demostración

```

```

lemma aux2 : A ⊆ B → ∅ A ⊆ ∅ B :=
λ h C hCA, subset.trans hCA h

-- 4ª demostración
example : A ⊆ B → ∅ A ⊆ ∅ B :=
powerset_mono.mpr

-----

-- Ej. 3. Demostrar
--   ∅ A ⊆ ∅ B ↔ A ⊆ B
-----

-- 1ª demostración
example : ∅ A ⊆ ∅ B ↔ A ⊆ B :=
iff.intro aux1 aux2

-- 2ª demostración
example : ∅ A ⊆ ∅ B ↔ A ⊆ B :=
-- by library_search
powerset_mono

-- 3ª demostración
example : ∅ A ⊆ ∅ B ↔ A ⊆ B :=
-- by hint
by finish

-- 4ª demostración
example : ∅ A ⊆ ∅ B ↔ A ⊆ B :=
by simp

```


Capítulo 5

Relaciones

5.1. Relaciones de orden

5.1.1. Las irreflexivas y transitivas son asimétricas

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Las irreflexivas y transitivas son asimétricas
-- =====

-- -----
-- Ej. 1. Demostrar que las relaciones irreflexivas y
-- transitivas son asimétricas.
-- -----

variable A : Type
variable R : A → A → Prop

-- #reduce irreflexive R
-- #reduce transitive R

-- 1ª demostración
example
  (h1 : irreflexive R)
  (h2 : transitive R)
  : ∀ x y, R x y → ¬ R y x :=
begin
  intros x y h3 h4,
  apply h1 x,
  apply h2 h3 h4,
end
```

```

-- 2ª demostración
example
  (h1 : irreflexive R)
  (h2 : transitive R)
  :  $\forall x y, R x y \rightarrow \neg R y x :=$ 
begin
  intros x y h3 h4,
  apply (h1 x) (h2 h3 h4),
end

-- 3ª demostración
example
  (h1 : irreflexive R)
  (h2 : transitive R)
  :  $\forall x y, R x y \rightarrow \neg R y x :=$ 
 $\lambda x y h3 h4, (h1 x) (h2 h3 h4)$ 

-- 4ª demostración
example
  (h1 : irreflexive R)
  (h2 : transitive R)
  :  $\forall x y, R x y \rightarrow \neg R y x :=$ 
assume x y,
assume h3 : R x y,
assume h4 : R y x,
have h5 : R x x, from h2 h3 h4,
have h6 :  $\neg R x x$ , from h1 x,
show false, from h6 h5

```

5.1.2. Las partes estrictas son irreflexivas

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Las partes estrictas son irreflexivas
-- =====

-- -----
-- Ej. 1. La parte estricta de una relación R es la
-- relación R' definida por
--    $R' a b := R a b \wedge a \neq b$ 
--
-- Demostrar que la parte estricta de cualquier

```

```

-- relación es irreflexiva.
-----

import tactic

section

parameter {A : Type}
parameter (R : A → A → Prop)

definition R' (a b : A) : Prop :=
  R a b ∧ a ≠ b

#reduce irreflexive R

-- 1ª demostración
example :
  irreflexive R' :=
begin
  intros a h,
  cases h with h1 h2,
  apply h2,
  refl,
end

-- 2ª demostración
example :
  irreflexive R' :=
assume a,
assume : R' a a,
have a ≠ a, from and.right this,
have a = a, from rfl,
show false, from <a ≠ a> <a = a>

end

```

5.1.3. Las partes estrictas de los órdenes parciales son transitivas

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Las partes estrictas de los órdenes parciales son transitivas
-- =====

-----
-- Ej. 1. La parte estricta de una relación R es la
-- relación R' definida por
--   R' a b := R a b ∧ a ≠ b
--
-- Demostrar que si R es un orden parcial, entonces su
-- parte estricta es transitiva.
-- -----

import tactic

section

parameter {A : Type}
parameter (R : A → A → Prop)
parameter (reflR   : reflexive R)
parameter (transR  : transitive R)
parameter (antisimR : anti_symmetric R)
variables {a b c : A}

definition R' (a b : A) : Prop :=
  R a b ∧ a ≠ b

include transR
include antisimR

-- 1ª demostración
example : transitive R' :=
begin
  rintros a b c ⟨h1,h2⟩ ⟨h3,h4⟩,
  split,
  { apply (transR h1 h3), },
  { intro h5,
    apply h4,
    apply (antisimR h3),
    rw ←h5,
    exact h1, },
end

-- 2ª demostración
-- =====

```

```

local infix ≤ := R
local infix < := R'

example : transitive (<) :=
  assume a b c,
  assume h1 : a < b,
  assume h2 : b < c,
  have a ≤ b, from and.left h1,
  have a ≠ b, from and.right h1,
  have b ≤ c, from and.left h2,
  have b ≠ c, from and.right h2,
  have a ≤ c, from transR <a ≤ b> <b ≤ c>,
  have a ≠ c, from
    assume : a = c,
    have c ≤ b, from eq.subst <a = c> <a ≤ b>,
    have b = c, from antisimR <b ≤ c> <c ≤ b>,
    show false, from <b ≠ c> <b = c>,
  show a < c, from and.intro <a ≤ c> <a ≠ c>

end

```

5.1.4. Las partes simétricas de las reflexivas son reflexivas

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Las partes simétricas de las reflexivas son reflexivas
-- =====

-- -----
-- Ej. 1. La parte simétrica de una relación R es la
-- relación S definida por
--   S x y := R x y ∧ R y x
--
-- Demostrar que la parte simétrica de una relación
-- reflexiva es reflexiva.
-- -----

section

parameter A : Type
parameter R : A → A → Prop

```

```

parameter reflR : reflexive R

include reflR

def S (x y : A) := R x y ∧ R y x

-- 1ª demostración
example : reflexive S :=
begin
  intro x,
  split,
  { exact (reflR x), },
  { exact (reflR x), },
end

-- 2ª demostración
example : reflexive S :=
assume x,
have R x x, from reflR x,
show S x x, from and.intro this this

-- 3ª demostración
example : reflexive S :=
assume x,
show S x x, from and.intro (reflR x) (reflR x)

-- 4ª demostración
example : reflexive S :=
assume x,
and.intro (reflR x) (reflR x)

-- 5ª demostración
example : reflexive S :=
λ x, ⟨reflR x, reflR x⟩

end

```

5.1.5. Las partes simétricas son simétricas

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Las partes simétricas son simétricas
-- =====

-----

-- Ej. 1. La parte simétrica de una relación R es la
-- relación S definida por
--    $S\ x\ y := R\ x\ y \wedge R\ y\ x$ 
--
-- Demostrar que la parte simétrica de cualquier
-- relación es simétrica.
-----

section
parameter A : Type
parameter R : A → A → Prop

def S (x y : A) := R x y ∧ R y x

-- 1ª demostración
example : symmetric S :=
begin
  intros x y h,
  split,
  { exact h.right, },
  { exact h.left, },
end

-- 2ª demostración
example : symmetric S :=
begin
  intros x y h,
  exact ⟨h.right, h.left⟩,
end

-- 3ª demostración
example : symmetric S :=
λ x y h, ⟨h.right, h.left⟩

-- 4ª demostración
example : symmetric S :=
assume x y,
assume h : S x y,
have h1 : R x y, from h.left,
have h2 : R y x, from h.right,
show S y x, from ⟨h2, h1⟩

```

```

-- 5ª demostración
example : symmetric S :=
assume x y,
assume h : S x y,
show S y x, from ⟨h.right, h.left⟩

-- 6ª demostración
example : symmetric S :=
λ x y h, ⟨h.right, h.left⟩

end

```

5.2. Órdenes sobre números

5.2.1. Pruebas de $n + 1 \leq m \vdash n < m + 1$

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Pruebas de  $n + 1 \leq m \vdash n < m + 1$ 
-- =====

-- -----
-- Ej. 1. Demostrar que si  $n$  y  $m$  son números naturales
-- tales que  $n + 1 \leq m$ , entonces  $n < m + 1$ .
-- -----

import tactic

variables n m : ℕ

-- 1ª demostración
example
  (h : n + 1 ≤ m)
  : n < m + 1 :=
calc
  n   < n + 1 : lt_add_one n
  ... ≤ m      : h
  ... < m + 1 : lt_add_one m

-- 2ª demostración
example

```



```

(h : n + 1 ≤ m)
: n < m + 1 :=
have h1 : n < n + 1, from lt_add_one n,
have h2 : n < m,      from lt_of_lt_of_le h1 h,
have h3 : m < m + 1, from lt_add_one m,
show n < m + 1,      from lt.trans h2 h3

-- 3ª demostración
example
(h : n + 1 ≤ m)
: n < m + 1 :=
begin
  apply lt_trans (lt_add_one n),
  apply lt_of_le_of_lt h (lt_add_one m),
end

-- 4ª demostración
example
(h : n + 1 ≤ m)
: n < m + 1 :=
lt_trans (lt_add_one n) (lt_of_le_of_lt h (lt_add_one m))

-- 5ª demostración
example
(h : n + 1 ≤ m)
: n < m + 1 :=
-- by suggest
nat.lt.step h

-- 6ª demostración
example
(h : n + 1 ≤ m)
: n < m + 1 :=
-- by hint
by omega

-- 7ª demostración
example
(h : n + 1 ≤ m)
: n < m + 1 :=
by linarith

-- 8ª demostración
example
(h : n + 1 ≤ m)

```

```

: n < m + 1 :=
by nlinarith

```

5.3. Relaciones de equivalencia

5.3.1. Las equivalencias son preórdenes simétricos

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Las equivalencias son preórdenes simétricos
-- =====

-- -----
-- Ej. 1. Un preorden es una relación reflexiva y
-- transitiva.
--
-- Demostrar que las relaciones de equivalencias son
-- los órdenes simétricos.
-- -----

import tactic

variable {A : Type}
variable R : A → A → Prop

def preorden (R : A → A → Prop) : Prop :=
  reflexive R ∧ transitive R

-- #print equivalence
-- #print symmetric

-- 1ª demostración
example :
  equivalence R ↔ preorden R ∧ symmetric R :=
begin
  split,
  { rintros ⟨h1, h2, h3⟩,
    exact ⟨⟨h1, h3⟩, h2⟩, },
  { rintros ⟨⟨h1, h3⟩, h2⟩,
    exact ⟨h1, h2, h3⟩, },
end

```

```

-- 2ª demostración
example :
  equivalence R ↔ preorden R ∧ symmetric R :=
  ⟨λ ⟨h1, h2, h3⟩, ⟨⟨h1, h3⟩, h2⟩,
   λ ⟨⟨h1, h3⟩, h2⟩, ⟨h1, h2, h3⟩⟩

-- 3ª demostración
example :
  equivalence R ↔ preorden R ∧ symmetric R :=
  iff.intro
    ( assume h1 : equivalence R,
      have h2 : reflexive R, from and.left h1,
      have h3 : symmetric R, from and.left (and.right h1),
      have h4 : transitive R, from and.right (and.right h1),
      show preorden R ∧ symmetric R,
        from and.intro (and.intro h2 h4) h3)
    ( assume h1 : preorden R ∧ symmetric R,
      have h2 : preorden R, from and.left h1,
      show equivalence R,
        from and.intro (and.left h2)
          (and.intro (and.right h1) (and.right h2)))

-- 4ª demostración
example :
  equivalence R ↔ preorden R ∧ symmetric R :=
  begin
    unfold equivalence preorden,
    tauto,
  end

-- 5ª demostración
example :
  equivalence R ↔ preorden R ∧ symmetric R :=
  by finish [preorden]

```

5.3.2. Las relaciones reflexivas y euclídeas son de equivalencia

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- Las relaciones reflexivas y euclídeas son de equivalencia
-- =====

```

```

-- -----
-- Una relación binaria ( $\approx$ ) es euclídea si
--    $\forall \{a\ b\ c\},\ a \approx b \rightarrow c \approx b \rightarrow a \approx c$ 
--
-- El objetivo de esta teoría es demostrar que si una
-- relación es reflexiva y euclídea, entonces es de
-- equivalencia.
-- -----

import tactic

section

parameter {A : Type}
parameter (R : A → A → Prop)

local infix  $\approx$  := R

parameter reflexivaR : reflexive ( $\approx$ )
parameter euclideanR :  $\forall \{a\ b\ c\},\ a \approx b \rightarrow c \approx b \rightarrow a \approx c$ 

include reflexivaR euclideanR

-- -----
-- Ej. 1. Demostrar que las relaciones reflexivas y
-- y euclídeas son simétricas.
-- -----

-- 1ª demostración
example : symmetric ( $\approx$ ) :=
begin
  intros a b h,
  exact euclideanR (reflexivaR b) h,
end

-- 2ª demostración
example : symmetric ( $\approx$ ) :=
 $\lambda$  a b h, euclideanR (reflexivaR b) h

-- 3ª demostración
lemma simetricaR : symmetric ( $\approx$ ) :=
assume a b (h1 : a  $\approx$  b),
have h2 : b  $\approx$  b, from (reflexivaR b),
show b  $\approx$  a, from euclideanR h2 h1

```

```

-----
-- Ej. 2. Demostrar que las relaciones reflexivas y
-- y euclídeas son transitivas.
-----

-- 1ª demostración
example : transitive ( $\approx$ ) :=
begin
  rintros a b c h1 h2,
  apply euclideanR h1,
  exact euclideanR (reflexivaR c) h2,
end

-- 2ª demostración
lemma transitivaR : transitive ( $\approx$ ) :=
 $\lambda$  a b c h1 h2, (euclideanR h1) (euclideanR (reflexivaR c) h2)

-- 3ª demostración
example : transitive ( $\approx$ ) :=
assume a b c (h1 : a  $\approx$  b) (h2 : b  $\approx$  c),
have h3 : c  $\approx$  b, from euclideanR (reflexivaR c) h2,
show a  $\approx$  c, from euclideanR h1 h3

-----
-- Ej. 3. Demostrar que las relaciones reflexivas y
-- y euclídeas son de equivalencia.
-----

-- 1ª demostración
example : equivalence ( $\approx$ ) :=
begin
  unfold equivalence,
  exact  $\langle$ reflexivaR, simetricaR, transitivaR $\rangle$ ,
end

-- 2ª demostración
example : equivalence ( $\approx$ ) :=
 $\langle$ reflexivaR, simetricaR, transitivaR $\rangle$ 

end

```


Capítulo 6

Funciones

6.1. Funciones en Lean

6.1.1. Definición de la composición de funciones

- Enlaces al [código](#) y a la [sesión en Lean Web](#).

```
-- Definición de la composición de funciones
-- =====

-----

-- Ej. 1. Definir la composición de dos funciones.
-----

namespace reservado

variables {X Y Z : Type}

def comp (f : Y → Z) (g : X → Y) : X → Z :=
  λ x, f (g x)

infixr `∘` := comp

end reservado

#print notation ∘
#print function.comp
```

6.1.2. Definición de la función identidad

- Enlaces al [código](#) y a la [sesión en Lean Web](#).

```
-- Definición de la función identidad
-- =====

-- -----
-- Ej. 1. Definir la función identidad.
-- -----

namespace reservado
variable {X : Type}

def id (x : X) : X :=
x

end reservado
```

6.1.3. Extensionalidad funcional

- Enlaces al [código](#), a la [sesión en Lean Web](#).

```
-- Extensionalidad funcional
-- =====

-- -----
-- Ej. 1. Sean f y g funciones de X en Y. Demostrar que
-- si
--    $\forall (x : X), f\ x = g\ x$ 
-- entonces, las funciones f y g son iguales.
-- -----

variables {X Y : Type}
variables (f g : X → Y)

example
  (h :  $\forall x, f\ x = g\ x$ )
  : f = g :=
funext h
```


6.1.4. Propiedades de la composición de funciones (elemento neutro y asociatividad)

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```
-- Propiedades de la composición de funciones
-- =====
```

```
import tactic
```

```
open function
```

```
variables {X Y Z W : Type}
```

```
-- -----
-- Ej. 1. Demostrar que
--   id ∘ f = f
-- -----
```

```
-- 1ª demostración
```

```
example
```

```
  (f : X → Y)
  : id ∘ f = f :=
```

```
begin
```

```
  ext,
```

```
  calc (id ∘ f) x = id (f x) : by rw comp_app
        ...      = f x      : by rw id.def,
```

```
end
```

```
-- 2ª demostración
```

```
example
```

```
  (f : X → Y)
  : id ∘ f = f :=
```

```
begin
```

```
  ext,
```

```
  rw comp_app,
  rw id.def,
```

```
end
```

```
-- 3ª demostración
```

```
example
```

```
  (f : X → Y)
  : id ∘ f = f :=
```

```
begin
```

```

    ext,
    rw [comp_app, id.def],
end

-- 4ª demostración
example
  (f : X → Y)
  : id ∘ f = f :=
begin
  ext,
  calc (id ∘ f) x = id (f x) : rfl
    ...         = f x      : rfl,
end

-- 5ª demostración
example
  (f : X → Y)
  : id ∘ f = f :=
rfl

-- 6ª demostración
example
  (f : X → Y)
  : id ∘ f = f :=
-- by library_search
left_id f

-- 7ª demostración
example
  (f : X → Y)
  : id ∘ f = f :=
comp.left_id f

-----
-- Ej. 2. Demostrar que
--   f ∘ id = f
-----

-- 1ª demostración
example
  (f : X → Y)
  : f ∘ id = f :=
begin
  ext,
  calc (f ∘ id) x = f (id x) : by rw comp_app

```

```

    ...           = f x           : by rw id.def,
end

-- 2ª demostración
example
  (f : X → Y)
  : f ∘ id = f :=
begin
  ext,
  rw comp_app,
  rw id.def,
end

-- 3ª demostración
example
  (f : X → Y)
  : f ∘ id = f :=
begin
  ext,
  rw [comp_app, id.def],
end

-- 4ª demostración
example
  (f : X → Y)
  : f ∘ id = f :=
begin
  ext,
  calc (f ∘ id) x = f (id x) : rfl
    ...           = f x           : rfl,
end

-- 5ª demostración
example
  (f : X → Y)
  : f ∘ id = f :=
rfl

-- 6ª demostración
example
  (f : X → Y)
  : f ∘ id = f :=
-- by library_search
right_id f

```

```

-- 7ª demostración
example
  (f : X → Y)
  : f ∘ id = f :=
comp.right_id f

-----

-- Ej. 3. Demostrar que
--   (f ∘ g) ∘ h = f ∘ (g ∘ h)
-----

-- 1ª demostración
example
  (f : Z → W)
  (g : Y → Z)
  (h : X → Y)
  : (f ∘ g) ∘ h = f ∘ (g ∘ h) :=
begin
  ext,
  calc ((f ∘ g) ∘ h) x
    = (f ∘ g) (h x)      : by rw comp_app
    ... = f (g (h x))     : by rw comp_app
    ... = f ((g ∘ h) x)   : by rw comp_app
    ... = (f ∘ (g ∘ h)) x : by rw comp_app
end

-- 2ª demostración
example
  (f : Z → W)
  (g : Y → Z)
  (h : X → Y)
  : (f ∘ g) ∘ h = f ∘ (g ∘ h) :=
begin
  ext,
  rw comp_app,
end

-- 3ª demostración
example
  (f : Z → W)
  (g : Y → Z)
  (h : X → Y)
  : (f ∘ g) ∘ h = f ∘ (g ∘ h) :=
begin

```

```

ext,
calc ((f ∘ g) ∘ h) x
      = (f ∘ g) (h x)      : rfl
...   = f (g (h x))        : rfl
...   = f ((g ∘ h) x)      : rfl
...   = (f ∘ (g ∘ h)) x    : rfl
end

-- 4ª demostración
example
  (f : Z → W)
  (g : Y → Z)
  (h : X → Y)
  : (f ∘ g) ∘ h = f ∘ (g ∘ h) :=
rfl

-- 5ª demostración
example
  (f : Z → W)
  (g : Y → Z)
  (h : X → Y)
  : (f ∘ g) ∘ h = f ∘ (g ∘ h) :=
comp.assoc f g h

```

6.1.5. Funciones inyectivas, suprayectivas y biyectivas

- Enlaces al [código](#) y a la [sesión en Lean Web](#).

```

-- Funciones inyectivas, suprayectivas y biyectivas
-- =====

-----
-- Ej. 1. Definir las funciones inyectivas,
-- suprayectivas y biyectivas.
-----

variables {X Y Z : Type}

def injective (f : X → Y) : Prop :=
∀ {x₁ x₂}, f x₁ = f x₂ → x₁ = x₂

def surjective (f : X → Y) : Prop :=

```

```

∀ y, ∃ x, f x = y

def bijective (f : X → Y) : Prop :=
  injective f ∧ surjective f

```

6.1.6. La identidad es biyectiva

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- La identidad es biyectiva
-- =====

import tactic

open function

variables {X : Type}

-- #print injective
-- #print surjective
-- #print bijective

-----

-- Ej. 1. Demostrar que la identidad es inyectiva.
-- -----

-- 1ª demostración
example : injective (@id X) :=
begin
  intros x₁ x₂ h,
  exact h,
end

-- 2ª demostración
example : injective (@id X) :=
λ x₁ x₂ h, h

-- 3ª demostración
example : injective (@id X) :=
λ x₁ x₂, id

-- 4ª demostración
example : injective (@id X) :=

```

```

assume x1 x2,
assume h : id x1 = id x2,
show x1 = x2, from h

-- 5ª demostración
example : injective (@id X) :=
--by library_search
injective_id

-- 6ª demostración
example : injective (@id X) :=
-- by hint
by tauto

-- 7ª demostración
example : injective (@id X) :=
-- by hint
by finish

-----
-- Ej. 2. Demostrar que la identidad es suprayectiva.
-----

-- 1ª demostración
example : surjective (@id X) :=
begin
  intro x,
  use x,
  exact rfl,
end

-- 2ª demostración
example : surjective (@id X) :=
begin
  intro x,
  exact ⟨x, rfl⟩,
end

-- 3ª demostración
example : surjective (@id X) :=
λ x, ⟨x, rfl⟩

-- 4ª demostración
example : surjective (@id X) :=
assume y,

```

```

show  $\exists x, id\ x = y$ , from exists.intro y rfl

-- 5ª demostración
example : surjective (@id X) :=
-- by library_search
surjective_id

-- 6ª demostración
example : surjective (@id X) :=
-- by hint
by tauto

-----
-- Ej. 3. Demostrar que la identidad es biyectiva.
-----

-- 1ª demostración
example : bijective (@id X) :=
and.intro injective_id surjective_id

-- 2ª demostración
example : bijective (@id X) :=
⟨injective_id, surjective_id⟩

-- 3ª demostración
example : bijective (@id X) :=
-- by library_search
bijective_id

```

6.1.7. La composición de funciones inyectivas es inyectiva

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al [vídeo](#).

```

-- La composición de funciones inyectivas es inyectiva
-- =====

-----
-- Ej. 1. Demostrar que la composición de dos funciones
-- inyectivas es una función inyectiva.
-----

```



```
import tactic

open function

variables {X Y Z : Type}
variable {f : X → Y}
variable {g : Y → Z}

-- 1ª demostración
example
  (Hf : injective f)
  (Hg : injective g)
  : injective (g ∘ f) :=
begin
  intros x y h,
  apply Hf,
  apply Hg,
  exact h,
end

-- 2ª demostración
example
  (Hf : injective f)
  (Hg : injective g)
  : injective (g ∘ f) :=
begin
  intros x y h,
  apply Hf,
  exact Hg h,
end

-- 3ª demostración
example
  (Hf : injective f)
  (Hg : injective g)
  : injective (g ∘ f) :=
begin
  intros x y h,
  exact Hf (Hg h),
end

-- 4ª demostración
example
  (Hf : injective f)
  (Hg : injective g)
```

```

: injective (g ∘ f) :=
λ x y h, Hf (Hg h)

-- 5ª demostración
example
  (Hf : injective f)
  (Hg : injective g)
  : injective (g ∘ f) :=
assume x y,
assume h1 : (g ∘ f) x = (g ∘ f) y,
have h2 : f x = f y, from Hg h1,
show x = y, from Hf h2

-- 6ª demostración
example
  (Hf : injective f)
  (Hg : injective g)
  : injective (g ∘ f) :=
assume x y,
assume h1 : (g ∘ f) x = (g ∘ f) y,
show x = y, from Hf (Hg h1)

-- 7ª demostración
example
  (Hf : injective f)
  (Hg : injective g)
  : injective (g ∘ f) :=
assume x y,
assume h1 : (g ∘ f) x = (g ∘ f) y,
Hf (Hg h1)

-- 8ª demostración
example
  (Hf : injective f)
  (Hg : injective g)
  : injective (g ∘ f) :=
λ x y h1, Hf (Hg h1)

-- 9ª demostración
example
  (Hg : injective g)
  (Hf : injective f)
  : injective (g ∘ f) :=
-- by library_search
injective.comp Hg Hf

```

```
-- 10ª demostración
example
  (Hg : injective g)
  (Hf : injective f)
  : injective (g ∘ f) :=
-- by hint
by tauto
```

6.1.8. La composición de funciones suprayectivas es suprayectiva

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al vídeo.

```
-- La composición de funciones suprayectivas es suprayectiva
-- =====

-- -----
-- Ej. 1. Demostrar que la composición de dos funciones
-- suprayectivas es una función suprayectiva.
-- -----

import tactic

open function

variables {X Y Z : Type}
variable {f : X → Y}
variable {g : Y → Z}

-- 1ª demostración
example
  (hf : surjective f)
  (hg : surjective g)
  : surjective (g ∘ f) :=
begin
  intro z,
  cases hg z with y hy,
  cases hf y with x hx,
  use x,
  simp,
  rw hx,
  exact hy,
```

```

end

-- 2ª demostración
example
  (hf : surjective f)
  (hg : surjective g)
  : surjective (g ∘ f) :=
begin
  intro z,
  cases hg z with y hy,
  cases hf y with x hx,
  use x,
  calc (g ∘ f) x = g (f x) : by rw comp_app
        ... = g y      : congr_arg g hx
        ... = z        : hy,
end

-- 3ª demostración
example
  (hf : surjective f)
  (hg : surjective g)
  : surjective (g ∘ f) :=
assume z,
exists.elim (hg z)
( assume y (hy : g y = z),
  exists.elim (hf y)
  ( assume x (hx : f x = y),
    have g (f x) = z, from eq.subst (eq.symm hx) hy,
    show ∃ x, g (f x) = z, from exists.intro x this))

-- 4ª demostración
example
  (hf : surjective f)
  (hg : surjective g)
  : surjective (g ∘ f) :=
-- by library_search
surjective.comp hg hf

-- 5ª demostración
example
  (hf : surjective f)
  (hg : surjective g)
  : surjective (g ∘ f) :=
λ z, exists.elim (hg z)
  (λ y hy, exists.elim (hf y)

```

```
(λ x hx, exists.intro x
  (show g (f x) = z,
    from (eq.trans (congr_arg g hx) hy))))
```

6.1.9. La composición de funciones biyectivas es biyectiva

- Enlaces al [código](#), a la [sesión en Lean Web](#) y al vídeo.

```
-- La composición de funciones biyectivas es biyectiva
-- =====

-- -----
-- Ej. 1. Demostrar que la composición de dos funciones
-- biyectivas es una función biyectiva.
-- -----

import tactic

open function

variables {X Y Z : Type}
variable {f : X → Y}
variable {g : Y → Z}

-- 1ª demostración
example
  (hf : bijective f)
  (hg : bijective g)
  : bijective (g ∘ f) :=
begin
  split,
  { apply injective.comp,
    { exact hg.left, },
    { exact hf.left, }},
  { apply surjective.comp,
    { exact hg.right, },
    { exact hf.right, }},
end

-- 2ª demostración
example
```

```

(hf : bijective f)
(hg : bijective g)
: bijective (g ∘ f) :=
begin
  split,
  { exact injective.comp hg.1 hf.1, },
  { exact surjective.comp hg.2 hf.2, },
end

-- 3ª demostración
example
  (hf : bijective f)
  (hg : bijective g)
  : bijective (g ∘ f) :=
  ⟨injective.comp hg.1 hf.1,
   surjective.comp hg.2 hf.2⟩

-- 4ª demostración
example
  (hf : bijective f)
  (hg : bijective g)
  : bijective (g ∘ f) :=
  have giny : injective g, from hg.left,
  have gsupr : surjective g, from hg.right,
  have finy : injective f, from hf.left,
  have fsupr : surjective f, from hf.right,
  show bijective (g ∘ f),
    from and.intro (injective.comp giny finy)
                  (surjective.comp gsupr fsupr)

-- 5ª demostración
example
  (hf : bijective f)
  (hg : bijective g)
  : bijective (g ∘ f) :=
  -- by library_search
  bijective.comp hg hf

-- 6ª demostración
example :
  bijective f → bijective g → bijective (g ∘ f) :=
begin
  rintros ⟨f_iny, f_supr⟩ ⟨g_iny, g_supr⟩,
  exact ⟨injective.comp g_iny f_iny,
        surjective.comp g_supr f_supr⟩,

```

```

end

-- 7ª demostración
example :
  bijective f → bijective g → bijective (g ∘ f) :=
begin
  rintros ⟨f_iny, f_supr⟩ ⟨g_iny, g_supr⟩,
  exact ⟨g_iny.comp f_iny,
        g_supr.comp f_supr⟩,
end

-- 8ª demostración
example :
  bijective f → bijective g → bijective (g ∘ f)
| ⟨f_iny, f_supr⟩ ⟨g_iny, g_supr⟩ :=
  ⟨g_iny.comp f_iny, g_supr.comp f_supr⟩

-- 9ª demostración
example :
  bijective f → bijective g → bijective (g ∘ f) :=
λ ⟨f_iny, f_supr⟩ ⟨g_iny, g_supr⟩,
  ⟨g_iny.comp f_iny, g_supr.comp f_supr⟩

```